

---

H.H. Schutte

**BASIC-programma's  
voor wetenschap en techniek**

---

42 origineel Nederlandstalige  
programma's

---

Kluwer  
Software-reeks

---



**BASIC-programma's  
voor  
wetenschap en techniek**



*Andere titels uit de Software-reeks zijn:*

E. Doppenberg/J. van 't Hof – BASIC-programma's voor elektronici

H.H. Schutte – BASIC-programma's voor huiscomputers

K. Tracton – BASIC-subroutines

M.Th.A.M. Vijftigschild – BASIC-programma's voor school en bedrijf

M.Th.A.M. Vijftigschild – BASIC-computerspellen



H.H. Schutte

# **BASIC-programma's voor wetenschap en techniek**

42 origineel Nederlandstalige programma's



**Kluwer Technische Boeken B.V. Deventer-Antwerpen**

Redactie: W.J.H. Akkermans

ISBN 90 201 1599 5

D/1983/0108/154

© 1983 Kluwer Technische Boeken B.V. Deventer

1e druk 1983

Niets uit deze uitgave mag worden verveelvoudigd en/of openbaar gemaakt door middel van druk, fotokopie, microfilm of op welke andere wijze ook, zonder voorafgaande schriftelijke toestemming van de uitgever.

No part of this book may be reproduced in any form, by print, photoprint, microfilm or any other means, without written permission from the publisher.

Ondanks alle aan de samenstelling van de tekst bestede zorg, kan noch de redactie nog de uitgever aansprakelijkheid aanvaarden voor eventuele schade, die zou kunnen voortvloeien uit enige fout, die in deze uitgave zou kunnen voorkomen.

# Woord vooraf

Een computer kan nauwkeurig rekenen. Dat is bekend. Bovendien rekent hij nog snel ook. Nu zitten in de BASIC-interpreters en -compilers een aantal rekenkundige functies ingebouwd, die het programmeren van ingewikkelde wiskundige problemen sterk vereenvoudigen.

Een computer die in BASIC kan worden geprogrammeerd is dan ook goed te gebruiken als wetenschappelijke rekenmachine. Het grote voordeel van de computer ten opzichte van die rekenmachine is dat de computer relevante gegevens op een overzichtelijke manier op een beeldscherm of regeldrukker kan weergeven.

Het doel van dit boek is tweeledig.

Het wil de reeds meer ervaren BASIC-programmeur laten zien wat de mogelijkheden van BASIC op het terrein van de wetenschap en techniek zijn.

De technicus die nog weinig of geen programmeerervaring heeft, verkrijgt door middel van dit boek een aantal directe probleemoplossingen met daarbij een groot aantal aanwijzingen die hem in staat zullen stellen ook oplossingen voor eigen wiskundige problemen te creëren.



# Inhoud

Inleiding .....	7
1. Dataparen-analyse .....	8
2. Curve fitting .....	10
3. Drie parameter curve fitting .....	14
4. Histogram .....	17
5. Verdelingsanalyse .....	20
6. Normaalverdeling .....	22
7. Inverse normaalverdeling .....	24
8. Faculteit en gammafunctie .....	25
9. Permutaties en combinaties .....	27
10. Binomiaal verdeling .....	28
11. Poisson-verdeling .....	30
12. Enkelvoudige variantie-analyse .....	32
13. Aitken-interpolatie .....	35
14. Lagrange-interpolatie .....	37
15. Voortschrijdend gemiddelde .....	39
16. Polynoom-interpolatie .....	41
17. Exponentiële interpolatie .....	44
18. Spline-interpolatie .....	50
19. Harmonische interpolatie .....	53
20. Driedimensionele interpolatie .....	55
21. Vierde orde vlak door negen punten .....	57
22. Parabolische extrapolatie .....	59
23. Polynoom evaluatie (reëel) .....	61
24. Polynoom evaluatie (complex) .....	63
25. Grenswaarde vlak Laplace .....	65
26. Polynomen vermenigvuldigen .....	68
27. Impleciete functies (Gauss/Seidel) .....	70
28. Tabel-interpolatie .....	72
29. Newton/Raphson .....	74
30. Regula falsi .....	76
31. Optimum/constraint .....	78
32. $F(X,Y)=0$ en $G(X,Y)=0$ .....	80
33. Polynomen met reële coëfficiënten (wortels) .....	82
34. Polynomen met complexe coëfficiënten (wortels) .....	84
35. Minimum, maximum, buigpunten .....	87
36. Lengte van een curve, rotatie-oppervlak en -inhoud .....	90
37. Systeem differentiaalvergelijkingen .....	92
38. Romberg/Gauss-integraal .....	95
39. Discrete Fourier-analyse .....	97
40. Fast Fourier-transformatie (FFT) .....	99
41. Systeem lineaire vergelijkingen ( + kleinste kwadraten) .....	102
42. Coëfficiënten polynomen .....	104
Appendix .....	106

# Inleiding

Dit boek bevat een selectie van 42 wiskundige problemen, opgelost met behulp van BASIC-programma's.

Bij ieder van de programma's is een korte beschrijving gevoegd, waarin aanwijzingen omtrent de bediening en de functie van het betreffende programma zijn gegeven. Na de 'programma-listing' volgt een voorbeeld van de 'computer-output', zoals die verkregen werd tijdens het testen van het programma.

Om u in staat te stellen de programma's naar eigen inzicht aan uw wensen of aan uw computer-systeem aan te passen, zijn, behalve de bij de programma's zelf gemaakte opmerkingen, ook nog een aantal zogenaamde conversietabellen opgenomen. Deze tabellen zijn in de appendix opgenomen. Het idee en de basisgegevens voor deze tabellen zijn afkomstig van de redactie van het maandblad 'Databus'. Wij kregen toestemming om het reeds in 'Databus' gepubliceerde materiaal te gebruiken voor dit boek, waarvoor onze hartelijke dank.

Voor een negental in Nederland en België veel gebruikte computers zijn in de tabellen gegevens over BASIC-statements, systeemcommando's, rekenkundige- en logische functies en string-functies opgenomen.

De programma's in dit boek zijn zoveel mogelijk in algemeen gangbaar BASIC gehouden. Dat wil zeggen dat het gebruik van die statements, die slechts op een enkele computer voorkomen, zoveel mogelijk is vermeden. De programma's zijn geschreven en getest op een PET-2000 personal computer. Met behulp van de conversietabellen kunt u snel zien of de programma's ongewijzigd op uw computer kunnen worden uitgevoerd. Ook stellen zij u in staat de programma's indien nodig op eenvoudige wijze geschikt te maken voor iedere in de tabellen genoemde computer. Mocht u desondanks problemen ondervinden, dan kunt u schrijven naar de heer H.H. Schutte, Holthuizerbrink 33, 7544 ML Enschede. Vergeet u dan vooral niet een antwoordpostzegel bij te sluiten.

Mocht u de BASIC-taal nog niet machtig zijn, dan attenderen wij u op de volgende twee uitgaven van Kluwer Technische Boeken.

BASIC en huiscomputers van K.L. Boon en

BASIC voor beginners van J.P.A. van Prooyen.

Beide boeken vormen een goede inleiding in de BASIC-taal.



# 1. Dataparen-analyse

Op de programmaregels 50 tot en met 100 kunnen de dataparen (X, Y) worden ingegeven. Dit moet worden gedaan vòòr het starten van het programma. Zoals blijkt uit het gegeven programma, behoeven niet alle dataregels te worden gebruikt. Na het laatste datapaar kan met 999,0 het einde van de reeks worden aangegeven.

Het programma voert de volgende berekeningen uit:

Het gemiddelde van X en Y

$$Gem_x = \frac{\sum X}{N} \quad Gem_y = \frac{\sum Y}{N}$$

De standaard afwijking voor X en Y

$$S_x = \sqrt{\frac{\sum(X \cdot X) - (\sum X)(\sum X)/N}{N-1}} \quad S_y = \sqrt{\frac{\sum(Y \cdot Y) - (\sum Y)(\sum Y)/N}{N-1}}$$

De variantie voor X en Y

$$V_x = \frac{\sum(X \cdot X)}{N} - \left(\frac{\sum X}{N}\right) \left(\frac{\sum X}{N}\right) \quad V_y = \frac{\sum(Y \cdot Y)}{N} - \left(\frac{\sum Y}{N}\right) \left(\frac{\sum Y}{N}\right)$$

De gemiddelde gelijkvormigheid

$$(\sum Y)/(\sum X)$$

De correlatie-coëfficiënt

$$R = \frac{(\sum X \cdot \sum Y)/N - \sum(X \cdot Y)}{(\sum X \cdot \sum X)/N - \sum(X \cdot X)} \cdot \sqrt{\left| \frac{V_x}{V_y} \right|}$$

## Programma 1

```
5 REM***X,Y DATAPAREN ANALYSE*****
10 PRINT"X EN Y IN DATALIJNEN 50-100."
20 PRINT"AFSLUITEN MET 999,0 VOOR EINDE."
30 PRINT"GEMIDDELDE,STANDAARDAFWIJKING EN"
35 PRINT"VARIANTIE VOOR X EN Y EN OOK DE"
40 PRINT"GEM.GELIJKVORMIGHEID EN"
45 PRINT"CORRELATIECOEFFICIENT WORDEN BEREKEND."
50 DATA101.3,609,103.7,626,98.6,586,99.9,594,97.2,579
   ,100.1,605,999,0
110 READ X,Y
120 IF X=999 THEN 155
125 PRINT"X=";X,"Y="Y
130 N=N+1: SX=SX+X: XX=XX+X*X
140 SY=SY+Y: YY=YY+Y*Y: XY=XY+X*Y
150 GOTO 110
155 PRINT"TYP LETTER VOOR DISPLAY"
160 GETA$: IFA$="" THEN 160: PRINT"HET AANTAL X,Y PAREN IS ";N
170 PRINT"HET GEMIDDELDE VAN X IS ";SX/N
```



```

180 PRINT"DE STANDAARDAFWIJKING VAN X IS";
190 S=SQR((XX-SX*SX/N)/(N-1))
200 PRINTS: L=XX/N-(SX/N)^2
210 PRINT"DE VARIANTIE VAN X IS",L
220 PRINT"HET GEMIDDELDE VAN Y IS ";SY/N
230 T=SQR((YY-SY*SY/N)/(N-1))
240 PRINT"DE STANDAARDAFWIJKING VAN Y IS";T: K=YY/N-(SY/N)^2
250 PRINT"DE VARIANTIE VAN Y IS ";K
260 PRINT"GEMIDDELDE GELIJKVORMIGHEID IS";SY/SX
270 M=(SX*SY/N-XY)/(SX*SX/N-XX)
280 R=M*SQR(ABS(L/K))
290 PRINT"DE CORRELATIECOEFFICIENT IS ";R
300 END

```

## Voorbeeld 1

\*\*\*X,Y DATAPAREN ANALYSE\*\*\*\*\*

X EN Y IN DATALIJNEN 50-100.  
 AFSLUITEN MET 999,0 VOOR EINDE.  
 GEMIDDELDE,STANDAARDAFWIJKING EN  
 VARIANTIE VOOR X EN Y EN OOK DE  
 GEM.GELIJKVORMIGHEID EN  
 CORRELATIECOEFFICIENT WORDEN BEREKEND.

X= 101.3	Y= 609
X= 103.7	Y= 626
X= 98.6	Y= 586
X= 99.9	Y= 594
X= 97.2	Y= 579
X= 100.1	Y= 605

TYP LETTER VOOR DISPLAY

HET AANTAL X,Y PAREN IS	6
HET GEMIDDELDE VAN X IS	100.133333
DE STANDAARDAFWIJKING VAN X IS	2.24023829
DE VARIANTIE VAN X IS	4.18221913
HET GEMIDDELDE VAN Y IS	599.833333
DE STANDAARDAFWIJKING VAN Y IS	17.057746
DE VARIANTIE VAN Y IS	242.47167
GEMIDDELDE GELIJKVORMIGHEID IS	5.99034621
DE CORRELATIECOEFFICIENT IS	.981505886

## 2. Curve fitting

Na het starten vraagt het programma de coördinaten  $x, y$  van een punt uit een curve. Na ingave vraagt het programma om het volgende punt. Dit gaat zo door totdat u voor  $x, y$  de waarde 0,999 hebt ingegeven. Het minimum aantal op te geven coördinaten is drie, het maximum is 100.

Het programma kent de volgende curves:

#1 lineair	$y = B \cdot x + A$
#2 exponentieel	$y = A \cdot e^{Bx}$
#3 exponentieel	$y = A \cdot e^{B/x}$
#4 macht van $x$	$y = A \cdot x^B$
#5 logaritmisch	$y = A + Bx \log(x)$
#6 hyperbolisch	$y = \frac{B}{x} + A$
#7 normaalverdeling	$y = A \cdot e^{-x^2 B}$

De gevonden curves worden door middel van de kleinste kwadraten methode bepaald. De best passende curve wordt bepaald door de gevonden correlatiecoëfficiënten van de zeven functies met elkaar te vergelijken.

Indien men hierna op de boodschap 'VOOR ALLE FUNCTIES TYP J' de letter J intikt, geeft het programma de berekende waarden van A en B en de correlatiecoëfficiënt voor ieder van de functies #1 tot en met #7. Na iedere berekening vraagt het systeem u een willekeurige letter in te tikken.

Na de laatste berekening komt de boodschap 'VOOR INTERPOLATIE TYP J'; wanneer de letter J wordt ingegeven, komt de boodschap 'TYP CYFER VOOR TYPE FUNCTIE:'. U kunt nu één van de functies #1 t/m #7 kiezen, waarna interpolatie van de dataparen  $x, y$  volgt.

### Programma 2

```
10 PRINT "***CURVE FITTING POS. DATAPAREN***"
20 A$(1)="#1 LINIAIR      Y=B*X+A"
30 A$(2)="#2 EXPONENTIEEL Y=A*EXP(B*X)"
40 A$(3)="#3 EXPONENTIEEL Y=A*EXP(B/X)"
50 A$(4)="#4 MACHT VAN X  Y=A*X^B"
60 A$(5)="#5 LOGARITMISCH Y=A+B*LOG(X)"
70 A$(6)="#6 HYPERBOLISCH Y=B/X+A"
75 A$(7)="#7 NORM. VERD.  Y=A*EXP(-X*X*B)"
80 PRINT:FOR K=1 TO 7:PRINT A$(K):NEXT K
90 PRINT:PRINT "NA INPUT DATA TYP 0,999 VOOR START"
100 DIM X(100), Y(100):N=0:SY=0: SX=0: YX=0: XX=0: YY=0
110 FOR I=1 TO 100
120 PRINT "TYP X,Y VOOR PUNT":I;
125 INPUT X(I),Y(I)
130 IF Y(I)=999 AND X(I)=0 THEN I=100:GOTO 300
135 IF X(I)=0 THEN X(I)=X(I)+.0000001
140 N=I:NEXT I
```



```

150 SY=0: SX=0: YX=0: YY=0: XX=0
155 FOR I=1 TO N
160 SY=SY+Y(I): SX=SX+X(I): YX=YX+Y(I)*X(I): YY=YY+Y(I)*Y(I)
170 XX=XX+X(I)*X(I)
180 NEXT I
220 R2=YX-SX*SY/N
230 R3=XX-SX*SX/N
240 R1=R2/R3
250 R4=SY/N-R1*SX/N
260 R5=R2*R2/(YY-SY*SY/N)/R3: RETURN
300 IF N<3 THEN 1000
305 GOSUB 150
310 B(1)=R1: A(1)=R4: C(1)=R5
320 FOR K=1 TO N: Y(K)=LOG(Y(K)): NEXT K
330 GOSUB 150
340 B(2)=R1: A(2)=EXP(R4): C(2)=R5
350 FOR K=1 TO N: X(K)=1/X(K): NEXT K
360 GOSUB 150
370 B(3)=R1: A(3)=EXP(R4): C(3)=R5
380 FOR K=1 TO N: X(K)=1/X(K)
390 X(K)=LOG(X(K)): NEXT K: GOSUB 150
400 B(4)=R1: A(4)=EXP(R4): C(4)=R5
420 FOR K=1 TO N: Y(K)=EXP(Y(K)): NEXT K
430 GOSUB 150
440 B(5)=R1: A(5)=R4: C(5)=R5
441 FOR K=1 TO N: X(K)=1/EXP(X(K)): NEXT K
442 GOSUB 150
443 B(6)=R1: A(6)=R4: C(6)=R5
445 FOR K=1 TO N: X(K)=1/X(K): X(K)=-X(K)*X(K): Y(K)=LOG(Y(K)): NEXT K
446 GOSUB 150
447 B(7)=R1: A(7)=EXP(R4): C(7)=R5
450 P=0: FOR K=1 TO 7
460 IF ABS(C(K))>P THEN Q=K: P=ABS(C(K))
470 NEXT K
480 PRINT: PRINT "DE BESTPASSENDE CURVE IS:"
490 PRINT: PRINT A(Q): PRINT
500 PRINT "A= "; A(Q)
510 PRINT "B= "; B(Q)
515 PRINT "TYPE FUNCTIE IS NR. "; Q
520 PRINT "CORRELATIE COEFFICIENT IS ";
530 PRINT C(Q): PRINT
540 PRINT "VOOR ALLE FUNCTIES TYP J"
550 INPUT B$: IF B$<>"J" THEN END
560 FOR K=1 TO 7: D$=""
570 PRINT: PRINT A$(K): PRINT "A= "; A(K): PRINT "B= "; B(K)
580 PRINT "CORR. COEFF. = "; C(K)
581 PRINT "TYP EEN LETTER"
585 GET D$: IF D$="" THEN 585
586 D$="": NEXT K
590 PRINT: PRINT "VOOR INTERPOLATIE TYP J"
600 INPUT C$: IF C$<>"J" THEN END
610 PRINT "TYP CYFER VOOR TYPE FUNCTIE: ";
620 INPUT U
630 ON U GOSUB 650, 700, 750, 800, 850, 900, 950
650 INPUT "X-WAARDE "; X

```



```

660 Y=B(U)*X+A(U)
670 PRINT"Y-WAARDE ";Y:GOTO650
700 INPUT"X-WAARDE ";X
710 Y=A(U)*EXP(B(U)*X)
720 PRINT"Y-WAARDE ";Y:GOTO700
750 INPUT"X-WAARDE ";X
760 Y=A(U)*EXP(B(U)/X)
770 PRINT"Y-WAARDE ";Y:GOTO750
800 INPUT"X-WAARDE ";X
810 Y=A(U)*(X↑B(U))
820 PRINT"Y-WAARDE ";Y:GOTO800
850 INPUT"X-WAARDE ";X
860 Y=A(U)+B(U)*LOG(X)
870 PRINT"Y-WAARDE ";Y:GOTO850
900 INPUT"X-WAARDE ";X
910 Y=B(U)/X+A(U)
920 PRINT"Y-WAARDE ";Y:GOTO 900
950 INPUT"X-WAARDE ";X
960 Y=A(U)*EXP(-X*X*B(U))
970 PRINT"Y-WAARDE ";Y
980 GOTO 950
1000 PRINT"TE WEINIG DATA. MINIMUM 3!"

```

## Voorbeeld 2

\*\*\*CURVE FITTING POS. DATAPAREN\*\*\*

```

#1 LINIAIR      Y=B*X+A
#2 EXPONENTIEEL Y=A*EXP(B*X)
#3 EXPONENTIEEL Y=A*EXP(B/X)
#4 MACHT VAN X  Y=A*X↑B
#5 LOGARITMISCH Y=A+B*LOG(X)
#6 HYPERBOLISCH Y=B/X+A
#7 NORM.VERD.   Y=A*EXP(-X*X*B)

```

```

NA INPUT DATA TYP 0,999 VOOR START
TYP X,Y VOOR PUNT 1 3 7.1
TYP X,Y VOOR PUNT 2 5 11
TYP X,Y VOOR PUNT 3 10 21.05
TYP X,Y VOOR PUNT 4 12 24.9
TYP X,Y VOOR PUNT 5 0 999

```

DE BESTPASSENDE CURVE IS:

```
#1 LINIAIR      Y=B*X+A
```

A= 1.12216985

B= 1.98537736

TYPE FUNCTIE IS NR.: 1

CORRELATIECOEFFICIENT IS .999949539

VOOR ALLE FUNCTIES TYP J

J

#1 LINIAIR  $Y=B*X+A$   
 A= 1.12216985  
 B= 1.98537736  
 CORR.COEFF.= .999949539  
 TYP EEN LETTER

#2 EXPONENTIEEL  $Y=A*EXP(B*X)$   
 A= 5.08506947  
 B= .137150789  
 CORR.COEFF.= .981215082  
 TYP EEN LETTER

#3 EXPONENTIEEL  $Y=A*EXP(B/X)$   
 A= 34.6679941  
 B= -4.9723384  
 CORR.COEFF.= .964909755  
 TYP EEN LETTER

#4 MACHT VAN X  $Y=A*X^B$   
 A= 2.58790552  
 B= .90938894  
 CORR.COEFF.= .999605819  
 TYP EEN LETTER

VOOR INTERPOLATIE TYP J  
 J

TYP CYFER VOOR TYPE FUNCTIE: 1

#5 LOGARITMISCH  $Y=A+B*LOG(X)$   
 A= -8.13400535  
 B= 12.8857956  
 CORR.COEFF.= .976059458  
 TYP EEN LETTER

X-WAARDE 5  
 Y-WAARDE 11.0490566  
 X-WAARDE 10  
 Y-WAARDE 20.9759434  
 X-WAARDE 12  
 Y-WAARDE 24.9466981  
 X-WAARDE 15  
 Y-WAARDE 30.9028302  
 X-WAARDE 3  
 Y-WAARDE 7.07830191  
 X-WAARDE 3.5  
 Y-WAARDE 8.07099059  
 X-WAARDE 4  
 Y-WAARDE 9.06367926  
 X-WAARDE 0  
 Y-WAARDE 1.12216985

#6 HYPERBOLISCH  $Y=B/X+A$   
 A= 28.3429948  
 B= -68.8213661  
 CORR.COEFF.= .898951536  
 TYP EEN LETTER

#7 NORM. VERD.  $Y=A*EXP(-X*X*B)$   
 A= 7.69507277  
 B= -8.83970334E-03  
 CORR.COEFF.= .93219239  
 TYP EEN LETTER

### 3. Drie parameter curve fitting

De parameters  $X$ ,  $Y$  en  $Z$  kunnen maximaal 100 keer worden ingegeven. Na de boodschap 'TYP X,Y,Z VOOR PUNT N' kunt u  $X$ ,  $Y$  en  $Z$  van elkaar gescheiden door een komma ingeven. Daarna komt dezelfde boodschap weer, nu voor het volgende punt. Dit gaat zo door tot u de waarden 9,9,9 hebt ingegeven.

Nu worden de best passende waarden voor  $A_0$ ,  $A_1$  en  $A_2$  berekend, voor de functie:

$$F_Z = A_0 + A_1 * F_X + A_2 * F_Y$$

Na de boodschap 'INTERPOLATIE, TYP X,Y' kunt u een waarde voor  $X$  en  $Y$  opgeven, waarna de bijbehorende  $Z$ -waarde door middel van interpolatie wordt berekend.

Om het programma berekeningen voor een andere curve te laten verrichten, moeten regel 20 en eventueel de regels 50, 60, 70 en 360 worden gewijzigd, al naar gelang de gewenste curve. Het bijgevoegde voorbeeld laat dit zien.

#### Programma 3

```
5 DIMX(100),Y(100),Z(100)
6 PRINT"*****"
10 PRINT"3 PARAMETER CURVEFITTING LSQ"
12 PRINT"*****"
20 PRINT"VORM : FZ=A0+A1*FX+A2*FY"
30 PRINT:PRINT"FUNCTIE: X(I)=F(X(I)) LIJN 50:RETURN"
35 PRINT"FUNCTIE: Y(I)=F(Y(I)) LIJN 60:RETURN"
40 PRINT"FUNCTIE: Z(I)=F(Z(I)) LIJN 70:RETURN"
41 PRINT"FUNCTIE INVERSE Z(I) LIJN 360"
45 GOTO100
50 X(I)=X(I):RETURN
60 Y(I)=Y(I):RETURN
70 Z(I)=Z(I):RETURN
100 PRINT"VOOR EINDE INPUT TYP 9,9,9"
110 FORI=1TO100
120 PRINT"TYP X,Y,Z VOOR PUNT ";I;
130 INPUT X(I),Y(I),Z(I)
140 IFX(I)=9ANDY(I)=9ANDZ(I)=9THENI=100:GOTO170
150 GOSUB50:GOSUB60:GOSUB70
160 R5=I:NEXT
170 FORI=1TOR5
180 R1=R1+X(I)*Y(I):R2=R2+X(I)*Z(I)
190 R3=R3+Y(I)*Z(I):R4=R4+Y(I)*Y(I)
200 R6=R6+X(I)*X(I):R7=R7+X(I)
210 R8=R8+Y(I):R9=R9+Z(I)
220 NEXT
230 A=(R5*R6-R7*R7)*(R5*R3-R8*R9)
240 B=(R5*R1-R7*R8)*(R5*R2-R7*R9)
```



```

250 A2=(A-B)/((R5*R6-R7*R7)*(R5*R4-R8*R8)-(R5*R1-R7*R8)
    *(R5*R1-R7*R8))
260 A1=(R5*R2-R7*R9)-A2*(R5*R1-R7*R8)
270 A1=A1/(R5*R6-R7*R7)
280 A0=(R9-A2*R8-A1*R7)/R5
290 PRINT"A0= ";A0
300 PRINT"A1= ";A1
310 PRINT"A2= ";A2
320 PRINT"INTERPOLATIE, TYP X,Y";
330 INPUT X(I),Y(I)
340 GOSUB50:GOSUB 60
350 Z(I)=A0+A1*X(I)+A2*Y(I)
360 Z(I)=Z(I):REM OF INVERSE VAN LIJN 70
370 PRINT"Z-WAARDE IS ";Z(I)
380 PRINT:GOTO320

```

### Voorbeeld 3.1

TEST OP EEN NIET LINIAIR VERBAND VAN X,Y EN Z  
 B.V. LIJN 20 WORDT  $FZ=A0+A1*FX+A2*FX*FX$  (PARABOLISCH)  
 LIJN 60 WORDT  $Y(I)=Y(I)*Y(I)$   
 DE REST VERANDERT NIET!

```

*****
3 PARAMETER CURVEFITTING LSQ
*****
VORM : FZ=A0+A1*FX+A2*FX*FX (PARABOLISCH)

FUNCTIE: X(I)=F(X(I)) LIJN 50:RETURN
FUNCTIE: Y(I)=F(Y(I)) LIJN 60:RETURN
FUNCTIE: Z(I)=F(Z(I)) LIJN 70:RETURN
FUNCTIE INVERSE Z(I) LIJN 360
VOOR EINDE INPUT TYP 9,9,9

```

TYP X,Y,Z VOOR PUNT	1	20		20	7.8
TYP X,Y,Z VOOR PUNT	2	30		30	10.6
TYP X,Y,Z VOOR PUNT	3	40		40	11.7
TYP X,Y,Z VOOR PUNT	4	50		50	10.5
TYP X,Y,Z VOOR PUNT	5	60		60	7.9
TYP X,Y,Z VOOR PUNT	6	70		70	5.1
TYP X,Y,Z VOOR PUNT	7	9		9	9

A0= -1.38285717  
 A1= .622714287  
 A2= -7.64285716E-03

INTERPOLATIE, TYP X,Y 30 30  
 Z-WAARDE IS 10.42

INTERPOLATIE, TYP X,Y 25 25  
 Z-WAARDE IS 9.40821429

INTERPOLATIE, TYP X,Y 45	45
Z-WAARDE IS 11.1625	
INTERPOLATIE, TYP X,Y 65	65
Z-WAARDE IS 6.8025	

### Voorbeeld 3.2

VOORBEELD VAN LISTING  
 TOEVOEGING STOF 1 IS INPUT X=1;TOEVOEGING STOF 2, DAN Y=2  
 GEEN TOEVOEGING X EN Y ZIJN BEIDE NUL, GEEN STOF 1 X=0  
 GEEN STOF 2 DAN Y=0  
 HET RESULTAAT Z IS DE INVOER VAN DE GEVONDEN WAARDE  
 HET PRG. GEEFT HET LINEAIRE VERBAND MET PARAMETERS (LIJN 20)  
 DUS VOOR DE 9 MOGELIJKE COMBINATIES

```
*****
3 PARAMETER CURVEFITTING LSQ
*****
VORM : FZ=A0+A1*FX+A2*FY
```

```
FUNCTIE: X(I)=F(X(I)) LIJN 50:RETURN
FUNCTIE: Y(I)=F(Y(I)) LIJN 60:RETURN
FUNCTIE: Z(I)=F(Z(I)) LIJN 70:RETURN
FUNCTIE INVERSE Z(I) LIJN 360
VOOR EINDE INPUT TYP 9,9,9
```

TYP X,Y,Z VOOR PUNT	1	0	0	18.5
TYP X,Y,Z VOOR PUNT	2	0	1	17.2
TYP X,Y,Z VOOR PUNT	3	0	2	18.6
TYP X,Y,Z VOOR PUNT	4	1	0	18.3
TYP X,Y,Z VOOR PUNT	5	1	1	19.2
TYP X,Y,Z VOOR PUNT	6	1	2	19.7
TYP X,Y,Z VOOR PUNT	7	2	0	19.5
TYP X,Y,Z VOOR PUNT	8	2	1	19.8
TYP X,Y,Z VOOR PUNT	9	2	2	20.5
TYP X,Y,Z VOOR PUNT	10	9	9	9

A0= 17.7  
 A1= .91666666  
 A2= .41666666

INTERPOLATIE, TYP X,Y  
 DE TOEVOEGING STOF X HEEFT RUIM 2X ZOVEEL EFFECT ALS STOF Y

## 4. Histogram

Na het opgeven van het aantal cellen, de celbreedte en de datalimiet (minimum celwaarde), komt de boodschap 'ALS FREQ. IS 1 TYP 1'. Indien u steeds een enkele waarde wilt opgeven, dan dient u hier 1 in te geven. Wilt u de mogelijkheid om bij de in te geven waarde ook de frequentie op te geven benutten, dan kunt u het cijfer 2 ingeven.

Hierna kunt u de waarde voor  $X$  (indien een 1 was ingetypt), of de frequentie en de waarde voor  $X$ , van elkaar gescheiden door een komma, ingeven. Indien de ingegeven waarde van  $X$  te hoog of te laag is, dan geeft het programma daarvan een melding. De ingave wordt dan beschouwd als niet te zijn gedaan en de volgende waarde voor  $X$  kan worden ingegeven.

Nadat u voor  $X$  de waarde 999 hebt ingegeven, worden het gemiddelde en de standaard afwijking van de ingegeven waarden berekend. Typt u als antwoord op de boodschap 'VOOR HISTOGRAM TYP 1' een 1 in, dan volgt een grafische weergave van de ingegeven waarden.

Wanneer u tenslotte op de boodschap 'VOOR MEER TYP 1' een 1 intypt, dan wordt een cumulatief frequentiediagram weergegeven.

### Programma 4

```
10 PRINT"*****HISTOGRAM*****":PRINT
20 INPUT"AANTAL CELLEN? ";R1:DIMC(R1)
30 INPUT"CELBREEDTE? ";R2
40 INPUT"MINIMUM DATALIMIET? ";R3:N=0
50 PRINT"VOOR START HISTOGRAM TYP X=999"
55 NN=1:PRINT:GOTO 70
60 N=N-F: SX= SX-R0: XX= XX-R0*R0: NN= NN-1
70 PRINT"ALS FREQ. IS 1 TYP 1": INPUT K
71 IF K=1 THEN 77
74 PRINT"TYP FREQ.,X";NN;" IS ";
75 INPUT F,R6: IFR6=999 THEN 130
76 GOTO 90
77 PRINT"TYP X ";NN;" IS ": INPUT R6
78 IFR6=999 THEN 130
79 F=1
90 N=N+F: R0=R6*F: SX= SX+R0: XX= XX+R0*R0: NN= NN+1
100 P=R6-R3: IF P<0 THEN PRINT"TE KLEIN!": GOTO 60
110 P=INT(P/R2): IF P>R1 THEN PRINT"TE GROOT!": GOTO 60
120 C(P)=C(P)+F
122 IF K=1 THEN 77
125 GOTO 74
130 PRINT"GEMIDDELDE IS ";SX/N
140 S=SQR((XX-SX*SX/N)/(N-1))
150 PRINT"STANDAARDAFWIJKING IS ";S
155 PRINT
160 PRINT"CEL NR.  VAN/TOT          AANTAL PTN.":R5=R3:R4=R3
170 FOR I=0 TO R1-1
```



```

180 R4=R4+R2:C=C(I)
190 PRINTI+1;TAB(8);INT(R5*100+.5)/100;TAB(14);
    INT(R4*100+.5)/100;TAB(26);C
200 R5=R4:NEXTI
205 PRINTTAB(14);"TOTAAL IS ";N
210 PRINT:PRINT"VOOR HISTOGRAM TYP 1 ";INPUTA:PRINT
220 IF A=1 THEN 240
230 END
240 FORI=0TOR1-1
250 IF C(I)>MATHENMA=C(I):R=I
260 NEXTI:R4=R3
265 PRINT" X";TAB(8);"MAX.IS ";MA;" PTN."
270 PRINT"=====
280 FORI=0TOR1
290 PRINTINT(R4*100+.5)/100;PRINTTAB(8);"I"
300 PRINTTAB(8);"I";IFC(I)=0THEN315
310 PRINTTAB(9);FORJ=1TO C(I)/MA*25:PRINT"*";NEXT
315 PRINT
320 R4=R4+R2
330 NEXT
340 PRINT"VOOR MEER TYP 1";INPUTB
350 PRINT:IF B=1THEN 365
360 END
365 PRINT" FR.    CUMULATIEF FREQUENTIE DIAGRAM"
370 PRINT"=====
380 FORI=0TOR1-1
390 PRINTTAB(8);"I":Q=Q+C(I)
400 PRINTQ;TAB(8);"I";TAB(Q*25/N+8);"*"
410 NEXTI:PRINT
420 PRINT"MODE#";R+1;": ";INT((R*R2+R3)*100+.5)/100;"TOT ";
430 PRINTINT((R*R2+R3+R2)*100+.5)/100;": ";MA;"PTN."

```

## Voorbeeld 4

\*\*\*\*\*HISTOGRAM\*\*\*\*\*

AANTAL CELLEN? 6  
 CELBREEDTE? 2  
 MINIMUM DATA LIMiet? 63  
 VOOR START HISTOGRAM TYP X=999

ALS FREQ.IS 1 TYP 1 1

```

1
TYP X  1 IS  66.5
TYP X  2 IS  65.3
TYP X  3 IS  68.4
TYP X  4 IS  69
TYP X  5 IS  73.6
TYP X  6 IS  75.9
TYP X  7 IS  71
TYP X  8 IS  69.8
TYP X  9 IS  69.7
TYP X 10 IS  70.8

```

TYP X 11 IS 67.7  
 TYP X 12 IS 71.1  
 TYP X 13 IS 71.5  
 TYP X 14 IS 69.9  
 TYP X 15 IS 74.4  
 TYP X 16 IS 999  
 GEMIDDELTE IS 70.3066667  
 STANDAARDAFWIJKING IS 2.85543822

CEL NR.	VAN/TOT	AANTAL	PTN.
1	63 65		0
2	65 67		2
3	67 69		2
4	69 71		5
5	71 73		3
6	73 75		2
TOTAAL IS		15	

VOOR HISTOGRAM TYP 1 1

X	MAX.	IS	5	PTN.
63	I			
65	I			
67	I		*****	
69	I		*****	
71	I		*****	
73	I		*****	
75	I		*****	

VOOR MEER TYP 1 1

FR.	CUMULATIEF	FREQUENTIE	DIAGRAM
0	I		*
2	I		*
4	I		*
9	I		*
12	I		*
14	I		*

MODE# 4 : 69 TOT 71 : 5 PTN.

## 5. Verdelingsanalyse

Het programma vraagt u waarden voor  $X$  in te typen. Nadat u de waarde 999 hebt ingegeven, worden de volgende berekeningen uitgevoerd:

$$\text{Eerste moment} \quad M_1 = \frac{\sum_{n=1}^N \cdot X_n}{N}$$

$$\text{Tweede moment} \quad M_2 = \frac{\sum_{n=1}^N \cdot X_n^2}{N} - M_1^2$$

$$\text{Derde moment} \quad M_3 = \frac{\sum_{n=1}^N \cdot X_n^3}{N} - \frac{3M_1 \cdot \sum_{n=1}^N \cdot X_n^2}{N} + 2M_1^3$$

$$\text{Vierde moment} \quad M_4 = \frac{\sum_{n=1}^N \cdot X_n^4}{N} - \frac{4M_1 \cdot \sum_{n=1}^N \cdot X_n^3}{N} + \frac{6M_1^2 \cdot \sum_{n=1}^N \cdot X_n^2}{N} - 3M_1^4$$

$$\text{Scheefheid} \quad S = \frac{M_3}{M_2^{3/2}}$$

$$\text{Welving} \quad W = \frac{M_4}{M_2^2}$$

Tevens worden de gemiddelde waarde en de variantie aangegeven, waarna afhankelijk van de waarden van  $S$  en  $W$  nog de volgende test kan volgen:

- $S < 0$  – linkszijdig scheef
- $S = 0$  – symmetrisch
- $S > 0$  – rechtszijdig scheef
- $W < 0$  – afgeplatte verdeling
- $W = 0$  – normale verdeling
- $W > 0$  – gepiekte verdeling

### Programma 5

```
5 PRINT"*****VERDELINGSANALYSE*****"
10 REM***1E VIER MOMENTEN, SCHEEFHEID
20 REM EN WELVING VOOR X1,X2 T/M XN
30 REM VOOR START NA INPUT TYP 999
40 PRINT:N=1:X1=0:X2=0:X3=0:X4=0
50 PRINT" TYP X":N:INPUT X
60 IF X=999 THEN 100
70 X1=X1+X:X2=X2+X*X
```



```

80 X3=X3+X*X*X:X4=X4+X*X*X*X
90 N=N+1:GOTO 50
100 N=N-1:M1=X1/N:M2=(X2/N)-M1*M1
110 M3=(X3/N)-(3*M1*X2)/N+(2*M1*M1*M1)
120 M4=(X4/N)-(4*M1*X3/N)
130 M4=M4+(6*M1*M1*X2/N)-(3*M1*M1*M1*M1)
140 S=M3/(M2↑(3/2))
150 W=M4/(M2*M2)
160 PRINT"1E.MOMENT = ";M1
170 PRINT"2E.MOMENT = ";M2
180 PRINT"3E.MOMENT = ";M3
190 PRINT"4E.MOMENT = ";M4
200 PRINT"SCHEEFHEID= ";S
210 PRINT"WELVING = ";W
220 REM SCHEEFHEID <0 LINKS SYMMETRISCH
230 REM SCHEEFHEID =0 SYMMETRISCH
240 REM SCHEEFHEID >0 RECHTS SYMMETRISCH
250 REM WELVING <3 AFGEPLAT
260 REM WELVING =3 NORMAAL
270 REM WELVING >3 GEPIEKTE VERDELING
280 REM 1E.MOMENT IS GEMIDDELDE
290 REM 2E.MOMENT IS VARIANTIE
300 PRINT:PRINT"GEMIDDELDE =";M1
310 PRINT"VARIANTIE =";M2
320 IF S<0 THEN PRINT"LINKSZIJDIG SCHEEF"
330 IF S=0 THEN PRINT"SYMMETRISCH"
340 IF S>0 THEN PRINT"RECHTSZIJDIG SCHEEF"
350 IF W<3 THEN PRINT"AFGEPLATTE VERDELING"
360 IF W=3 THEN PRINT"NORMALE VERDELING"
370 IF W>3 THEN PRINT"GEPIEKTE VERDELING"
380 END

```

## Voorbeeld 5

\*\*\*\*\*VERDELINGSANALYSE\*\*\*\*\*

```

TYP X 1 25
TYP X 2 20
TYP X 3 20
TYP X 4 21
TYP X 5 22
TYP X 6 26
TYP X 7 10
TYP X 8 999
1E.MOMENT = 20.5714286
2E.MOMENT = 23.3877549
3E.MOMENT = -133.13702
4E.MOMENT = 1963.81423
SCHEEFHEID= -1.17710724
WELVING = 3.59023878

```

```

GEMIDDELDE = 20.5714286
VARIANTIE = 23.3877549
LINKSZIJDIG SCHEEF
GEPIEKTE VERDELING

```

## 6. Normaalverdeling

Na het starten biedt het programma de keuze uit drie soorten normaalverdelingen, waarvoor u de volgende ingave kunt doen:

S = standaard

GS = niet-standaard

GA = garantie

Indien u hebt gekozen voor 'GS' of 'GA', dan wordt u eerst gevraagd naar de gemiddelde waarde en de standaard afwijking.

Daarna komt de vraag naar de X-waarde, een vraag die ook wordt gesteld indien u 'S' had gekozen. De in te geven X-waarde moet groter zijn dan nul.

Het programma berekent de kans in procenten.

Hierna komt de boodschap 'VOOR MEER TYP 1'. Indien hierop het cijfer 1 wordt ingetikt, kan een nieuwe X-waarde worden opgegeven, waarop weer een kansberekening volgt.

### Programma 6

```
10 PRINT"*****NORMAAL-VERDELING*****"
20 PRINT"VOOR X-WAARDEN >=0"
30 PRINT:PRINT"INDIEN STANDAARD TYP S"
40 PRINT"EN GEEN STANDAARD TYP GS"
42 PRINT"OF INDIEN GARANTIE TYP GA";
45 INPUT A$
50 IF A$="S"THEN 80
60 INPUT"GEMIDDELDE IS";M
70 INPUT"STANDAARDAFWIJKING IS";S1
80 INPUT"X IS ";X
90 IFA$="S"THEN 110
100 X=(X-M)/S1: W=X:X=ABS(X)
110 F=EXP(-X*X/2)/SQRT(2*3.14159265)
120 T=1/(1+.2316419*X)
121 B1=.31938153:B2=-.356563782
122 B3=1.781477937:B4=-1.821255978
123 B5=1.330274429
125 Q=.5-F*(B1*T+B2*T*T+B3*(T^3)+B4*(T^4)+B5*(T^5))
130 F=INT(F*1000+.5)/1000
140 Q=INT(Q*1000+.5)/10
141 IF A$="S"THEN Q=Q+50
142 IF W<0 THEN Q=100-Q
143 IF A$="GA"THEN Q=Q-50
150 PRINT:PRINT"FUNCTIE Z(X)IS ";F
160 PRINT" DE KANS IS ";Q;"PROCENT"
170 PRINT:PRINT"VOOR MEER TYP 1";
175 INPUT B
180 IF B=1 THEN PRINT:GOTO80
190 END
```

## Voorbeeld 6.1

\*\*\*\*NORMAAL-VERDELING\*\*\*\*  
VOOR X-WAARDEN  $\geq 0$

INDIEN STANDAARD TYP S  
EN GEEN STANDAARD TYP GS  
OF INDIEN GARANTIE TYP GA? GA  
GEMIDDELDE IS? 2.2  
STANDAARDAFWIJKING IS? .45  
X IS ? 1.98

FUNCTIE Z(X) IS .354  
DE KANS IS 31.2 PROCENT

VOOR MEER TYP 1? 1

X IS ? 1.75

FUNCTIE Z(X) IS .242  
DE KANS IS 15.9 PROCENT

VOOR MEER TYP 1?

## Voorbeeld 6.2

\*\*\*\*NORMAAL-VERDELING\*\*\*\*  
VOOR X-WAARDEN  $\geq 0$

INDIEN STANDAARD TYP S  
EN GEEN STANDAARD TYP GS  
OF INDIEN GARANTIE TYP GA? S  
X IS ? 1.25

FUNCTIE Z(X) IS .183  
DE KANS IS 89.4 PROCENT

VOOR MEER TYP 1? 1

X IS ? .89

FUNCTIE Z(X) IS .268  
DE KANS IS 81.3 PROCENT

VOOR MEER TYP 1?



## 7. Inverse normaalverdeling

Na het invoeren van de waarde voor  $P_{(x)}$ , die moet liggen tussen 0,5 en 1, berekent het programma  $F_{(x)}$  (zie afbeelding 1).

Hierna kan men een nieuwe waarde voor  $P_{(x)}$  opgeven, waarna de bijbehorende  $F_{(x)}$  weer wordt berekend. Dit kan men zo vaak herhalen als men wil.

### Programma 7

```
10 PRINT"****INVERSE NORMAALVERDELING****"
20 PRINT:PRINT"P(X) IS >=.5 EN <1"
30 C0=2.515517: C1=.802853
40 C2=.010328: D1=1.432788
50 D2=.189269: D3=.001308:PRINT
60 INPUT"TYP P(X)";P
70 P=ABS(P)
80 IF P<.5 OR P>=1 THEN 60
90 Q=1-P
100 T=SQR(LOG(1/(Q*Q)))
110 A=C0+C1*T+C2*T*T
120 B=1+D1*T+D2*T*T+D3*T*T*T
130 FX=T-A/B
140 PRINT"F(X) IS ";INT(FX*100+.5)/100
150 PRINT: GOTO 60
```

### Voorbeeld 7

\*\*\*\*INVERSE NORMAALVERDELING\*\*\*\*

P(X) IS >=.5 EN <1

TYP P(X)? .55  
F(X) IS .13


TYP P(X)? .65  
F(X) IS .38

TYP P(X)? .70  
F(X) IS .52

TYP P(X)? .78  
F(X) IS .77

TYP P(X)? .95  
F(X) IS 1.65

TYP P(X)?



afb. 1

## 8. Faculteit en gammafunctie

Voor het berekenen van de  $N$  faculteit dient voor  $N$  een geheel getal te worden ingevoerd.  $N$  moet kleiner dan of gelijk aan 33 zijn. Indien voor  $N$  een niet-geheel getal wordt ingevoerd, dan moet ook de  $X$ -waarde worden opgegeven. Toegestane  $X$ -waarden zijn  $-10 \leq X \leq 33$ .

Het is mogelijk om van het berekenen van faculteiten over te gaan tot het berekenen van gammafuncties. Omgekeerd is dit niet mogelijk.

Het aantal berekeningen van faculteiten en gammafuncties is onbeperkt.

De volgende berekeningen kunnen door het programma worden uitgevoerd:

De faculteit van  $N = N!$

De gammafunctie voor  $X$ :

$$\text{Indien } X > 0 \rightarrow \Gamma_{(x)} \equiv \int_0^{\infty} t^{x-1} e^{-t} dt$$

$$\text{Indien } X < 0 \rightarrow \Gamma_{(x)} = \frac{\Gamma(X+1)}{X}$$

### Programma 8

```
10 PRINT"*****FACULTEITEN*****"
20 PRINT:INPUT"N KLEINER DAN 33 ";N
25 M=N-INT(N):F=N
30 IF M > 0 THEN 100
35 IF N<=0 THEN 100
40 FOR I=1 TO N-1
50 F=F*(N-I)
60 NEXT I: PRINT
70 PRINT"FACULTEIT ";N;" IS ";F
80 PRINT:GOTO 20
100 PRINT
105 PRINT"*****GAMMA-FUNCTIE*****"
107 PRINT"X IS GEEN GEHEEL GETAL."
110 PRINT"X IS MIN. -10 EN MAX.+33"
120 PRINT:INPUT"X IS ";R4:R7=R4
125 IF R4<-10 OR R4>33 THEN 120
130 R1=R4+ 10
140 R5=LOG(2*3.14159265*R1)/2
150 R5=R5+R1*LOG(R1)-R1
160 R6=1/(12*R1)-1/(360*R1*R1*R1)
170 R6=1/(1260*R1*R1*R1*R1)+R6
180 R6=-1/(1680*R1*R1*R1*R1*R1)+R6
190 R5=EXP(R5+R6)
200 R2=R5/R1
220 R1=R1-1
230 IF(R4-R1-.001)>0 THEN PRINT:GOTO280
235 IF R1=0 THEN 120
240 R2=R2/R1
250 GOTO 220
```

```
280 PRINT"GAMMAFUNCTIE ";R7;" IS ";R2
285 PRINT"-----"
290 PRINT:GOTO 120
```

### Voorbeeld 8.1

```
*****FACULTEITEN*****
N KLEINER DAN 33 ? 12
FACULTEIT 12 IS 479001600

N KLEINER DAN 33 ? 8
FACULTEIT 8 IS 40320

N KLEINER DAN 33 ? 24
FACULTEIT 24 IS 6.20448402E+23

N KLEINER DAN 33 ?
```

### Voorbeeld 8.2

```
*****FACULTEITEN*****
N KLEINER DAN 33 ? .5

*****GAMMA-FUNCTIE*****
X IS GEEN GEHEEL GETAL.
X IS MIN. -10 EN MAX.+33

X IS ? 1.35
GAMMAFUNCTIE 1.35 IS .891151443
-----

X IS ? 1.75
GAMMAFUNCTIE 1.75 IS .919062527
-----

X IS ?
```



## 9. Permutaties en combinaties

De variabelen  $M$  en  $N$  moeten, van elkaar gescheiden door een komma, worden ingevoerd.  $M$  is het aantal verschillende objecten en  $N$  is de grootte van de rij. Zowel  $M$  als  $N$  moeten hele getallen zijn.

Het programma berekent achtereenvolgens het aantal permutaties:

$$P = \frac{M!}{(M-N)!} \qquad Q = \frac{M!}{(M-N)! N!}$$

### Programma 9

```
10 PRINT"**PERMUTATIES&COMBINATIES**"
12 PRINT"-----"
15 PRINT
20 PRINT" M VERSCHILLENDE OBJECTEN"
30 PRINT" N GROOTTE VAN RIJ":Q=1
32 PRINT" M EN N ZIJN GEHELE GETALLEN"
35 PRINT
40 PRINT" TYP M,N      ":INPUTM,N:P=M
50 FORI=1 TO (N-1)
60 P=P*(M-I):Q=Q*I
70 NEXT I:PRINT
80 PRINT"HET AANTAL PERMUTATIES IS "P
85 PRINT"-----"
90 Q=P/Q/N
95 PRINT
100 PRINT"HET AANTAL COMBINATIES IS ";Q
105 PRINT"-----"
110 PRINT:Q=1:GOTO 40
```

### Voorbeeld 9

```
**PERMUTATIES&COMBINATIES**
-----
M VERSCHILLENDE OBJECTEN
N GROOTTE VAN RIJ
M EN N ZIJN GEHELE GETALLEN
TYP M,N      ? 10,5

HET AANTAL PERMUTATIES IS  30240
-----

HET AANTAL COMBINATIES IS  252
-----
TYP M,N      ?
```

## 10. Binomiaalverdeling

De parameters die voor dit programma moeten worden ingegeven zijn:

$P$  voor de kans op succes van een proef,

$N$  voor het totaal aantal te nemen proeven en

$X$  voor het gewenste aantal succesvolle proeven.

Het programma berekent dan achtereenvolgens:

Het gemiddelde  $M = P \cdot N$ ;

De standaard afwijking  $S = \sqrt{M(1-P)}$ ;

De kans  $F$  dat er  $X$  succesvolle proeven optreden

$$F_{(X,N,P)} = \frac{N!}{X!(N-X)!} \cdot P^X (1-P)^{N-X};$$

De kans  $P$  dat er 0 tot en met  $X$  succesvolle proeven optreden

$$P_{(X,N,P)} = \sum_{x=1}^N \frac{X \cdot N! \cdot P}{X!(N-X)!} \cdot (1-P)^{N-X}$$

Hierna kan men door een nieuwe  $X$ -waarde in te geven  $F$  en  $P$  opnieuw laten berekenen.

### Programma 10

```
10 PRINT"****BINOMIAAL VERDELING****"
20 PRINT
30 PRINT"F(X,N,P) VOOR 0<=X<=N"
40 PRINT"P(X,N,P) VOOR 0<=X<=N"
50 PRINT"X EN N ZIJN GEHELE GETALLEN"
60 PRINT
70 INPUT"TYP P ";R1
80 INPUT"TYP N ";R2
90 PRINT:INPUT"TYP X ";R3
100 IF R3<0 THEN 250
110 R7=R3:R6=0:R4=-1
120 R5=(1-R1)*R2
130 R6=R6+R5
140 R4=R4+1
150 IF(R4-R7)>=0 THEN 180
160 R5=R5*R1*(R2-R4)/(R4+1)/(1-R1)
170 GOTO 130
180 PRINT"F("R3;" ";R2;" ";R1;")= ";R5
190 PRINT"P("R3;" ";R2;" ";R1;")= ";R6
195 IF K=1 THEN 240
200 M=R2*R1
210 PRINT"GEMIDDELDE IS ";M
220 S=SQR(M*(1-R1)):K=1
230 PRINT"STANDAARDAFWIJKING IS ";S
240 PRINT"-----"
245 GOTO 90
```

```
250 PRINT"F(X;N;P) IS 0"  
260 PRINT"P(X;N;P) IS 0"  
270 GOTO 90
```

## Voorbeeld 10

\*\*\*\*BINOMIAAL VERDELING\*\*\*\*

F(X,N,P) VOOR  $0 \leq X \leq N$   
P(X,N,P) VOOR  $0 \leq X \leq N$   
X EN N ZIJN GEHELE GETALLEN

TYP P ? .75  
TYP N ? 15

TYP X ? 2  
F( 2 ; 15 ; .75 )= 8.80099833E-07  
P( 2 ; 15 ; .75 )= 9.22940672E-07  
GEMIDDELDE IS 11.25  
STANDAARDAFWIJKING IS 1.67705098  
-----

TYP X ? 3  
F( 3 ; 15 ; .75 )= 1.14412978E-05  
P( 3 ; 15 ; .75 )= 1.23642385E-05  
-----

TYP X ? 4  
F( 4 ; 15 ; .75 )= 1.0297168E-04  
P( 4 ; 15 ; .75 )= 1.15335919E-04  
-----

TYP X ?



## 11. Poisson-verdeling

Het programma kent drie variabelen die u achtereenvolgens moet ingeven:

$P$  – de kans op een gebeurtenis (moet groter dan nul en kleiner dan één zijn).

$N$  – het aantal monsters.

$X$  – het aantal gebeurtenissen (indien per ongeluk nul werd opgegeven kan dit aantal opnieuw worden ingegeven).

Hierna berekent het programma eerst de kans dat bij  $N$  monsters  $X$  gebeurtenissen plaatsvinden, wanneer de kans op één gebeurtenis gelijks is aan  $P$ .

$$F_{(x)} = \frac{N \cdot P^x \cdot e^{-NP}}{X!}$$

Vervolgens wordt de totale kans berekend. De totale kans is de som van alle afzonderlijke kansen voor  $X=0$  tot en met  $X$ .

Nu volgt de berekening van de kans op  $X$  gebeurtenissen voor  $X=0$  tot en met  $X=X$  in procenten, waarna het programma om een nieuwe waarde voor  $X$  vraagt.

### Programma 11

```
10 PRINT"*****POISSON VERDELING*****"
15 PRINT"-----"
20 PRINT
30 PRINT" P IS KANS:  0<P<1  "
40 PRINT" N IS AANTAL MONSTERS"
50 PRINT" X IS AANTAL GEBEURTENISSEN"
60 PRINT
70 INPUT" P IS ";P
80 INPUT" N IS ";N:N=INT(N)
90 INPUT" X IS ";X:X=INT(X)
95 IF X<=0 THEN 90
100 L=N*P :F=1
110 FOR I= 1 TO X
120 F=F*I
130 FX=FX+EXP(-L)*(L↑I)/F
140 NEXT I
150 FS=EXP(-L)*(L↑(I-1))/F
160 PRINT
165 PRINT"-----"
170 PRINT"F(";X;") IS ";FS
180 PRINT:FX=FX+EXP(-L)
190 PRINT"SOM VAN F(X) VOOR X VAN 0 T/M ";X
200 PRINT"IS DE TOTALE KANS: ";FX
205 PRINT"-----"
210 PRINT:F=1
220 FOR I=0 TO X
230 IF I=0 THEN F=1:GOTO240
235 F=F*I
```

```

240 PX=N*EXP(-L)*(L↑I)/F:SU=SU+PX
245 PX=INT(PX*100+.5)/100
250 PRINT"P(";I;") IS ";PX
260 NEXT I
265 SU=INT(SU*100+.5)/100
270 PRINT"SOM VAN P(X) IS ";SU
280 PRINT"-----"
290 PRINT:GOTO 90

```

## Voorbeeld 11

\*\*\*\*\*POISSON VERDELING\*\*\*\*\*

```

P IS KANS: 0<P<1
N IS AANTAL MONSTERS
X IS AANTAL GEBEURTENISSEN
P IS ? .05
N IS ? 100
X IS ? 2

```

-----

F( 2 ) IS .0842243375

SOM VAN F(X) VOOR X VAN 0 T/M 2  
IS DE TOTALE KANS: .12465202

-----

P( 0 ) IS	.67	%
P( 1 ) IS	3.37	%
P( 2 ) IS	8.42	%
SOM VAN P(X) IS	12.47	%

-----

X IS ? 17

P( 0 ) IS	.67	%
P( 1 ) IS	3.37	%
P( 2 ) IS	8.42	%
P( 3 ) IS	14.04	%
P( 4 ) IS	17.55	%
P( 5 ) IS	17.55	%
P( 6 ) IS	14.62	%
P( 7 ) IS	10.44	%
P( 8 ) IS	6.53	%
P( 9 ) IS	3.63	%
P( 10 ) IS	1.81	%
P( 11 ) IS	.82	%
P( 12 ) IS	.34	%
P( 13 ) IS	.13	%
P( 14 ) IS	.05	%
P( 15 ) IS	.02	%
P( 16 ) IS	0	%
P( 17 ) IS	0	%
SOM VAN P(X) IS	100	%

-----

## 12. Enkelvoudige variantie-analyse

Het programma begint met het creëren van rijen. Op de vraag 'RIJNUMMER 1 T/M N' kunt u het rijnummer ingeven. Hierna komt het programma met de vraag 'TYP X (<rijnummer> <plaats-in-rij>)', waarop u de X-waarde kunt ingeven. Deze vraag wordt herhaald tot u voor X de waarde nul hebt ingegeven. U kunt nu een volgende rij maken, waarvoor u weer naar de X-waarden zult worden gevraagd. Indien u voor de X-waarde 999 ingeeft, is het creëren van rijen geëindigd. Het programma zal nu de volgende berekeningen gaan uitvoeren:

Het totaal van de groepen –  $TT = \sum_{n=1}^N X_n^2 - \frac{\sum_{n=1}^N X_n}{N}$

Het totaal tussen/groepen –  $TS = \frac{\sum_{n=1}^{N_1} X_n^2}{N_1} + \frac{\sum_{n=1}^{N_2} X_n^2}{N_2} - \frac{\sum_{n=1}^N X_n}{N}$   
( $N = N_1 + N_2$ )

Het totaal binnen/groepen –  $ES = TT - TS$

De vrijheidsgrootte tussen/groepen –  $TD = \text{aantal rijen} - 1$

De variantie tussen/groepen –  $MS = \frac{TS}{TD}$

De vrijheidsgrootte binnen/groepen –  $ED = N - TD$

De variantie binnen/groepen –  $E = \frac{ES}{ED}$

De F-waarde voor de H0-test –  $F = \frac{MS}{E}$

Hierna vraagt het programma u om het gewenste betrouwbaarheidspercentage, waarna de berekening van F-test H0 volgt. Ook geeft het programma nog aan of H0 is geaccepteerd of verworpen.

### Programma 12

```
10 PRINT "**VARIANTIEANALYSE, ENKELVOUDIG**"
20 PRINT "-----"
30 PRINT:R1=0:R2=0:R3=0:R4=0
40 INPUT"RIJNUMMER 1 T/M N ";Q
50 PRINT"VOOR EINDE RIJ TYP 0"
60 PRINT"VOOR BEREKENING TYP 999"
70 IF Q=999 THEN 180
80 R4=R4+1:R8=1
90 PRINT" TYP X(";Q;R8;") ";:INPUTX
100 IF X=0 THEN 160
```



```

110 R1=R1+X:R2=R2+1
120 R5=R5+1:R6=R6+X*X
130 R7=R7+X:R8=R8+1
140 GOTO 90
160 R8=0:R3=R3+R1*R1/R2
170 R1=0:R2=0:GOTO40
180 TT=R6-R7*R7/R5:PRINT
185 PRINT"-----"
190 PRINT"TOTAAL VAN GROEPEN ";TT
200 TS=R3-R7*R7/R5
210 PRINT"TOT.TUSSEN/GROEPEN ";TS
220 ES=TT-TS
230 PRINT"TOT.BINNEN/GROEPEN ";ES
240 TD=R4-1
250 PRINT"VRIJH.GR.T/GROEPEN ";TD
260 MS=TS/TD
270 PRINT"VARIANTIE T/GROEPEN";MS
280 ED=R5-R4
290 PRINT"VRIJH.GR. B/GROEPEN ";ED
300 E=ES/ED
310 PRINT"VARIANTIE B/GROEPEN ";E
320 F=MS/E
330 PRINT"FWAARDE VOOR H0-TEST";F
335 PRINT"-----"
340 PRINT"TYP GEWENST BETROUWBAARHEIDS"
345 INPUT"PERCENTAGE ";P:P=1-P/100:D=1
350 PRINT"BEZIG MET H0 TEST F-STATISTIEK"
430 S=S+D
440 RC=S:RE=S:RD=ED/2:R0=TD/2
450 IFR0=INT(R0)THEN500
460 U=RD:RD=R0:R0=U
470 RE=1/RC:K=1
500 RF=1/(RD/R0/RE+1)
520 T=(-RF+1)*RD:RG=1:RH=1:SU=1
530 R0=R0-1:IFR0<=0THEN 600
540 V=RD*RF/RG
550 RH=RH*V
560 RG=RG+1:RD=RD+1
570 SU=SU+RH:GOTO 530
600 W=T*SU
610 IF K=1 THEN W=1-W
620 IFW>PTHEN 660
630 IF W=P THEN 670
640 S=S-D:D=D/10:R=R+1
650 IF R=4 THEN 670
660 GOTO 430
670 PRINT"F-TEST H0 VOLGENS THEORIE";INT(S*100+.5)/100
675 PRINT"-----"
680 IF F<S THEN 700
690 PRINT" H0 VERWORPEN !!!":END
700 PRINT" H0 GEACCEPTEERD !!!":END

```

## Voorbeeld 12

\*\*VARIANTIEANALYSE, ENKELVOUDIG\*\*

---

RIJNUMMER 1 T/M N 1  
VOOR EINDE RIJ TYP 0  
VOOR BEREKENING TYP 999

TYP X( 1 1 )	1.11
TYP X( 1 2 )	1.15
TYP X( 1 3 )	1.14
TYP X( 1 4 )	1.1
TYP X( 1 5 )	1.09
TYP X( 1 6 )	1.11
TYP X( 1 7 )	1.12
TYP X( 1 8 )	1.15
TYP X( 1 9 )	1.13
TYP X( 1 10 )	1.14
TYP X( 1 11 )	0

RIJNUMMER 1 T/M N 2  
VOOR EINDE RIJ TYP 0  
VOOR BEREKENING TYP 999

TYP X( 2 1 )	1.12
TYP X( 2 2 )	1.06
TYP X( 2 3 )	1.02
TYP X( 2 4 )	1.08
TYP X( 2 5 )	1.11
TYP X( 2 6 )	1.05
TYP X( 2 7 )	1.06
TYP X( 2 8 )	1.03
TYP X( 2 9 )	1.05
TYP X( 2 10 )	1.08
TYP X( 2 11 )	0

RIJNUMMER 1 T/M N 999  
VOOR EINDE RIJ TYP 0  
VOOR BEREKENING TYP 999

---

TOTAAL VAN GROEPEN	.0301000066
TOT. TUSSEN/GROEPEN	.0168200247
TOT. BINNEN/GROEPEN	.0132799819
VRIJH. GR. T/GROEPEN	1
VARIANTIE T/GROEPEN	.0168200247
VRIJH. GR. B/GROEPEN	18
VARIANTIE B/GROEPEN	7.37776773E-04
FWAARDE VOOR H0-TEST	22.7982573

---

TYP GEWENST BETROUWBAARHEIDS  
PERCENTAGE 99  
BEZIG MET H0 TEST F-STATISTIEK  
F-TEST H0 VOLGENS THEORIE 8.29

---

H0 VERWORPEN !!!

## 13. Aitken-interpolatie

Het aantal dataparen en de waarden van die dataparen moet worden ingegeven. Bij het ingeven van de waarde van een datapaar moeten de waarden voor  $X$  en  $Y$  door middel van een komma van elkaar worden gescheiden. De  $X$ -waarden behoeven geen gelijke onderlinge afstand te hebben.

Na het ingeven van de dataparen kan men een bepaalde  $X$ -waarde opgeven, waarvoor het programma dan de bijbehorende  $Y$ -waarde berekent door middel van Aitken-interpolatie. Het aantal keren dat men een  $X$ -waarde kan ingeven is onbeperkt.

### *Opmerking:*

Alleen wanneer de  $X$ -waarde die u opgeeft binnen de  $X$ -waarden uit de dataparen valt, verkrijgt u een goed resultaat.

### Programma 13

```
10 PRINT"****AITKEN INTERPOLATIE****"
20 PRINT"-----"
25 PRINT"X OOK VOOR NIET GELIJKE AFSTANDEN"
30 PRINT:DIMD(200):B=1:E=1
40 INPUT"HET AANTAL DATAPAREN IS ";C
50 F=C+B
60 FORI=1TOC
70 PRINT"TYPE X,Y #";I;:INPUTX,Y
80 D(E)=X:D(F)=Y:E=E+1:F=F+1
90 NEXT
100 PRINT:PRINT"INTERPOLATIE: X= ";:INPUTD
110 E=B+C:A=C:F=E+C
120 FORI=1TOA
130 D(F)=D(E):F=F+1:E=E+1
140 NEXT
150 A=C-1:E=B+2*C:G=B
160 F=E+1:H=G+1
170 K=D(G)-D:L=A
180 D(F)=(D(E)*(D(H)-D)-D(F)*K)/(D(H)-D(G))
190 F=F+1:H=H+1
200 L=L-1:IFL=0THEN230
210 GOTO 180
230 E=E+1:G=G+1
240 A=A-1:IFA=0THEN260
250 GOTO160
260 F=F-1
270 PRINT"                Y-WAARDE= ";D(F)
280 PRINT:GOTO100
```



## Voorbeeld 13

\*\*\*\*AITKEN INTERPOLATIE\*\*\*\*

X OOK VOOR NIET GELIJKE AFSTANDEN

HET AANTAL DATAPAREN IS 5  
TYP X,Y # 1 1 2  
TYP X,Y # 2 3 5  
TYP X,Y # 3 4 6  
TYP X,Y # 4 7 10  
TYP X,Y # 5 10 3

INTERPOLATIE: X= 4  
Y-WAARDE= 6

INTERPOLATIE: X= 7  
Y-WAARDE= 10

INTERPOLATIE: X= 2.5  
Y-WAARDE= 4.50111607

INTERPOLATIE: X= 7.5  
Y-WAARDE= 10.3394097

INTERPOLATIE: X= 3.15  
Y-WAARDE= 5.14330872

INTERPOLATIE: X= 0  
Y-WAARDE= -1.81481481

## 14. Lagrange-interpolatie

De dataparen  $X, Y$  kunnen worden ingevoerd. Bij het invoeren moeten  $X$  en  $Y$  van elkaar gescheiden door een komma ingetypt worden. De  $X$ -waarden behoeven niet op onderling gelijke afstanden van elkaar te liggen. De waarde 99,99 is voor het programma een teken dat er niet meer dataparen zullen worden ingevoerd. Het maximum aantal in te voeren dataparen is 75.

Na het invoeren van de dataparen kan men een bepaalde  $X$ -waarde opgeven, waarvoor het programma dan de bijbehorende  $Y$ -waarde berekent met behulp van de Lagrange-interpolatie.

Voor het verkrijgen van een juiste  $Y$ -waarde moet de op te geven  $X$ -waarde tussen de  $X$ -waarden van de dataparen liggen.

### Programma 14

```
10 PRINT "***LAGRANGE INTERPOLATIE***"
15 PRINT"-----"
20 PRINT"X WEL OF NIET GELIJKE AFSTANDEN"
30 PRINT:DIM X(75),Y(75):N=1
40 PRINT"VOOR START BEREKENING TYP 99,99"
50 PRINT:PRINT"X,Y #";N;"IS ";;INPUTX(N),Y(N)
60 IF X(N)=99 THEN 80
70 N=N+1:GOTO 50
80 IF Y(N)=99 THEN PRINT:GOTO100
90 N=N+1:GOTO 50
100 PRINT"INTERPOLATIE X-WAARDE ";
110 INPUTX:S=0
120 FOR I=1 TO N-1
130 P=1
140 FOR J=1 TO N-1
150 IF I=J THEN 170
160 P=P*(X-X(J))/(X(I)-X(J))
170 NEXT J
180 S=S+P*Y(I)
190 NEXT I
200 PRINT "DE Y-WAARDE IS:          ";S
210 PRINT"-----"
220 PRINT
230 GOTO 100
240 END
```

## Voorbeeld 14

\*\*\*LAGRANGE INTERPOLATIE\*\*\*

X WEL OF NIET GELIJKE AFSTANDEN

VOOR START BEREKENING TYP 99,99

X,Y # 1 IS 5.6 2.366

X,Y # 2 IS 5.85 2.419

X,Y # 3 IS 6 2.449

X,Y # 4 IS 6.13 2.476

X,Y # 5 IS 99 99

INTERPOLATIE X-WAARDE 5.6  
DE Y-WAARDE IS: 2.366

---

INTERPOLATIE X-WAARDE 6.13  
DE Y-WAARDE IS: 2.476

---

INTERPOLATIE X-WAARDE 5.75  
DE Y-WAARDE IS: 2.39865665

---

INTERPOLATIE X-WAARDE 5.93  
DE Y-WAARDE IS: 2.4349676

---

INTERPOLATIE X-WAARDE 6.1  
DE Y-WAARDE IS: 2.46960548

---

INTERPOLATIE X-WAARDE 5.68  
DE Y-WAARDE IS: 2.38383993

---



## 15. Voortschrijdend gemiddelde

De data die dient te worden gemiddeld, moet vòòr het starten van het programma op de datalijnen 300 en hoger worden ingevoerd. Nadat het programma is gestart dienen het aantal monsters en het maximum aantal te middelen waarden te worden opgegeven. In het voorbeeld wordt uitgegaan van 17 monsters en maximaal drie te middelen waarden. Voor dit voorbeeld zal het programma de volgende berekeningen uitvoeren:

Het voorschrijdend gemiddelde van  $X_1$

Het voortschrijdend gemiddelde van  $X_1$  en  $X_2$

Het voortschrijdend gemiddelde van  $X_1$ ,  $X_2$  en  $X_3$

Het voortschrijdend gemiddelde van  $X_2$ ,  $X_3$  en  $X_4$

Het voortschrijdend gemiddelde van  $X_3$ ,  $X_4$  en  $X_5$  enz.

Na zes berekeningen wordt u gevraagd een letter in te tikken, waarna weer zes berekeningen volgen.

Wanneer alle gemiddelden zijn berekend, kan men een nieuw maximum aantal te middelen waarden opgeven, waarna de berekeningen opnieuw worden uitgevoerd, nu met deze nieuwe waarde.

### Programma 15

```
10 PRINT"***VOORTSCHRIJDEND GEMIDDELDE***"
20 PRINT"-----"
30 PRINT:PRINT"DATARIJ IN LIJN 300 EN OP"
40 INPUT"AANTAL MONSTERS=";A
50 DIMX(A+1):S=0
60 INPUT"MAX.AANTAL TE MIDDELEN WAARDEN";N
70 FORI=1TON
80 READX(I)
90 PRINT"X(";I;")=";X(I)
100 S=S+X(I):VG=S/I
110 PRINT"V.G.";N;" IS ";VG
115 PRINT"-----"
120 NEXTI:K=I
130 FORI=ITOA
140 READ X(I)
150 PRINT"X(";I;")=";X(I)
160 S=S+X(I)-X(I-N):VG=S/N
170 PRINT"V.G.";N;" IS ";VG
175 PRINT"-----"
177 K=K+1
178 IF K=7 THEN 183
180 NEXT I:GOTO 190
183 PRINT"TYP EEN LETTER"
184 GETA$:IFA$=""THEN184
185 K=0:I=I+1:GOTO 140
190 RESTORE:S=0:GOTO60
300 DATA75,66,87,82,66,85,99,71,88,73,88,75,91,86,83,95,82,99
```

## Voorbeeld 15

\*\*\*VOORTSCHRIJDEND GEMIDDELDEN\*\*\*

DATARIJ IN LIJN 300 EN OP  
AANTAL MONSTERS= 17  
MAX.AANTAL TE MIDDELEN WAARDEN 3

X( 1 )= 75

V.G. 3 IS 75

X( 2 )= 66

V.G. 3 IS 70.5

X( 3 )= 87

V.G. 3 IS 76

X( 4 )= 82

V.G. 3 IS 78.3333334

X( 5 )= 66

V.G. 3 IS 78.3333334

X( 6 )= 85

V.G. 3 IS 77.6666667

TYP EEN LETTER

K

X( 7 )= 99

V.G. 3 IS 83.3333334

X( 8 )= 71

V.G. 3 IS 85

X( 9 )= 88

V.G. 3 IS 86

X( 10 )= 73

V.G. 3 IS 77.3333334

X( 11 )= 88

V.G. 3 IS 83

X( 12 )= 75

V.G. 3 IS 78.6666667

X( 13 )= 91

V.G. 3 IS 84.6666667

TYP EEN LETTER

K

X( 14 )= 86

V.G. 3 IS 84

X( 15 )= 83

V.G. 3 IS 86.6666667

X( 16 )= 95

V.G. 3 IS 88

X( 17 )= 82

V.G. 3 IS 86.6666667

## 16. Polynoom-interpolatie

De dataparen moeten voor het starten van het programma worden ingevoerd in datastrings. Alle data moeten van elkaar worden gescheiden door een komma. De datastrings kunnen worden ingevoerd in datalijnen 60 tot en met 90. Om aan te geven dat er niet meer dataparen volgen moet het laatste datapaar 99,99 zijn.

Na het starten van het programma kunt u de orde van de polynoom opgeven. De orde der polynoom is gelijk aan het aantal dataparen plus 1.

Het programma berekent vervolgens met behulp van de kleinste kwadraten methode de coëfficiënten voor de machten van  $X$ :

$$F_{(x)} = A \cdot X^0 + B \cdot X^1 + C \cdot X^2 + D \cdot X^3 \dots\dots$$

De macht van  $X$  is afhankelijk van de orde der polynoom.

Vervolgens berekent het programma de kwadratische- en relatieve fout. Na deze berekening krijgt u de keuze uit:

1: Dezelfde berekening nog eens, echter nu voor een andere orde.

2: Berekening van  $F_{(x)} = Y$ , met gebruikmaking van de in de vorige berekening gevonden coëfficiënten en de nog in te voeren  $X$ -waarde.

Om terug te kunnen gaan naar de keuzemogelijkheid dient u 111 in te tikken.

### Programma 16

```
10 PRINT"***POLYNOOM INTERPOLATIE VOLGENS"
15 PRINT"-----":PRINT
20 PRINT"DE METHODE V.D. KLEINSTE KWADRATEN."
30 PRINT"DATALIJNEN BEVINDEN ZICH IN LIJN 60-90."
40 PRINT"X,Y PAREN, AF TE SLUITEN MET 99,99"
45 PRINT"ORDE MAX.9; PRAKTISCH 7 EN GROTER"
48 PRINT"OF GELIJK AAN AANTAL PAREN PLUS EEN."
50 DIM X(100),Y(100),A(10,10),C(20)
55 DIM B(20),D(20)
60 DATA 0,576,1,326,2,92,3,-162,4,-400,5,-514,6,-324,7,422,99,99
100 FOR I=1 TO 100
110 READ X(I),Y(I)
120 IF X(I)=99 AND Y(I)=99 THEN S=I-1:I=100
130 NEXT
140 INPUT"ORDE VAN POLYNOOM ";N:N=INT(N):F=N+N
145 IF N>9 THEN 140
148 IF N=S THEN 140
149 REM VUL MATRIX VOOR GAUSS-ELIMINATIE
150 N=N+1
160 FOR J=1 TO F
170 D(J)=0
180 FOR I=1 TO S
190 D(J)=D(J)+X(I)^J
200 NEXT I
205 NEXT J
210 D(0)=S:B(1)=0
```



```

220 FORI=1TOS
230 B(1)=B(1)+Y(I)
240 NEXT
250 FORJ=2TON
260 B(J)=0
270 FORI=1TOS
280 B(J)=B(J)+Y(I)*X(I)↑(J-1)
290 NEXT:NEXT
300 FORJ=1TON
310 FORI=1TON
320 A(J,I)=D(J+I-2)
330 NEXT:NEXT
340 FORH=1TON-1:I=H+1:K=H
350 REM ZOEK PIVOT
360 IFABS(A(I,H))>ABS(A(K,H))THENK=I
370 IFI<NTHENI=I+1:GOTO360
380 IFK=HTHEN420
390 FORJ=HTON:U=A(H,J):A(H,J)=A(K,J)
400 A(K,J)=U:NEXT
410 U=B(H):B(H)=B(K):B(K)=U
415 REM ELIMINEER MATRIXRIJ
420 I=H+1
430 U=A(I,H)/A(H,H):A(I,H)=0
440 FORJ=H+1TON
445 A(I,J)=A(I,J)-U*A(H,J):NEXT
450 B(I)=B(I)-U*B(H)
455 IFI<NTHENI=I+1:GOTO430
460 NEXT
470 C(N)=B(N)/A(N,N)
475 FORI=N-1TO1 STEP -1
480 U=0
485 FORJ=I+1TON:U=U+A(I,J)*C(J)
490 C(I)=(B(I)-U)/A(I,I)
495 NEXT:NEXT
500 PRINT:PRINT"MACHT VAN X      COEFFICIENT"
510 FORI=1TON
515 PRINT"      ";I-1,"      ";C(I):NEXT
520 PRINT:PRINT
525 W=0
530 FORJ=1TOS
535 U=0
540 FORK=1TON:U=U+C(K)*X(J)↑(K-1):NEXT
550 W=W+((Y(J)-U)*(Y(J)-U)):NEXT
560 PRINT"SOM V.D. KWADRATISCHE FOUT IS: ";INT(W*10000+.5)/10000
570 PRINT" 'CHI' IS: ";INT(SQR(W/S)*10000+.5)/10000
585 PRINT:PRINT"VOOR HERHALING TYP 111 'INTERPOLATIE'"
590 PRINT"TYP 1 VOOR ANDERE ORDE "
595 INPUT"TYP 2 VOOR INTERPOLATIE";Z:Z=INT(Z)
600 IFZ=1 THEN 140
605 IFZ<>2 THEN 585
610 PRINT:INPUT"X IS ";H:Y=0
615 IFH=111 THEN 585
620 FORI=1TON
630 Y=Y+C(I)*H↑(I-1):NEXT
640 PRINT"Y IS ";Y
650 GOTO 610

```

## Voorbeeld 16

### \*\*POLYNOOM INTERPOLATIE VOLGENS

DE METHODE V.D. KLEINSTE KWADRATEN.  
DATALIJNEN BEVINDEN ZICH IN LIJN 60-90.  
X,Y PAREN, AF TE SLUITEN MET 99,99  
ORDE MAX.9; PRAKTISCH 7 EN GROTER  
OF GELIJK AAN AANTAL PAREN PLUS EEN.  
ORDE VAN POLYNOOM 2

MACHT VAN X	COEFFICIENT
0	728.99999
1	-536.999991
2	65.8571416

SOM V.D. KWADRATISCHE FOUT IS: 208748.571  
'CHI' IS: 161.535

VOOR HERHALING TYP 111 'INTERPOLATIE'  
TYP 1 VOOR ANDERE ORDE  
TYP 2 VOOR INTERPOLATIE 1  
ORDE VAN POLYNOOM 3

MACHT VAN X	COEFFICIENT
0	540.000014
1	-42.0000509
2	-123.142837
3	17.999998

SOM V.D. KWADRATISCHE FOUT IS: 16292.571  
'CHI' IS: 45.1284

VOOR HERHALING TYP 111 'INTERPOLATIE'  
TYP 1 VOOR ANDERE ORDE  
TYP 2 VOOR INTERPOLATIE 1  
ORDE VAN POLYNOOM 4

MACHT VAN X	COEFFICIENT
0	575.999875
1	-287.999365
2	58.9995744
3	-23.9999077
4	2.99999369

SOM V.D. KWADRATISCHE FOUT IS: 0  
'CHI' IS: 1E-04

VOOR HERHALING TYP 111 'INTERPOLATIE'  
TYP 1 VOOR ANDERE ORDE  
TYP 2 VOOR INTERPOLATIE 2

X IS	1
Y IS	326.000171
X IS	2.5
Y IS	-33.0625021
X IS	4
Y IS	-400.000106
X IS	3.5
Y IS	-288.062608
X IS	111

## 17. Exponentiële interpolatie

Dit algoritme is alleen relevant voor fysische verschijnselen die goed kunnen worden beschreven door een stelsel van maximaal twee eerste orde differentiaal vergelijkingen.

Er wordt tevens gebruik gemaakt van een kleinste kwadraten methode om de eventuele meetfouten te minimaliseren. De output 'gemiddelde fout' is hiervoor een indicatie.

Het programma vraagt u eerst het aantal ( $N$ ) in te voeren  $Y$ -waarden. Hierna kunt u het opgegeven aantal  $Y$ -waarden invoeren. Het maximum aantal  $Y$ -waarden is 100. Vervolgens vraagt het programma u om de startwaarde  $X_0$  en de stapgrootte  $DX$ .

Na opgave van bovenstaande gegevens berekent het programma afhankelijk van de discriminant de waarden  $C_1$ ,  $R_1$ ,  $C_2$  en  $R_2$ . De berekening geschiedt volgens één van de volgende formules:

$$D > 0 \rightarrow Y = C_1 R_1^x + C_2 R_2^x$$

$$D < 0 \rightarrow Y = R^x (C_1 \cdot \cos(Q \cdot X) + C_2 \cdot \sin(Q \cdot X))$$

$$D = 0 \rightarrow Y = R^x (C_1 + C_2 \cdot X)$$

Hierna kunt u een  $X$ -waarde opgeven waarvoor dan de bijbehorende  $Y$ -waarde wordt berekend. Indien u voor  $X$  de waarde 999 geeft, dan start de berekening voor een homogene lineaire differentiaal vergelijking:

$$D_2 \cdot Y + P \cdot D_1 \cdot Y + Q \cdot Y = 0$$

gevolgd door de berekening van  $P$ ,  $Q$ , de dempingscoëfficiënt, de tijdconstante en de natural period.

### Programma 17

```
10 PRINT"EXPONENTIELE EN GONIOMETRISCHE I N T E R P O L A T I E"
12 PRINT"MAX.AANTAL N IS 100"
15 PRINT"PUNTEN EQ.DST.:Y0, Y1, Y2 ENZ."
20 DIM Y(100)
21 INPUT"N";N
22 FORI=0TON-1
23 PRINT"Y";I;:INPUTY(I)
24 NEXTI
25 FORI=2TON-1
30 R1=R1+Y(I-1)*Y(I-1)
31 R2=R2+Y(I-1)*Y(I-2)
32 R3=R3+Y(I-2)*Y(I-2)
35 R4=R4+Y(I-1)*Y(I)
36 R5=R5+Y(I-2)*Y(I)
37 NEXTI
45 R6=R1*R3-R2*R2
```



```

46 B=(R3*R4-R2*R5)/R6
50 A=(R1*R5-R2*R4)/R6
55 A$="## Y=C1*R1↑X+C2*R2↑X ##"
56 B$="## Y=R↑X*(C1*COS(Q*X)+C2*SIN(Q*X)) ##"
57 C$="## Y=R↑X*(C1+C2*X) ##"
58 D$="X=(X-X0)/DX FORMATTING"
59 E$="VDOOR HOM.LIN.DV(L5000) TYP 999"
60 D=B*B+4*A: P=π
61 PRINT" TYP X0": INPUT X0
62 PRINT" TYP DX": INPUT DX
70 IF D>0 THEN E=1
80 IF D<0 THEN E=2
90 IF D=0 THEN E=3
100 ON E GOTO 250,500,750
250 R1=(B+SQR(D))/2
260 R2=(B-SQR(D))/2
270 C2=(Y(1)-Y(0)*R1)/(R2-R1)
280 C1=Y(0)-C2
285 GOSUB 1000:PRINTD$:PRINTA$:PRINTE$
286 PRINT"C1=";C1:PRINT"C2=";C2:PRINT"R1=";R1:PRINT"R2=";R2
290 PRINT"INTERPOLATIE, TYP X": INPUT X
295 IFX=999 THEN5000
296 X=(X-X0)/DX
300 IF R1<0 AND R2<0 THEN 350
310 IF R1<0 THEN 400
320 IF R2<0 THEN 425
330 PRINT C1*R1↑X+C2*R2↑X:GOTO 290
350 PRINT C1*ABS(R1)↑X*COS(P*X)+C2*ABS(R2)↑X*COS(P*X):GOTO290
400 PRINT C1*ABS(R1)↑X*COS(P*X)+C2*R2↑X:GOTO290
425 PRINTC1*R1↑X+C2*ABS(R2)↑X*COS(P*X):GOTO290
500 IM=SQR(ABS(D/4)):RE=B/2
510 R=SQR(RE*RE+IM*IM):Q=ATN(IM/RE)
515 Q=ABS(Q)
516 IFRE<0ANDIM>0THENQ=3*π-Q
520 C1=Y(0):C2=(Y(1)-R*Y(0)*COS(Q))/R/SIN(Q)
525 GOSUB 2000:PRINTD$:PRINTB$:PRINTE$
526 PRINT"R=";R:PRINT"C1=";C1:PRINT"C2=";C2:PRINT"Q=";Q
530 PRINT"INTERPOLATIE, TYP X": INPUT X
531 IFX=999 THEN5000
535 X=(X-X0)/DX
540 PRINT R↑X*(C1*COS(Q*X)+C2*SIN(Q*X)):GOTO530
750 R=B/2:C1=Y(0):C2=Y(1)/R-Y(0)
755 GOSUB 3000:PRINTD$:PRINTC$:PRINTE$
756 PRINT"R=";R:PRINT"C1=";C1:PRINT"C2=";C2
760 PRINT"INTERPOLATIE": INPUT X
761 X=(X-X0)/DX
762 IFX=999 THEN5000
765 IF R<0 THEN 780
770 PRINT R↑X*(C1+C2*X):GOTO 760
780 PRINT ABS(R)↑X*(C1+C2*X):GOTO 760
1000 IFR1<0 AND R2<0 THEN 1050
1001 IFR1<0 THEN 1100
1002 IFR2<0 THEN 1200
1003 GOTO 1300
1050 FORI=0TON-1

```

```

1055 ER=ER+SQR(ABS((Y(I)-C1*ABS(R1)↑I*COS(P*I)-C2*ABS(R2)↑I
      *COS(P*I))↑2))
1060 NEXTI
1065 PRINT"GEM.FOUT ABS. IS **";ER/N
1070 RETURN
1100 FORI=0TON-1
1110 ER=ER+SQR(ABS((Y(I)-C1*ABS(R1)↑I*COS(P*I)-C2*ABS(R2)↑I)↑2))
1112 NEXTI
1115 PRINT"GEM.FOUT ABS. IS **";ER/N
1120 RETURN
1200 FORI=0TON-1
1205 ER=ER+SQR(ABS((Y(I)-C1*R1↑I-C2*ABS(R2)↑I*COS(P*I))↑2))
1210 NEXTI
1215 PRINT"GEM.FOUT ABS. IS **";ER/N
1220 RETURN
1300 FORI=0TON-1
1310 ER=ER+SQR((Y(I)-C1*(R1↑I)-C2*(R2↑I))↑2)
1320 NEXTI
1330 PRINT"GEM.FOUT IS ABS. ";ER/N
1340 RETURN
2000 FORI=0TON-1
2010 ER=ER+ABS(Y(I)-(R↑I)*(C1*COS(Q*I)+C2*SIN(Q*I)))
2020 NEXTI
2030 PRINT"GEM.FOUT IS ABS. ";ER/N
2040 RETURN
3000 IF R<0 THEN 3200
3005 FORI=0TON-1
3010 ER=ER+SQR((Y(I)-R↑I*(C1+C2*I))↑2)
3020 NEXTI
3030 PRINT"GEM. FOUT IS ABS. ";ER/N
3040 RETURN
3200 FORI=0TON-1
3210 ER=ER+SQR((Y(I)-ABS(R)↑I*(C1+C2*I))↑2)
3220 PRINT"GEM. FOUT IS ABS.** ";ER/N
3230 RETURN
5000 IFE=2 THEN 6000
5010 IFE=3 THEN 7000
5020 P=-1*(LOG(R1)+LOG(R2))
5030 Q=LOG(R1)*LOG(R2)
5040 PRINT"D2Y+P*D1Y+QY=0"
5050 PRINT"P= ";P
5060 PRINT"Q= ";Q
5065 PRINT"DAMPINGSCOEFF.= ";P/2
5066 PRINT"TYDCONSTANTE = ";P/Q
5070 END
6000 P=(LOG(R)*-2)
6005 IF ABS(Q-2*π)<.005 THENQ=0
6006 IFQ<.005 THENQ=0
6010 Q1=LOG(R)*LOG(R)+Q*Q
6020 PRINT"D2Y+PD1Y+QY=0"
6030 PRINT"P= ";P
6040 PRINT"Q= ";Q1
6045 PRINT"DAMPINGSCOEFF.= ";P/2
6046 FF=2*π/SQR(Q1-P*P/4+.000001)
6048 PRINT"NATURAL PERIOD= ";FF;" SEC."

```

```

6049 PRINT"AMPLITUDE AT ZERO ";C1
6050 PRINT"ALS P=Q CRITICAL DAMPING"
6051 PRINT"OMEGA+2*π:TYP P";:INPUTO$
6052 IF O$="P"THENQ=Q+2*π:GOTO6000
6053 END
7000 P=LOG(R)*-2
7010 Q=P*P/4
7020 PRINT"D2Y+PD1Y+QY=0
7030 PRINT"P= ";P
7040 PRINT"Q= ";Q
7045 PRINT"LINIAIRE FUNCTIE"
7050 END

```

## Voorbeeld 17.1

```

EXPONENTIELE EN GONIOMETRISCHE
MAX.AANTAL N IS 100
PUNTEN EQ.DST.:Y0, Y1, Y2 ENZ.
N 4
Y 0 0
Y 1 1.8
Y 2 2.16
Y 3 2.31
TYP X0 0
TYP DX 3
GEM.FOUT IS ABS. 3.48823192E-10
X=(X-X0)/DX FORMATTING
## Y=C1*R1↑X+C2*R2↑X ##
VOOR HOM.LIN.DV(L5000) TYP 999
C1= 1.99589744
C2=-1.99589744
R1= 1.05092498
R2= .149075026
INTERPOLATIE, TYP X 1
.970923949
INTERPOLATIE, TYP X 2
1.50195101
INTERPOLATIE, TYP X 4
1.97479102
INTERPOLATIE, TYP X 5
2.08450609
INTERPOLATIE, TYP X 7
2.21763781
INTERPOLATIE, TYP X 8
2.26610164
INTERPOLATIE, TYP X 999
D2Y+P*D1Y+QY=0
P= 1.85363487
Q= -.0945385285
DEMPINGSCOEFF.= .926817433
TYDCONSTANTE = -19.6071897

```

## I N T E R P O L A T I E



## Voorbeeld 17.2

```
EXPONENTIELE EN GONIOMETRISCHE  
MAX.AANTAL N IS 100  
PUNTEN EQ.DST.:Y0, Y1, Y2 ENZ.  
N 6  
Y 0 1  
Y 1 .14723  
Y 2 -.96976  
Y 3 -.42741  
Y 4 .84844  
Y 5 .67315  
TYP X0 0  
TYP DX 1  
GEM.FOUT IS ABS. 1.16497299E-03  
X=(X-X0)/DX FORMATTING  
## Y=R1%*(C1*COS(Q*X)+C2*SIN(Q*X)) ##  
VOOR HOM.LIN.DV(L5000) TYP 999  
R= 1.00339866  
C1= 1  
C2= 3.28635316E-03  
Q= 1.42682047  
INTERPOLATIE, TYP X 2  
-.964416471  
INTERPOLATIE, TYP X 2.5  
-.919976451  
INTERPOLATIE, TYP X 4.5  
1.00625821  
INTERPOLATIE, TYP X 5  
.673087148  
INTERPOLATIE, TYP X 999  
D2Y+PD1Y+QY=0  
P= -6.78579828E-03  
Q= 2.03582816  
DAMPINGSCOEFF.= -3.39289914E-03  
NATURAL PERIOD= 4.40362603 SEC.  
AMPLITUDE AT ZERO 1  
ALS P=Q CRITICAL DAMPING
```

I N T E R P O L A T I E

## Voorbeeld 17.3

```
EXPONENTIELE EN GONIOMETRISCHE  
MAX.AANTAL N IS 100  
PUNTEN EQ.DST.:Y0, Y1, Y2 ENZ.  
N 5  
Y 0 2  
Y 1 4  
Y 2 6  
Y 3 8  
Y 4 10  
TYP X0 0  
TYP DX 1  
GEM. FOUT IS ABS. 0
```

I N T E R P O L A T I E

```

X=(X-X0)/DX  FORMATTING
## Y=R1X*(C1+C2*X) ##
VOOR HOM.LIN.DV(L5000) TYP 999
R= 1
C1= 2
C2= 2
INTERPOLATIE 1
 4
INTERPOLATIE 3
 8
INTERPOLATIE 4
10
INTERPOLATIE 4.5
11
INTERPOLATIE 1.5
 5
INTERPOLATIE 999
D2Y+PD1Y+QY=0
P=  0
Q=  0
LINIAIRE FUNCTIE

```

## 18. Spline-interpolatie

De dataregels 50 tot en met 90 moeten worden ingevuld voordat het programma wordt gestart. De indeling van de data in de dataregels is als volgt:

$N$ ,  $FOUT$ ,  $X(I)$ ,  $Y(I)$ ,  $X(I)$ ,  $Y(I)$ , ....

Hierin is  $N$  het aantal datapunten ( $X(I)$ ,  $Y(I)$ ).

$FOUT$  staat voor de maximaal toelaatbare fout die tijdens de berekening mag worden gemaakt.

$X(I)$ ,  $Y(I)$  zijn de gegeven datapunten.

Na het starten van het programma kunt u kiezen uit drie modes.

Kiest u mode 1, dan zal het programma voor ieder datapunt een uitgebreide interpolatie  $Y$ ,  $Y'$ ,  $Y''$  uitvoeren.

Kiest u mode 2, dan zal het programma, nadat u de waarden voor  $A$  en  $B$  hebt ingetypt, de integraal berekenen van  $X(A)$  tot  $X(B)$ .

Kiest u mode 3, dan zal het programma op alleen  $Y$  interpoleren, voor de door u op te geven begin- en eindwaarde van  $X$  en de stapgrootte.

De gevolgde methode heeft de eigenschap de meest 'gladde' curve door de opgegeven punten te berekenen (collocatie), waarbij de eerste en tweede afgeleiden continue zijn.

Het door de curve omsloten oppervlak is minimaal en dit geeft een vorm van stroomlijn, waarbij de eerste en tweede afgeleiden aan het begin- en eindpunt nul zijn.

### Programma 18

```
10 PRINT"****SPLINE FUNCTIE****"
15 PRINT"-----"
20 PRINT"DATA PUNTEN LINES 50-90"
30 PRINT"FORMAAT:N,FOUT,X(I),Y(I):I=1TON"
40 PRINT" @ BETEKENT CARIAGE RETURN"
50 DATA 5,.000001,0,576,1,326,2,92,3,-162,4,-400
100 READ N
110 READ EPSLN
115 DIM H(N),X(N),Y(N),H2(N),B(N),DY(N),DS(N),S2(N),C(N)
116 DIM S3(N)
120 FOR I=1 TO N
130 READ X(I):READ Y(I)
140 NEXT I
150 N1=N-1
160 FOR I=1TON1
170 H(I)=X(I+1)-X(I)
180 DY(I)=(Y(I+1)-Y(I))/H(I)
190 NEXTI
200 FOR I=2 TO N1
210 H2(I)=H(I-1)+H(I)
220 B(I)=.5*H(I-1)/H2(I)
230 DS(I)=(DY(I)-DY(I-1))/H2(I)
```



```

240 S2(I)=2*DS(I)
250 C(I)=3*DS(I)
260 S2(1)=0:S2(N)=0
270 OM=4*(2-SQR(3))
275 NEXT I
280 FOR I=2 TO N1
290 W=(C(I)-B(I)*S2(I-1)-(.5-B(I))*S2(I+1)-S2(I))*OM
310 S2(I)=S2(I)+W: NEXT I
315 IF ABS(W)<EPSLN THEN I=N1:GOTO330
320 GOTO 280
330 FOR I=1 TO N1
340 S3(I)=(S2(I+1)-S2(I))/H(I)
350 NEXT I
360 RESTORE:PRINT
361 PRINT" MODE 1: INTERP.UITGEBREID Y,Y',Y""
362 PRINT" MODE 2: INTEGRAAL X(A)-X(B) 'INTERVAL'"
363 PRINT" MODE 3: INTERPOLATIE OP Y ALLEEN"
364 PRINT"          MET FOR NEXT LOOP & TAFEL"
365 PRINT:PRINT" TYP 1, 2 OR 3 AND @":INPUT M
366 IF M=1 THEN 370
367 IF M=2 THEN 640
368 IF M=3 THEN 700
369 GOTO 361
370 PRINT:INPUT"INTERPOLATIE X=";T
390 IF T<X(1)THEN 490
400 IF T=X(N)THEN I=N:GOTO540
401 IF T>X(N) THEN 490
405 FOR J=1 TO N
410 IF T<X(J) THEN I=J:J=N:GOTO 540
420 NEXT J
490 PRINT"NIET IN INTERVAL":GOTO 370
540 I=I-1:HT=T-X(I)
550 HU=T-X(I+1)
560 PR=HT*HU
570 SS=S2(I)+HT*S3(I)
580 DQ=(S2(I)+S2(I+1)+SS)/6
590 SJ=Y(I)+HT*DY(I)+PR*DQ
600 TT=DY(I)+(HT+HU)*DQ+PR*S3(I)/6
610 PRINT:PRINT "Y=: ";SJ
620 PRINT "Y'=: ";TT
630 PRINT "Y''=: ";SS:PRINT:GOTO 370
640 PRINT:PRINT"INTEGRAAL VAN X(A) TOT X(B)"
643 INPUT" TYP A,B: ";AA,B
645 B=B-1:A=0
650 FOR I=AA TO B
660 A=A+.5*H(I)*(Y(I)+Y(I+1))-H(I)*13*(S2(I)+S2(I+1))/24
665 NEXT I
670 PRINT
680 PRINT"          INTEGRAAL=: ";A
690 GOTO 640
700 PRINT:INPUT"START INTERP.LOOP MET";K
710 INPUT"EIND OF INTERPOLATIE LOOP";L
720 PRINT" TYP DELTA VOOR LOOP"
722 INPUT"20 PUNTEN IS PRIMA!";M
725 PRINT " X",:PRINT" Y"

```

```

730 GOSUB 790
740 K=INT(K*1000+.5)/1000:PRINTK,:PRINT SJ
750 K=K+M:IF K>L THEN 770
760 GOTO 730
770 INPUT"MEER? TYP J & @":A$
780 IF A$="J" THEN 700
785 END
790 IF K<X(1) THEN 830
800 IF K=X(N) THEN I=N:GOTO 840
801 IF K>X(N) THEN 830
805 FOR J=1 TO N
810 IF K<X(J) THEN I=J:J=N:GOTO 840
820 NEXT J
830 PRINT"NIET IN INTERVAL":PRINT:GOTO 700
840 I=I-1:HT=K-X(I)
850 HU=K-X(I+1)
860 PR=HT*HU
870 SS=S2(I)+HT*S3(I)
880 DQ=(S2(I)+S2(I+1)+SS)/6
890 SJ=Y(I)+HT*DY(I)+PR*DQ
900 RETURN

```

## Voorbeeld 18

\*\*\*\*SPLINE FUNCTIE\*\*\*\*

-----  
DATA PUNTEN LINES 50-90  
FORMAAT:N,FOUT,X(I),Y(I):I=1TON  
@ BETEKENT CARIAGE RETURN

MODE 1: INTERP.UITGEBREID Y,Y',Y''  
MODE 2: INTEGRAAL X(A)-X(B) 'INTERVAL'  
MODE 3: INTERPOLATIE OP Y ALLEEN  
MET FOR NEXT LOOP & TAFEL

TYP 1, 2 OR 3 AND @ 3

START INTERP.LOOP MET 0  
EIND OF INTERPOLATIE LOOP 4  
TYP DELTA VOOR LOOP  
20 PUNTEN IS PRIMA!: .2

X	Y		
0	576	2	92
.2	524.848	2.2	42.352
.4	473.984	2.4	-8.54400009
.6	423.696	2.6	-60.0160002
.8	374.272	2.8	-111.392
1	326	3	-162
1.2	279.008	3.2	-211.328
1.4	232.784	3.4	-259.504
1.6	186.656	3.6	-306.816
1.8	139.952	3.8	-353.552
		4	-400

## 19. Harmonische interpolatie

Na het starten van het programma kunt u het aantal punten dat u wilt gaan definiëren opgeven. Dit moet een oneven aantal zijn. Geeft u per ongeluk een even aantal op, dan zal het programma u vragen het aantal punten opnieuw op te geven.

Hierna moet u de startwaarde en de stapgrootte voor  $X$  opgeven.

Het programma zal nu voor iedere  $X$ -waarde de bijbehorende  $Y$ -waarde van u opvragen, waarna het de coëfficiënten berekent.

Daarna vraagt het programma u om een  $X$ -waarde in te tikken, waarvoor het de bijbehorende  $Y$ -waarde zal berekenen met behulp van de zojuist berekende coëfficiënten.

### Programma 19

```
10 PRINT"***HARMONISCHE INTERPOLATIE***"
20 PRINT"-----"
30 PRINT:PRINT"ONEVEN AANTAL PUNTEN Y0-Y2N"
40 PRINT"WAARVOOR GELDT OP GELIJKE AFSTAND"
50 INPUT"HOEVEEL PUNTEN";N
60 IF(N/2-INT(N/2))=0THEN50
70 INPUT"STARTWAARDE VOOR X";X0:XS=X0
80 INPUT"STAPGROOTTE DELTA X";DX
85 DIMF(N),A(N),B(N)
90 FORI=1TON
100 PRINT"VOOR X=";X0;" : Y IS";:INPUTF(I-1)
110 X0=X0+DX
120 NEXTI:PRINT"BEN BEZIG!!"
130 C=1:S=0:P=0
140 C1=COS((2*PI)/N):S1=SIN((2*PI)/N)
150 U2=0:U1=0:N1=N-1
160 FORI=N1 TO 1 STEP-1
170 U0=F(I)+2*C*U1-U2
180 U2=U1:U1=U0
190 NEXTI
200 A(P)=2*(F(0)+C*U1-U2)/N
210 B(P)=2*S*U1/N
215 PRINTA(P),B(P)
220 IFP=INT(N/2)THEN 260
230 Q=C1*C-S1*S
240 S=C1*S+S1*C
250 C=Q:P=P+1:GOTO 150
260 INPUT"INTERPOLATIE X";X1
270 X=(X1-XS)/DX
280 S=A(0)/2
290 FORI=1TOINT(N/2)
300 S=S+A(I)*COS(2*PI*X*I/N)
310 S=S+B(I)*SIN(2*PI*X*I/N)
320 NEXTI:PRINT
330 PRINT"VOOR X=";X1;" : Y IS";S
340 PRINT"-----"
350 PRINT:GOTO 260
```



## Voorbeeld 19

\*\*\*HARMONISCHE INTERPOLATIE\*\*\*

---

ONEVEN AANTAL PUNTEN Y0-Y2N  
WAARVOOR GELDT OP GELIJKE AFSTAND  
HOEVEEL PUNTEN 6  
HOEVEEL PUNTEN 5  
STARTWAARDE VOOR X 0  
STAPGROOTTE DELTA X 1  
VOOR X= 0 :Y IS 4  
VOOR X= 1 :Y IS 6  
VOOR X= 2 :Y IS 3  
VOOR X= 3 :Y IS 2  
VOOR X= 4 :Y IS 3  
BEN BEZIG!!

7.2                      0

1.09442719                      1.37638192

-.694427193                      .324919701

INTERPOLATIE X 0

VOOR X= 0 :Y IS 4

---

INTERPOLATIE X .5

VOOR X= .5 :Y IS 5.38885438

---

INTERPOLATIE X 3.5

VOOR X= 3.5 :Y IS 2.70557281

---

INTERPOLATIE X 4

VOOR X= 4 :Y IS 3

---

INTERPOLATIE X 1.7

VOOR X= 1.7 :Y IS 4.17737086

---

INTERPOLATIE X 3.3667

VOOR X= 3.3667 :Y IS 2.53637436

---

## 20. Driedimensionale interpolatie

Er van uitgaande dat  $Z$  een functie is van  $X$  en  $Y$ , en dat de oorsprong ligt op  $X=0$  en  $Y=0$ , heeft het programma nog de volgende gegevens nodig om voor iedere gewenste  $X$ - en  $Y$ -waarde  $Z$  te kunnen berekenen:

De maximale  $X$ -waarde

De maximale  $Y$ -waarde

De waarde van  $Z$  bij  $X=0$  en  $Y=0$  (is  $Z_0$ )

De waarde van  $Z$  bij  $X=X_{\max}$  en  $Y=0$  (is  $Z_1$ )


De waarde van  $Z$  by  $X=0$  en  $Y=Y_{\max}$  (is  $Z_2$ )

De waarde van  $Z$  by  $X=X_{\max}$  en  $Y=Y_{\max}$  (is  $Z_3$ )

Voor alle waarden die liggen tussen  $X=0$  en  $X=X_{\max}$ , en  $Y=0$  en  $Y=Y_{\max}$  berekent het programma de bijbehorende waarde van  $Z$  met behulp van de formule:

$$Z = (1-P)(1-Q)Z_0 + P(1-Q)Z_1 + Q(1-P)Z_2 + P \cdot Q \cdot Z_3$$

Hierin staat  $P$  voor  $\frac{X}{X_{\max}}$  en  $Q$  voor  $\frac{Y}{Y_{\max}}$ .



afb. 2

### Programma 20

```

10 PRINT "*** 3 DIM. INTERPOLATIE ***"
15 PRINT "-----"
20 PRINT "VOOR Z= FUNCTIE VAN X EN Y"
30 PRINT "DE OORSPRONG IS OP X=0 EN Y=0"
40 PRINT: INPUT "X MAXIMUM"; XM
50 INPUT "Y MAXIMUM"; YM
60 INPUT "Z VOOR PNT. 0,0"; Z0
70 INPUT "Z VOOR PNT. 0,XMAX"; Z1
80 INPUT "Z VOOR PNT. 0,YMAX"; Z2
90 INPUT "Z VOOR PNT. XMAX,YMAX"; Z3
100 PRINT: INPUT "INTERPOLATIE VOOR X,Y"; X,Y
110 P=X/XM:Q=Y/YM
120 Z=(1-P)*(1-Q)*Z0+P*(1-Q)*Z1+Q*(1-P)*Z2+P*Q*Z3

```

```

130 Z=INT(Z*100+.5)/100
140 PRINT"VOOR DEZE X,Y IS Z: ";Z
150 PRINT"-----"
160 PRINT:GOTO100

```

## Voorbeeld 20

\*\*\* 3 DIM.INTERPOLATIE\*\*\*

-----  
 VOOR Z= FUNCTIE VAN X EN Y  
 DE OORSPRONG IS OP X=0 EN Y=0

X MAXIMUM 100  
 Y MAXIMUM 50  
 Z VOOR PNT. 0,0 20  
 Z VOOR PNT. 0,XMAX 7  
 Z VOOR PNT. 0,YMAX 5  
 Z VOOR PNT. XMAX,YMAX 12

INTERPOLATIE VOOR X,Y 0 0  
 VOOR DEZE X,Y IS Z: 20  
 -----

INTERPOLATIE VOOR X,Y 100 0  
 VOOR DEZE X,Y IS Z: 7  
 -----

INTERPOLATIE VOOR X,Y 0 50  
 VOOR DEZE X,Y IS Z: 5  
 -----

INTERPOLATIE VOOR X,Y 100 50  
 VOOR DEZE X,Y IS Z: 12  
 -----

INTERPOLATIE VOOR X,Y 25 25  
 VOOR DEZE X,Y IS Z: 11.75  
 -----

INTERPOLATIE VOOR X,Y 10 10  
 VOOR DEZE X,Y IS Z: 16.1  
 -----

INTERPOLATIE VOOR X,Y 7.89 23.14  
 VOOR DEZE X,Y IS Z: 12.76  
 -----




## 21. Vierde orde vlak door negen punten

Voor ieder van de in afbeelding 3 vermelde coördinaten vraagt het programma de waarde van  $\bar{Z}$ .

Daarna kunt u iedere gewenste waarde voor de coördinaten  $X, Y$  opgeven, waarvoor het programma dan de bijbehorende waarde van  $\bar{Z}$  berekent.

De maximale waarde voor zowel  $X$  als  $Y$  is 1.



afb. 3

### Programma 21

```

10 PRINT"**** 4E.ORDE  VLAK DOOR 9 PNT.****"
20 PRINT"-----"
40 PRINT:INPUT"Z(-1,+1) ";Z
50 INPUT"Z( 0,+1) ";A
60 INPUT"Z(+1,+1) ";B
70 INPUT"Z(-1, 0) ";C
80 INPUT"Z( 0, 0) ";D
90 INPUT"Z(+1, 0) ";E
100 INPUT"Z(-1,-1) ";F
110 INPUT"Z( 0,-1) ";G
120 INPUT"Z(+1,-1) ";H
130 PRINT:INPUT"INTERPOLATIE X,Y ";X,Y
140 J=X:L=X+1:M=X-1
150 N=Y:O=Y+1:P=Y-1
160 Q=Z*M*O+B*L*O+F*M*P+H*L*P
170 Q=Q*J*N/4
180 Q=Q-(((A*L*N+C*J*P)*M*O+(E*J*O+G*M*N)*L*P))/2
190 Q=Q+D*L*M*O*P
200 Q=INT(100*Q+.5)/100
210 PRINT"?VOOR DEZE X,Y IS Z ";Q
220 PRINT"-----"
230 PRINT:GOTO130

```

## Voorbeeld 21

\*\*\*\* 4E. ORDE VLAK DOOR 9 PNT. \*\*\*\*

---

Z(-1,+1) 17  
Z( 0,+1) 1  
Z(+1,+1) 8  
Z(-1, 0) 21  
Z( 0, 0) 30  
Z(+1, 0) 11  
Z(-1,-1) 13  
Z( 0,-1) 23  
Z(+1,-1) 5

INTERPOLATIE X,Y 0 0  
VOOR DEZE X,Y IS Z 30

---

INTERPOLATIE X,Y 0 .5  
VOOR DEZE X,Y IS Z 20

---

INTERPOLATIE X,Y 0 -.5  
VOOR DEZE X,Y IS Z 31

---

INTERPOLATIE X,Y .5 0  
VOOR DEZE X,Y IS Z 24

---

INTERPOLATIE X,Y .23 .48  
VOOR DEZE X,Y IS Z 19.17

---

INTERPOLATIE X,Y .77 -.89  
VOOR DEZE X,Y IS Z 13.27

---

INTERPOLATIE X,Y 1 .5  
VOOR DEZE X,Y IS Z 10.63

---

INTERPOLATIE X,Y .11 .15  
VOOR DEZE X,Y IS Z 27.25

---

## 22. Parabolische extrapolatie

Na het ingeven van de coördinaten  $X, Y$  van drie punten, waarbij de onderlinge afstand in de  $X$ -richting gelijk dient te zijn, berekent het programma de coëfficiënten  $A, B$  en  $C$  voor de functie:

$$Y = AX^2 + BX + C$$

Daarna kan, indien men daarom vraagt, het programma de  $Y$ -waarde voor iedere gewenste  $X$ -waarde berekenen.

Wil men niet zelf een  $X$ -waarde opgeven waarvoor het programma de bijbehorende  $Y$ -waarde moet berekenen, dan zal het programma voor een zelfgegenereerde tabel van  $X$ -waarden de bijbehorende  $Y$ -waarden berekenen.

### Programma 22

```
10 PRINT"***PARABOLISCHE EXTRAPOLATIE***"
20 PRINT"-----"
30 PRINT:PRINT"DRIE PUNTEN X,Y (GELIJKE AFSTAND)"
35 PRINT"BEREKENT OOK DE COEFFICIENTEN!"
36 PRINT"mits X#1 is ongelijk nul"
40 DIMX(10),Y(10):I=1
50 PRINT"TYP PUNT #";I;"X,Y";
60 INPUT X(I),Y(I):I=I+1
70 IF I<4 THEN 50
80 H=X(3)-X(2):PRINT:GOTO 250
90 GOTO 150
100 Y(4)=Y(1)-3*Y(2)+3*Y(3)
110 Y(1)=Y(2)
120 Y(2)=Y(3)
130 Y(3)=Y(4)
140 PRINT X(4);TAB(15);Y(4)
145 RETURN
150 PRINT" X-WAARDE          Y-WAARDE"
160 FOR I=1 TO 15
170 X(4)=X(3)+H
180 X(3)=X(4)
190 GOSUB 100
200 NEXT:PRINT
210 PRINT"VOOR MEER TYP J, ANDERS N";
220 INPUT A$
230 IF A$="J" THEN 90
240 END
250 PRINT"DE PARABOOL Y=A*X*X+B*X+C"
251 PRINT"-----"
255 PRINT"VOOR COEFFICIENTEN TYP J";
256 INPUT B$
257 IF B$<>"J" THEN 150
260 P1=X(1)*X(1):Q1=X(1):R1=1:S1=Y(1)
270 P2=X(2)*X(2):Q2=X(2):R2=1:S2=Y(2)
280 P3=X(3)*X(3):Q3=X(3):R3=1:S3=Y(3)
290 T1=((Q1*P2)/P1)-Q2
```



```

300 T2=((R1*P2)/P1)-R2
310 T3=((Q1*P3)/P1)-Q3
320 T4=((R1*P3)/P1)-R3
330 T5=((T1*T4)-(T2*T3))
340 IF T5=0 THEN PRINT"FOUT 0 DETERMINANT"
350 T6=((S1*P2)/P1)-S2
360 T7=((S1*P3)/P1)-S3
370 B=((T6*T4)-(T2*T7))/T5
380 C=((T1*T7)-(T6*T3))/T5
390 A=(S1/P1)-(Q1/P1*B)-(R1/P1*C)
400 PRINT "COEFFICIENT A= ";A
410 PRINT "COEFFICIENT B= ";B
420 PRINT "COEFFICIENT C= ";C
430 INPUT"VOOR INTERPOLATIE TYP J";C$
440 IF C$="J"THEN INPUT"X= ";X
450 IF C$<>"J"THEN 150
460 PRINT"Y= ";A*X*X+B*X+C:GOTO 440

```

## Voorbeeld 22

\*\*\*PARABOLISCHE EXTRAPOLATIE\*\*\*

DRIE PUNTEN X,Y (GELIJKE AFSTAND)  
BEREKENT OOK DE COEFFICIENTEN!  
MITS X#1 IS ONGELIJK NUL

TYP PUNT # 1 X,Y	1	9
TYP PUNT # 2 X,Y	3	5
TYP PUNT # 3 X,Y	5	1.51

DE PARABOOL  $Y=A*X*X+B*X+C$

-----  
VOOR COEFFICIENTEN TYP J  
COEFFICIENT A= .0637499988  
COEFFICIENT B= -2.255  
COEFFICIENT C= 11.19125  
VOOR INTERPOLATIE TYP J  
X= 1  
Y= 9

X= 3	Y= 5
X= 5	Y= 1.50999999

X= 1.5	Y= 7.9521875
--------	--------------

X= 2.5	Y= 5.9521875
--------	--------------

X= 3.5	Y= 4.0796875
--------	--------------

X= 4	Y= 3.19125
------	------------

X= 6	Y= -.0437500179
------	-----------------

X= 12.5	Y= -7.03531263
---------	----------------

X= 17	Y= -8.72000027
-------	----------------

X= 7	Y= -1.47000003
------	----------------

## 23. Polynoom-evaluatie (reëel)

Afhankelijk van de opgegeven orde ( $N$ ) zal het programma u vragen de waarden van de coëfficiënten  $A_n$  tot en met  $A_0$  te specificeren.

Hierna berekent het programma de waarde van  $P_{(x)}$  voor de door u op te geven waarde van  $X$  volgens de formule:

$$P = A_0 + A_1X + A_2X^2 + \dots + A_nX^n$$

Ook worden voor die waarde van  $X$  de eerste en tweede afgeleiden berekend.

### Programma 23

```
10 PRINT"***POLYNOOM EVALUATIE***"
20 PRINT"-----"
30 PRINT:INPUT"ORDE  N";N
40 DIMA(N),F(N+2)
50 FORI=NT00 STEP-1
60 PRINT"A";I;" = ";:INPUT A(I)
70 NEXT I
80 PRINT
90 INPUT"X= ";X
100 B=0: C=0:E=0
110 FOR I=N TO 0 STEP -1
120 B=A(I)+B*X:D=C
130 C=B+C*X
135 E=C+E*X:F(I)=E
140 NEXT I
150 PRINT"P(";X;")=";TAB(18);B
160 PRINT"1E.AFGELEIDE=";TAB(18);D
165 PRINT"2E.AFGELEIDE=";TAB(18);F(2)*2
170 PRINT:GOTO 90
```

### Voorbeeld 23

\*\*\*POLYNOOM EVALUATIE\*\*\*

-----

```
ORDE  N 4
A 4  =  3
A 3  = -24
A 2  =  59
A 1  = -288
A 0  =  576
```

```
X= 0
P( 0 )=          576
1E.AFGELEIDE=      -288
2E.AFGELEIDE=       118
```

X= 1		
P( 1 )=	326	
1E.AFGELEIDE=		-230
2E.AFGELEIDE=		10
X= 2		
P( 2 )=	92	
1E.AFGELEIDE=		-244
2E.AFGELEIDE=		-26
X= 3		
P( 3 )=	-162	
1E.AFGELEIDE=		-258
2E.AFGELEIDE=		10
X= 4		
P( 4 )=	-400	
1E.AFGELEIDE=		-200
2E.AFGELEIDE=		118
X= 2.45		
P( 2.45 )=	-20.3094813	
1E.AFGELEIDE=		-254.6065
2E.AFGELEIDE=		-18.71
X= 2.37		
P( 2.37 )=	-3.47517151E-03	
1E.AFGELEIDE=		-253.012164
2E.AFGELEIDE=		-21.0716
X= 6		
P( 6 )=	-324	
1E.AFGELEIDE=		420
2E.AFGELEIDE=		550



## 24. Polynoom-evaluatie (complexe coëfficiënten)

Nadat u hebt opgegeven wat de orde van de polynoom is, vraagt het programma om de coëfficiënten. De coëfficiënten bestaan uit een reëel deel ( $A$ ) en een imaginair deel ( $IB$ ).  $A + IB$  moet ongelijk aan nul zijn. De coëfficiënten moeten achtereenvolgens worden ingegeven van de hoogste tot de laagste orde.

Er van uitgaande dat  $Z = X + iY$ , berekent het programma de waarde van  $P_{(z)}$  voor de opgegeven waarden van  $X$  en  $Y$ .

De berekende waarde van  $P_{(z)} = U + iV$ , waarin  $U$  het reële en  $V$  het imaginaire deel is.

Na iedere berekening kunnen nieuwe waarden voor  $X$  en  $Y$  worden ingegeven, waarvoor  $P_{(z)}$  weer wordt berekend.

### Programma 24

```
10 PRINT"***POLYNOMEN, COMPLEXE COEFF.***"
20 PRINT"-----"
30 REM COEFFICIENTEN ZIJN A+IB<>0"
40 INPUT"ORDE  =";N
45 DIM X(N+3),Y(N+3),A(N),B(N)
50 FOR I=N TO 0 STEP-1
60 PRINT" A(";I;")=";:INPUT A(I)
70 PRINT"IB(";I;")=";:INPUT B(I)
80 PRINT
90 NEXT I
100 INPUT"VOOR Z=X+iY: TYP X,Y";X,Y
105 PRINT"+++++"
110 X(0)=1: X(1)=X
120 Y(0)=0: Y(1)=Y
130 FOR I=0 TO N
140 X(I+2)=2*X*X(I+1)-(X*X+Y*Y)*X(I)
150 Y(I+2)=2*X*Y(I+1)-(X*X+Y*Y)*Y(I)
160 NEXT I
170 U=0: V=0
180 FOR I=0 TO N
190 U=U+A(I)*X(I)-B(I)*Y(I)
200 V=V+A(I)*Y(I)+B(I)*X(I)
210 NEXT I
215 PRINT:PRINT"P(Z) IS U+iV-----"
220 PRINT"REELE U = ";U
230 PRINT"IMAG. V = ";V
235 PRINT"-----"
240 PRINT
250 GOTO 100
```

## Voorbeeld 24

\*\*\*POLYNOMEN, COMPLEXE COEFF.\*\*\*

ORDE = 3

A( 3 ) = 1

IB( 3 ) = 1

A( 2 ) = 3

IB( 2 ) = 5

A( 1 ) = 5

IB( 1 ) = 7.1

A( 0 ) = 4

IB( 0 ) = 1.5

VOOR Z=X+IY: TYP X,Y 4 3

+++++

P(Z) IS U+IV-----

REELE U = -257.3

IMAG. V = 224.9

VOOR Z=X+IY: TYP X,Y 4.25 3.59

+++++

P(Z) IS U+IV-----

REELE U = -373.113296

IMAG. V = 227.749196

VOOR Z=X+IY: TYP X,Y 1.5 4

+++++

P(Z) IS U+IV-----

REELE U = -149.775

IMAG. V = -106.225

VOOR Z=X+IY: TYP X,Y 1 1

+++++

P(Z) IS U+IV-----

REELE U = -12.1

IMAG. V = 19.6

VOOR Z=X+IY: TYP X,Y 7.3 5

+++++

P(Z) IS U+IV-----

REELE U = -1107.963

IMAG. V = 954.647

## 25. Grenswaarde vlak 'Laplace'

De in te geven waarde voor  $N$  bepaalt de grootte van de matrix. Nadat u  $N$  hebt gespecificeerd, vraagt het programma u de randvoorwaarden ( $R$ ) op te geven. Afbeelding 4 geeft een voorbeeld van een matrix van 3 bij 3, met daaromheen de nummering van de randvoorwaarden ( $R_{(ij)}$ ).

$I \downarrow$	$J \rightarrow$			
	$R_{(0\ 1)}$	$R_{(0\ 2)}$	$R_{(0\ 3)}$	
	$R_{(1\ 0)}$			$R_{(1\ 4)}$
	$R_{(2\ 0)}$			$R_{(2\ 4)}$
	$R_{(3\ 0)}$			$R_{(3\ 4)}$
	$R_{(4\ 1)}$	$R_{(4\ 2)}$	$R_{(4\ 3)}$	

afb. 4

De mogelijkheid bestaat om behalve voor de randvoorwaarden ook voor een punt binnen de matrix een vaste waarde op te geven. Hiervoor moeten zowel de waarde als de coördinaten worden opgegeven, van elkaar gescheiden door een komma (waarde,  $I,J$ ).

Na het opgeven van de gewenste nauwkeurigheid berekent het programma de waarde van de matrix met behulp van 'Laplace', uitgaande van de randvoorwaarden.

De waarden van de matrix worden in matrixvorm uitgeprint, gevolgd door een tabel waarin zowel de waarden van de matrix als de randvoorwaarden staan.

### Programma 25

```

5 PRINT"****GRENSWAARDEVLAKE: 'LAPLACE'****"
6 PRINT"-----"
7 PRINT
10 INPUT"ANTAL N+2 PTS KLOKWIJZE: TYP N";N
20 DIMR(N+1,N+1)
30 I=0
40 FORJ=1TON
50 GOSUB350:NEXT
60 J=N+1
70 FORI=1TON

```



```

80 GOSUB350:NEXT
90 I=N+1
100 FORJ=1TON
110 GOSUB350:NEXT
120 J=0
130 FORI=1TON
140 GOSUB350:NEXT
145 PRINT"WENST U 1 SPECIAAL PUNT TYP J/N":INPUTB$
146 IF B$="N"THEN150
147 INPUT"R(I,J):TYP WAARDE,I,J";R,K,L:R(K,L)=R
150 INPUT"RELAXATIEFACTOR 1.5-3.5 ";W
160 INPUT"GEWENSTE AFBREEKFOUT ";E
165 PRINT"ITERATIEFOUT:"
170 FORI=1TON STEP2 :REM ONEVEN RIJEN
180 FORJ=1TON
190 GOSUB300
200 NEXTJ:NEXTI
210 FORI=2TON STEP2 :REM EVEN RIJEN
220 FOR J=1TON
230 GOSUB300
240 NEXTJ:NEXTI
245 IFABS(S-U-B)<=ETHEN 265
250 IF ABS(S-U)<=ETHEN265
255 B=S-U
260 PRINT"FOUT";(S-U):U=S:GOTO170
265 PRINT:PRINT:PRINT"RESULTAAT MATRIX";N;"X";N:PRINT
270 FORI=1TON:FORJ=1TON
280 PRINTINT(R(I,J)*100+.5)/100;
290 NEXTJ:PRINT:PRINT:NEXTI
295 GETA$:IFA$=""THEN295
296 GOTO370
300 Z=W*(.25*(R(I-1,J)+R(I,J-1)+R(I+1,J)+R(I,J+1))-R(I,J))
310 S=S+Z*Z
315 IFK=IANDL=JTHEN330
320 R(I,J)=R(I,J)+Z
330 RETURN
350 PRINT"TYP R";I;J;" = ";:INPUT R(I,J)
360 RETURN
370 FORI=0TON+1:FORJ=0TON+1
380 PRINT"R";I;J;" = ";R(I,J)
390 NEXTJ:NEXTI:END

```

## Voorbeeld 25

\*\*\*\*GRENSWAARDEVLAKE: 'LAPLACE'\*\*\*\*

```

AANTAL N*N PTS KLOKWIJZE:TYP N 3
TYP R 0 1 = 1
TYP R 0 2 = 2
TYP R 0 3 = 3
TYP R 1 4 = 4
TYP R 2 4 = 5
TYP R 3 4 = 6

```

TYP R 4 1 = 9  
TYP R 4 2 = 8  
TYP R 4 3 = 7  
TYP R 1 0 = 3  
TYP R 2 0 = 5  
TYP R 3 0 = 7

WENST U 1 SPECIAAL PUNT TYP J/N  
N

RELAXATIEFACTOR 1.5-3.5 1.5

GEWENSTE AFBREEKFOUT .01

ITERATIEFOUT:

FOUT 215.589807

FOUT 26.6853931

FOUT 6.46188462

FOUT 2.91613621

FOUT 1.51445317

FOUT .451113522

FOUT .0450331569

FOUT .0214226842

RESULTAAT MATRIX 3 X 3

3.12 3.49 3.87

5 4.99 5

6.88 6.49 6.12

R 0 0 = 0

R 0 1 = 1

R 0 2 = 2

R 0 3 = 3

R 0 4 = 0

R 1 0 = 3

R 1 1 = 3.124614

R 1 2 = 3.48927739

R 1 3 = 3.86862474

R 1 4 = 4

R 2 0 = 5

R 2 1 = 4.99625629

R 2 2 = 4.98865505

R 2 3 = 5.00346723

R 2 4 = 5

R 3 0 = 7

R 3 1 = 6.87824968

R 3 2 = 6.49352202

R 3 3 = 6.12425801

R 3 4 = 6

R 4 0 = 0

R 4 1 = 9

R 4 2 = 8

R 4 3 = 7

R 4 4 = 0

## 26. Polynomen vermenigvuldigen

Voor de polynoom  $A_0 + A_1X + A_2X^2 + \dots + A_nX^n$  kunnen de coëfficiënten  $A_0$  tot en met  $A_n$  worden ingegeven. Het beëindigen van de ingave geschiedt door op de vraag naar de waarde van de coëfficiënt 99 in te geven.

Onmiddellijk hierna vraagt het programma om de coëfficiënten van de volgende polynoom, waarna het produkt van POL.I \* POL.II wordt berekend. De uitkomst, in coëfficiënten, wordt geprint of op het beeldscherm zichtbaar gemaakt.

Indien u dat wilt kunt u hierna POL.I met een andere POL.II vermenigvuldigen.

### Programma 26

```
10 PRINT"***PRODUCT:POL.I * POL.II ENZ.***"
20 PRINT"===== "
25 PRINT"POL.=A0+A1*X+A2*X^2+.....ENZ."
30 DIMR(100):N=7:T=42:B=0
40 PRINT:PRINT"COEFFICIENTEN POL.I;STOP MET 99"
50 N=N+1
60 PRINT"A";B;"="":INPUTR(N)
70 IFR(N)=99 THENR(N)=0:GOTO110
80 A=A+1
90 IFA>TTHENPRINT"TE VEEL DATA":END
100 B=B+1:GOTO 50
110 B=0:N=A:C=A
120 C=C-1:D=7:E=48
130 FORI=NT00STEP-1
140 D=D+1:E=E+1
150 R(E)=R(D):R(D)=0
160 NEXT: D=7
180 PRINT:PRINT"COEFFICIENTEN POL.II;STOP MET 99"
185 F=D:E=48:T=0
190 PRINT"A";B;"="":INPUT G
200 IFG=99 THEN G=0:GOTO300
210 IFG=T THEN 280
220 T=49:N=C
230 IFC+F>TTHENPRINT"TE VEEL DATA":END
240 FORI=NT00STEP-1
250 F=F+1:E=E+1
260 R(F)=R(F)+(G*R(E))
270 NEXT
280 A=A+1:B=B+1:D=D+1
290 GOTO 185
300 N=A-2:C=7
310 PRINT"HET PRODUCT IS IN COEFF.:"
320 FORI=NT00STEP-1
330 C=C+1
340 PRINT"A";C-8;"="":R(C)
350 NEXT
360 PRINT:PRINT"VOOR MEER TYP 1":INPUTM
370 IF M=1 THENA=A-1:GOTO110
```



## Voorbeeld 26

\*\*\*PRODUCT:POL.I \* POL.II ENZ.\*\*\*

=====

POL.=A0+A1\*X+A2\*X<sup>2</sup>+.....ENZ.

COEFFICIENTEN POL.I;STOP MET 99

A 0 = 2

A 1 = 5

A 2 = 3

A 3 = 99

COEFFICIENTEN POL.II            STOP MET 99

A 0 = 5

A 1 = 1

A 2 = 4

A 3 = 99

HET PRODUCT IS IN COEFF.:

A 0 = 10

A 1 = 27

A 2 = 28

A 3 = 23

A 4 = 12

VOOR MEER TYP 1 1

COEFFICIENTEN POL.II            STOP MET 99

A 0 = 5

A 1 = 99

HET PRODUCT IS IN COEFF.:

A 0 = 50

A 1 = 135

A 2 = 140

A 3 = 115

A 4 = 60

VOOR MEER TYP 1 7

## 27. Impleciete functies (Gauss/Seidel)

Dit programma kan maximaal 5 functies berekenen. Die functies staan op de programmaregels 100, 200, 300, 400 en 500. Indien u dat wenst kunt u deze functies zelf invullen door die programmaregels te wijzigen. In het gegeven programma zijn de volgende vijf functies opgenomen:

$$F(Y_0, Z_0, U_0, V_0) = X$$

$$F(X, Z_0, U_0, V_0) = Y$$

$$F(X, Y, U_0, V_0) = Z$$

$$F(X, Y, Z, V_0) = U$$

$$F(X, Y, Z, U) = V$$

Afhankelijk van het aantal gewenste functies zal het programma u vragen de nulwaarden voor de betreffende functies te specificeren. Daarna kunt u nog de maximaal toegestane grootte van de fout opgeven.

### Programma 27

```
10 PRINT"***IMPLICIETE FUNCTIES(MAX.5)***"
15 PRINT"-----"
20 PRINT"GAUSZ SEIDEL ITERATIE":PRINT
30 INPUT"HOEVEEL FUNCTIES? ";N
40 INPUT"Y0= ";Y0:IF N=2 THEN80
50 INPUT"Z0= ";Z0:IF N=3 THEN80
60 INPUT"U0= ";U0:IF N=4 THEN80
70 INPUT"V0= ";V0
80 INPUT"FOUT= ";ER:K=2
90 REM LIJN 100 IS X=F(Y0,Z0,U0,V0)
95 PRINT:PRINT"DE ITERATIEFOUT IS:"
100 X=.91-COS(Y0)
190 REM LIJN 200 IS Y=F(X,Z0,U0,V0)
200 Y=ATN(.67-X*X+EXP(Z0))
280 IFK=NTHEN600
290 REM LIJN 300 IS Z=F(X,Y,U0,Z0)
300 Z=(-4.21-Y)/5
380 K=K+1:IFK=NTHEN600
390 REM LIJN 400 IS U=F(X,Y,Z,V0)
400 U=0
480 K=K+1:IFK=NTHEN600
490 REM LIJN 500 IS V=F(X,Y,Z,U)
500 V=0
600 E=ABS(Y-Y0)+ABS(Z-Z0)+ABS(U-U0)+ABS(V-V0)
610 IFE<=ER THEN PRINT:GOTO690
620 Y0=Y:Z0=Z:U0=U:V0=V
630 T=T+1:PRINTE
640 IF T=100 THEN 800
650 GOTO 100
690 PRINT"DE RESULTATEN ZIJN:"
700 PRINT"X= ";X
710 PRINT"Y= ";Y0:IFK=2 THEN END
```

```

720 PRINT"Z= ";Z0:IFK=3 THEN END
730 PRINT"U= ";U0:IFK=4 THEN END
740 PRINT"V= ";V0:END
800 PRINT"100 ITERATIES !!!!!"
810 PRINT"KIES ANDERE STARTWAARDEN"
820 GOTO 40

```

### Voorbeeld 27.1

\*\*\*IMPLICIETE FUNCTIES(MAX.5)\*\*\*

---

GAUSZ SEIDEL ITERATIE

HOEVEEL FUNCTIES? 3

Y0= 0

Z0= 0

FOUT= .01

DE ITERATIEFOUT IS:

2.07693489

.315898134

.0707144623

DE RESULTATEN ZIJN:

X= .201854938

Y= .783928732

Z= -1.04782248

### Voorbeeld 27.2

\*\*\*IMPLICIETE FUNCTIES(MAX.5)\*\*\*

---

GAUSZ SEIDEL ITERATIE

HOEVEEL FUNCTIES? 3

Y0= 1

Z0= 1

FOUT= 1E-03

DE ITERATIEFOUT IS:

2.36892342

.714175798

.227247999

.01903704

2.70944135E-03

DE RESULTATEN ZIJN:

X= .191519286

Y= .769180784

Z= -1.09648724



## 28. Tabel-interpolatie

Het programma vraagt u om de X- en Y-waarden in te voeren, totdat u de waarde 999,0 hebt ingegeven. De waarde die u ingeeft voor X en Y moet steeds hoger zijn dan de daarvoor ingegeven waarde.

Van de ingegeven waarden maakt het programma een tabel.

Hierna vraagt het programma u om een X-waarde, waarvoor het dan de bijbehorende Y-waarde berekent. Geeft u een X-waarde op die niet in het interval zit, dan wordt het programma beëindigd.

### Programma 28

```
10 PRINT"INTERPOLATIE UIT TABEL"
20 PRINT"-----"
25 DIMTD(60),FD(60)
30 PRINT:PRINT"TYP X(I),Y(I).... X(I)<X(I+1)..."
40 PRINT"EN EINDIG MET 999.0"
50 PRINT
60 FOR I=1TO60
70 PRINT"X";I;" ,Y";I;" IS ";;INPUTTD(I),FD(I)
80 IFTD(I)=999THEN 100
90 NEXT
100 N=I-1
110 PRINT:PRINT"X IS ";;INPUTT
120 IFT<TD(1) THEN 210
130 IFT>TD(N) THEN 210
140 FORI=2TON
150 K=I
160 IFTD(I)>TTHEN180
170 NEXT
180 KM=K-1
190 PW=FD(KM)+(FD(K)-FD(KM))*(T-TD(KM))/(TD(K)-TD(KM))
200 PRINT"Y IS ";PW: GOTO 110
210 PRINT"FOUT, X NIET IN INTERVAL!!":END
```

### Voorbeeld 28

INTERPOLATIE UIT TABEL

-----  
TYP X(I),Y(I).... X(I)<X(I+1)..  
EN EINDIG MET 999,0

X 1 ,Y 1 IS	0	
X 2 ,Y 2 IS	1	
X 3 ,Y 3 IS	1.5	8
X 4 ,Y 4 IS	4	9
X 5 ,Y 5 IS	6	7

X 6 ,Y 6 IS 8.5  
X 7 ,Y 7 IS 10  
X 8 ,Y 8 IS 999

5  
2  
0

X IS 1  
Y IS 4

X IS 4  
Y IS 9

X IS 5  
Y IS 8

X IS 3.57  
Y IS 8.828

X IS 9  
Y IS 4

X IS 1.1  
Y IS 4.8

X IS 1.2  
Y IS 5.6

X IS 1.3  
Y IS 6.4

X IS 6.897  
Y IS 6.2824

X IS 50  
FOUT, X NIET IN INTERVAL!!

## 29. Newton – Raphson

Dit programma berekent de  $X$ -waarde voor de functie waarvoor geldt  $F=0$ . De functie  $F$  staat in programmaregel 250. In regel 100 wordt de eerste afgeleide benaderd. Regels 150–190 bevatten een methode waardoor de iteratiesnelheid tot een factor 2,8 wordt opgevoerd.

U wordt gevraagd de startwaarde voor  $X$ , de maximaal toegestane fout en het maximum aantal iteraties te specificeren.

Hierna berekent het programma de  $X$ -waarde voor  $F=0$ . Bovendien geeft het programma aan hoeveel iteraties voor die berekening nodig waren.

Op regel 250 van dit programma kan iedere gewenste functie worden geplaatst. In voorbeeld 2 is de functie  $F=4 \sin(X) + 1 - X$  op regel 250 gezet, waarna het programma opnieuw is uitgevoerd.

### Programma 29

```
10 PRINT"****NEWTON-RAPHSON F=0****"
20 PRINT"-----"
30 PRINT"FUNCTIE F=0 VAN X IN LIJN 250"
40 PRINT
50 INPUT"START X ";X0
60 INPUT"FOUT      ";E
70 INPUT"ITERMAX. ";M
80 X=X0:GOSUB250:Y0=F
90 X=X0+E:GOSUB250
100 D0=(F-Y0)/E
110 X1=X0-Y0/D0
120 FORI=1TOM
130 X=X1:GOSUB250:Y1=F
140 X=X+E:GOSUB250
150 D1=(F-Y1)/E
160 D=X1-X0
170 A=(2*D1+D0-3*(Y1-Y0)/D)/D
180 U=Y1/D1
190 X2=X1-U*(1+U*A/D1)
200 IFABS((X2-X1)/X2)<=ETHEN230
210 X0=X1:X1=X2:Y0=Y1:D0=D1
220 NEXT:PRINT"ITERMAX BEREIKT";:END
230 PRINT:PRINT"X VOOR F=0 IS ";X2
235 PRINT
240 PRINT"AANTAL ITERATIES";I:END
250 F=3*X^4-24*X^3+59*X^2-288*X+576
300 RETURN
```



### Voorbeeld 29.1

\*\*\*\*NEWTON-RAPHSON F=0\*\*\*\*

-----  
FUNCTIE F=0 VAN X IN LIJN 250

EXTRA SNEL DOOR LIJNEN 170-180!!

START X -2.1  
FOOT 1E-06  
ITERMAX. 100

X VOOR F=0 IS -2.21008395

AANTAL ITERATIES 2

250 F=4\*SIN(X)+1-X  
300 RETURN  
READY.

### Voorbeeld 29.2

\*\*\*\*NEWTON-RAPHSON F=0\*\*\*\*

-----  
FUNCTIE F=0 VAN X IN LIJN 250

EXTRA SNEL DOOR LIJNEN 170-180!!

START X 1  
FOOT 1E-06  
ITERMAX. 100

X VOOR F=0 IS 2.36998627

AANTAL ITERATIES 3

## 30. Regula falsi

Voor het starten van het programma moet op regel 200 de functie voor  $Y = F_{(x)}$  worden ingevuld. Hoewel in het bijgaande programma is gekozen voor  $Y = 3X^4 - 24X^3 + 59X^2 - 288X + 576$ , kunt u iedere gewenste functie invullen.

Nu kan het programma worden gestart. Het programma vraagt achtereenvolgens naar de startwaarde van  $X$ , de te nemen stapgrootte en de foutlimiet.

Hierna wordt het nulpunt voor  $F_{(x)}$  berekend. Nadat de eerste wortel is gevonden, vraagt het programma of het nog meer wortels moet berekenen. Antwoordt u 'J', dan wordt de waarde van de eerste wortel verhoogd met de stapgrootte en probeert het programma de tweede wortel te berekenen. Hierna kunt u weer een volgende wortel laten bepalen, enz.

### Programma 30

```
10 PRINT "**REGULA FALSI: F(X)=0: REELE WORTELS**"
15 PRINT "-----"
20 PRINT: PRINT "Y=F(X) IN LIJN 200"
30 PRINT: INPUT "STARTWAARDE X0"; X
33 INPUT "STAPGROOTTE X"; DX
34 INPUT "FOUTLIMIET"; E
35 PRINT "X-WAARDE"; TAB(20); "Y-WAARDE"
40 GOSUB 200: R2=Y
50 X=X+DX: GOSUB 200: R4=Y
60 IF R2*R4>0 THEN R2=Y: GOTO 50
70 R1=X-DX: R3=X
75 IF ABS(R4-R2)<=E THEN 120
80 X=R3-R4*(R3-R1)/(R4-R2)
90 GOSUB 200
100 R2=R4: R4=Y: R1=R3: R3=X
110 GOTO 75
120 PRINT "DE WORTEL IS"; X
122 PRINT "-----"
125 PRINT "MEER WORTELS? TYP J";: INPUT A$
130 IF A$="J" THEN X=X+DX: GOSUB 35
140 END
200 Y=3*X^4-24*X^3+59*X^2-288*X+576
210 PRINT X; TAB(20); Y
300 RETURN
```

### Voorbeeld 30

\*\*REGULA FALSI: F(X)=0: REELE WORTELS\*\*

-----

Y=F(X) IN LIJN 200

STARTWAARDE X0 0

STAPGROOTTE X .5

FOUTLIMIET 1E-06

X-WAARDE

Y-WAARDE

0

576

.5

443.9375

1

326

1.5

210.9375

2

92

2.5

-33.0625

2.36781609

.549029589

2.36997526

2.784729E-03

2.36998627

-2.38418579E-07

2.36998627

0

DE WORTEL IS 2.36998627

-----

MEER WORTELS? TYP JJ

X-WAARDE

Y-WAARDE

2.86998627

-128.397454

3.36998627

-256.108023

3.86998627

-373.05183

4.36998627

-464.648999

4.86998627

-511.819655

5.36998627

-490.983919

5.86998627

-374.061918

6.36998627

-128.473774

6.86998627

282.860394

6.52615342

-19.6637218

6.54850217

-2.70159483

6.5520617

.0330498219

6.55201868

-5.31673431E-05

6.55201875

-2.86102295E-06

6.55201875

2.38418579E-07

6.55201875

2.38418579E-07

DE WORTEL IS 6.55201875

-----

MEER WORTELS? TYP JN



## 31. Optimum/constraint

Voor het starten van het programma kunt u op regels 250, 300 en 350 de gewenste functies specificeren.

Is het programma eenmaal gestart dan zal het eerst naar de startwaarde voor  $X$  vragen en vervolgens naar de maximaal toegestane foutgrootte. Tijdens de berekeningen die hierna volgen zal  $X$  steeds in stappen van 0,1 worden verhoogd, totdat de in berekening zijnde functie van teken is veranderd. In dat geval gaat het programma terug naar de voorlaatste waarde van  $X$  en wordt de ophoging door 10 gedeeld. Dit is de zogenaamde 'bracketing'-methode.

Het programma berekent de  $X$ - en  $Y$ -waarden en het optimum, waarbij  $X$  wordt verhoogd totdat het optimum binnen de toegestane foutgrootte valt.

### Programma 31

```
10 PRINT"***OPTIMUM F(X,Y),MITS Y=G(X)***"
20 PRINT"-----"
25 PRINT"SUBROUTINE 250 IS H=G(X,Y)=0"
30 PRINT"SUBROUTINE 300 IS L=F(X,Y)OPTIMUM"
40 PRINT"SUBROUTINE 350 IS Y=G(X)EXPLICIET"
45 PRINT"F EN G ZIJN FUNCTIES!!":PRINT
50 INPUT"STARTWAARDE X";X
60 PRINT:V=10
70 T=.1:INPUT"FOUT CA '.0001' ";E
80 PRINT"X-WAARDE      Y-WAARDE      OPTIMUM"
90 X0=X:X=X+T
100 GOSUB350:B=Y
110 GOSUB300:G=L
120 R=R+1:IFR>30THEN END
130 X=X+E
140 GOSUB300:C=(L-G)/E
150 GOSUB250:F=H/E
160 X=X-E:Y=Y+E
170 GOSUB300:D=(L-G)/E
180 GOSUB250:P=H/E
190 Y=Y-E:U=C/F-D/P
195 IF ABS(U)>ABS(V)THEN X=X0-(U*T)/(U-V):GOTO90
200 V=U
210 IF U=0 THEN PRINTX;TAB(13);Y;TAB(26);L:GOTO245
220 PRINT X;TAB(13);Y;TAB(26);L
230 IFABS(U)<E/2THEN 245
240 GOTO 90
245 PRINT:PRINT"FOUT PARTIELE AFGELEIDEN= ";INT(U*1E9+.5)/1E9:END
250 H=X*X+Y*Y-6*(X+Y)+15.75:RETURN
300 L=(X*X/50+Y*Y/32):RETURN
350 Y=3-SQR(2.25-(X-3)*(X-3)):RETURN
360 REM STARTWAARDE X CA. 1.6
```

## Voorbeeld 31.1

\*\*\*OPTIMUM F(X,Y),MITS Y=G(X)\*\*\*

-----  
 SUBROUTINE 250 IS H=G(X,Y)=0  
 SUBROUTINE 300 IS L=F(X,Y)OPTIMUM  
 SUBROUTINE 350 IS Y=G(X)EXPLICIET  
 F EN G ZIJN FUNCTIES!!

STARTWAARDE X 1.6

FOOT CA '.0001' 1E-04

X-WAARDE	Y-WAARDE	OPTIMUM
1.7	2.25166852	.216251921
1.8	2.1	.202625625
1.9	1.9801961	.194749145
2	1.88196601	.19069289
2.1	1.8	.18946125
2.101648	1.79876576	.189460925

FOOT PARTIELE AFGELEIDEN= 4.92E-06

## Voorbeeld 31.2

\*\*\*OPTIMUM F(X,Y),MITS Y=G(X)\*\*\*

-----  
 SUBROUTINE 250 IS H=G(X,Y)=0  
 SUBROUTINE 300 IS L=F(X,Y)OPTIMUM  
 SUBROUTINE 350 IS Y=G(X)EXPLICIET  
 F EN G ZIJN FUNCTIES!!

STARTWAARDE X 1.62

FOOT CA '.0001' 1E-04

X-WAARDE	Y-WAARDE	OPTIMUM
1.72	2.21795141	.212910251
1.82	2.07393305	.200673409
1.92	1.95903891	.193672539
2.02	1.86439443	.190243609
2.12	1.78525723	.189497389
2.10230978	1.79827113	.189460964
2.10170367	1.79872414	.189460925

FOOT PARTIELE AFGELEIDEN= -1.708E-06

## 32. $F_{(x,y)} = 0$ en $G_{(x,y)} = 0$

Voordat het programma wordt gestart moeten op de regels 250 en 300 de functies  $F$  respectievelijk  $G$  worden ingevuld.

Na het starten van het programma moet u de startwaarden van  $X$  en  $Y$  opgeven. Zoals het voorbeeld laat zien is het erg belangrijk hiervoor goede startwaarden te kiezen.

Als laatste vraagt het programma nog naar de toegestane grootte van de fout.

Het programma berekent vervolgens de waarden van  $X$  en  $Y$ , waarvoor  $F_{(x,y)} = 0$  en  $G_{(x,y)} = 0$ .

### Programma 32

```
10 PRINT"****F(X,Y)=0 EN G(X,Y)=0****"
20 PRINT"-----"
25 PRINT
30 PRINT"F IN LIJN 200 EN G IN LIJN 300"
40 PRINT
50 INPUT"X0,Y0 START";X,Y
60 INPUT"FOUT";E
70 PRINT:PRINT"X"
80 GOSUB200:GOSUB300:K=F:H=G:EE=X+Y
90 X=X+E:GOSUB200:C=(F-K)/E
100 GOSUB300:L=(G-H)/E
110 X=X-E:Y=Y+E:GOSUB200
120 D=(F-K)/E:GOSUB300
130 B=(G-H)/E:Y=Y-E
140 A=C*B-L*D
150 IFA=0THENPRINT"FOUT 0-MATRIX":END
160 X=X+(H*D-K*B)/A
170 Y=Y+(L*K-C*H)/A
180 IF ABS(EE-X-Y)<ETHEN 400
190 PRINTX;TAB(19);Y
195 GOTO 80
200 F=SIN(X+Y)-X
210 RETURN
300 G=COS(X-Y)-Y
310 RETURN
400 PRINTX;TAB(19);Y
410 PRINT:PRINT"F(X,Y) IS ";F
420 PRINT"G(X,Y) IS ";G
430 END
```



## Voorbeeld 32

\*\*\*\*F(X,Y)=0 EN G(X,Y)=0\*\*\*\*

-----  
F IN LIJN 200 EN G IN LIJN 300

X0,Y0 START 0 0  
FOOT 1E-05

X	Y	
16460398.6		.999992385

FOOT 0-MATRIX

\*\*\*\*F(X,Y)=0 EN G(X,Y)=0\*\*\*\*

-----  
F IN LIJN 200 EN G IN LIJN 300

X0,Y0 START .5 .5  
FOOT 1E-05

X	Y	
1.83047907		.999996193
2.29493596		-1.55366082
47.3723756		16.6783035
-48.8944465		-31.1666083
8.28741227		-232.061846
240.017143		-781.986818
-4.17863923		-464.15427
522.527322		-1522.05879
2656.52502		-468.497055
586.671228		6678.20675
-377.633944		-325.449065
-2919.65425		-1390.15834
-121147.679		-34774.5869

\*\*\*\*F(X,Y)=0 EN G(X,Y)=0\*\*\*\*

-----  
F IN LIJN 200 EN G IN LIJN 300

X0,Y0 START 1 1  
FOOT 1E-05

X	Y	
.935953212		1
.9350847		.998020853
.935082064		.998020058

F(X,Y) IS -7.39633106E-06  
G(X,Y) IS -1.13081187E-05

X= .935082064 Y= .998020058  
-----

### 33. Polynomen met reële coëfficiënten (wortels)

De coëfficiënten  $A_n$  tot en met  $A_0$  moeten op regel 5 van het programma worden gezet, voor het programma wordt gestart.

Na het starten van het programma wordt u gevraagd naar de startwaarden van  $X$  en  $Y$  en de orde ( $N$ ) van de polynoom. De maximale orde van de polynoom is 20.

Vervolgens zal het programma de nulpunten berekenen van de polynoom:

$F(z) = 0 = A_0 Z^n + A_1 Z^{n-1} + \dots + A_n$ . Hierin is  $Z$  een complex getal ( $Z = X + iY$ ).

De toegepaste methode is die van Lin-Bairstow.

#### Programma 33

```
1 PRINT "IN LIJN 5 STAAN DE COEFFICIENTEN!"
2 PRINT " A4,A3,A2,A1,A0 !!"
3 OPEN 4,4:CMD4
4 DATA 3,-24,59,-288,576
5 PRINT "***NULPUNTEN VAN POLYNOMEN***"
10 PRINT "F(Z)=0=A0Z^N+A1Z^(N-1)+...AN"
20 PRINT "-----"
25 PRINT "DATARIJ IN LIJN 5 T/M 9. A0-AN!"
27 PRINT "STARTWAARDE Z= X + iY . N=ORDE "
30 DIMA(20),AR(20),BR(20),CR(20),DR(20),E(20)
40 PRINT:INPUT "X,Y,N ";X,Y,N
45 PRINT
50 AR(0)=1:CR(0)=1:BR(0)=0:DR(0)=0
55 READ A(0):U=A(0):A(0)=1
60 FOR I=1 TO N
70 READ A(I):A(I)=A(I)/U:NEXT I
80 FOR I=1 TO N
90 AR(I)=A(I)+X*AR(I-1)-Y*BR(I-1)
100 BR(I)=X*BR(I-1)+Y*AR(I-1)
110 CR(I)=AR(I)+X*CR(I-1)-Y*DR(I-1)
120 DR(I)=BR(I)+X*DR(I-1)+Y*CR(I-1)
130 NEXT I
140 X=X-(AR(N)*CR(N-1)+BR(N)*DR(N-1))/(CR(N-1)^2+DR(N-1)^2)
150 Y=Y-(BR(N)*CR(N-1)-AR(N)*DR(N-1))/(CR(N-1)^2+DR(N-1)^2)
160 IF ABS(X-A)<=.00000001 THEN 175
170 A=X:GOTO 80
175 IF ABS(Y)<1E-9 THEN Y=0
177 PRINT "ORDE      REELE W.      IMAGINAIRE W."
180 PRINTN;TAB(8)X;TAB(23)Y
185 IF N=1 OR N=0 THEN END
190 IF Y=0 THEN 300
200 Q=X*X+Y*Y:P=(2*X)*-1
210 E(1)=1:E(0)=0:A(0)=1
220 FOR I=2 TO N+1
230 E(I)=A(I-1)-P*E(I-1)-Q*E(I-2)
240 NEXT I
250 FOR I=2 TO N+1
260 A(I-1)=E(I)
```

```

270 NEXT I: N=N-2
280 IF N=1 OR N=0 THEN X=-E(2): Y=0: GOTO 180
290 GOTO 80
300 E(0)=1
310 FOR I=1 TO N
320 E(I)=E(I-1)*X+A(I)
330 NEXT I
340 FOR I=1 TO N-1: A(I)=E(I): NEXT I
350 N=N-1: X=.5: Y=1: GOTO 80

```

### Voorbeeld 33

\*\*\*NULPUNTEN VAN POLYNOMEN\*\*\*  
 $F(Z)=0=A_0Z^N+A_1Z^{N-1}+\dots+A_N$

-----  
 DATARIJ IN LIJN 5 T/M 9. A0-AN!  
 STARTWAARDE Z= X + IY . N=ORDE

X,Y,N 0 0 4

ORDE	REELE W.	IMAGINAIRE W.
4	2.36998627	0
ORDE	REELE W.	IMAGINAIRE W.
3	-.461002508	3.48598369
1	6.55201875	0



## 34. Polynomen met complexe coëfficiënten (wortels)

Na het starten vraagt het programma u eerst om de orde ( $N$ ) van de polynoom te specificeren. Daarna wordt u gevraagd de complexe coëfficiënten  $A_0$  en  $B_0$  tot en met  $A_n$  en  $B_n$  op te geven. De maximale orde is 30.

De polynoom heeft het formaat:

$$F(z) = C_0 z^n + C_1 z^{n-1} + \dots + C_n$$

Hierin is  $C_0 = A_0 + iB_0$  .....  $C_n = A_n + iB_n$

Het programma berekent de reële en imaginaire wortels voor de verschillende orden.

Naast de complexe rekenwijze wordt in dit programma ook de methode 'steepest decent' toegepast.

### Programma 34

```
10 DIMA(30),B(30)
15 PRINT"***WORTELS POLYNOOM, MAX.ORDE 30***"
16 PRINT"COMPLEXE COEFFICIENTEN A0&B0 T/M AN&BN"
17 PRINT"A0&B0 IS DE CONSTATE TERM!!"
18 PRINT
20 INPUT"ORDE POLYNOOM";N
25 FOR I=0 TO N
30 PRINT"A";I;:INPUTA(I)
40 PRINT"B";I;:INPUTB(I)
45 IFA(1)=0THENA(1)=.0000001
50 NEXT I
55 PRINT:PRINT"ORDE "; " REELE W. "; " IMAGINAIRE W."
60 X=0
70 Y=1
80 ER=1E19
90 Y0=0:X0=1:X1=X:Y1=Y:H=X*X+Y*Y
100 I=0
110 U=A(I)*X0:V=B(I)*X0
120 IF I=N GOTO280
130 I=I+1
140 U=U+A(I)*X1-B(I)*Y1
150 V=V+A(I)*Y1+B(I)*X1
160 DV=B(I):DU=A(I)
170 IF I=N GOTO 280
180 I=I+1
190 X2=2*X*X1-H*X0
200 Y2=2*X*Y1-H*Y0
210 U=U+A(I)*X2-B(I)*Y2
220 V=V+A(I)*Y2+B(I)*X2
230 DU=DU+(A(I)*X1-B(I)*Y1)*I
240 DV=DV+(A(I)*Y1+B(I)*X1)*I
250 A=Y1:Y1=Y2:Y0=A
260 A=X1:X1=X2:X0=A
270 GOTO 170
```

```

280 A=1E-09
290 B=DX:C=DY:G=U*U+V*V
291 IF G<1E-10 THEN 400
295 IF (G-ER)>=A THEN 380
300 H=DU*DU+DV*DV
310 DX=(U*DU+V*DV)/H
320 X=X-DX
330 IF ABS(DX)<A THEN 400
340 DY=(U*DV-V*DU)/H
350 Y=Y+DY
360 IF ABS(DY)<A THEN 400
370 GOTO 390
380 X=X+.25*B:Y=Y-.25*C
390 ER=G:GOTO 90
400 IF ABS(Y)<.0000001 THEN Y=0
405 PRINT "N=";N;:PRINT " ";:PRINT X;:PRINT TAB(18);Y:PRINT
410 IF ABS(Y)>=1E-2 THEN 520
420 A=0:D=1+N:E=N
430 A(E)=A(D)*X+A(E)
440 IF E=A THEN 460
450 D=D-1:E=E-1:GOTO 430
460 N=N-1
470 FOR I=0 TO N+1
480 A(I)=A(I+1)
490 NEXT I
500 IF N=A THEN END
510 GOTO 60
520 Y2=X*X+Y*Y:Y1=2*X
530 E=N:D=0:DV=0:A=0
540 A(E)=A(E)+Y1*D-Y2*DV
550 Q=D:D=A(E):DV=Q
560 IF (E-2)=A THEN 580
570 E=E-1:GOTO 540
580 FOR I=0 TO N+1
590 A(I)=A(I+2)
600 NEXT I
610 E=N-1:D=0:DV=0
620 B(E)=B(E)+Y1*D-Y2*DV
630 Q=D:D=B(E):DV=Q
640 IF (E-1)=A THEN 660
650 E=E-1:GOTO 620
660 FOR I=0 TO N+1
670 B(I)=B(I+2)
680 NEXT I
690 N=N-2:GOTO 500

```

## Voorbeeld 34.1

\*\*\*\*WORTELS POLYNOM, MAX.ORDE 30\*\*\*\*  
COMPLEXE COEFFICIENTEN A0&B0 T/M AN&BN  
A0&B0 IS DE CONSTATE TERM!!

ORDE POLYNOM 4

A 0 576  
B 0 0  
A 1 -288  
B 1 0  
A 2 59  
B 2 0  
A 3 -24  
B 3 0  
A 4 3  
B 4 0

ORDE	REELE W.	IMAGINAIRE W.
N= 4	2.36998626	0
N= 3	-.461002508	3.48598369
N= 1	6.55201875	0

## Voorbeeld 34.2

\*\*\*\*WORTELS POLYNOM, MAX.ORDER 30\*\*\*\*  
COMPLEXE COEFFICIENTEN A0&B0 T/M AN&BN  
A0&B0 IS DE CONSTATE TERM!!

ORDER POLYNOM 2

A 0 6  
B 0 -3  
A 1 4  
B 1 4  
A 2 1  
B 2 -2

ORDER	REELE W.	IMAGINAIRE W.
N= 2	-.278233	.834698992



## 35. Minimum, maximum, buigpunten

$Y = F(x)$  moet op regel 200 worden gespecificeerd, voordat het programma wordt gestart. In het bijgaande programma is gekozen voor:  $Y = 8X^3 - 20X^2 - 2X + 5$ .

Nadat het programma is gestart, komt eerst de vraag naar maximum of minimum, waarop u kunt ingeven 1 (max.) of -1 (min.). Vervolgens moet u de startwaarde van  $X$  en de stapgrootte ingeven. Als laatste vraagt het programma naar het aantal cijfers. Het maximum aantal is 8. Dit aantal cijfers geeft aan hoeveel maal de stapgrootte door 10 wordt gedeeld.

Nu berekent het programma de  $X$ -waarde waarvoor  $Y$  een maximum dan wel een minimum heeft. De output van het programma is een tabel van  $X$ -waarden en bijbehorende  $Y$ -waarden, die start nadat de stapgrootte voor de eerste maal door 10 is gedeeld.

Hierna kunt u nog een tabel laten maken die start met  $X_0$  en waarin de  $X$ -waarden een onderlinge afstand hebben die gelijk is aan de opgegeven stapgrootte. Deze tabel komt in blokken van 20 stappen. Na ieder blok hebt u de keuze om te stoppen (door op de RUN- of STOP-toets te drukken) of om door te gaan met een volgend blok (door een letter in te tikken).

### Programma 35

```
10 PRINT"***MAX.,MIN.OF BUIGPUNT Y=F(X)***"
20 PRINT"-----"
30 PRINT:REM BRACKETING METHODE
35 REM Y=F(X):RETURN LIJN 200 OP
40 INPUT"MAX.1 MIN.-1";I
50 INPUT"START X0";X0
60 INPUT"STARTSTAP";D
70 INPUT"AANTAL CIJFERS";N
75 PRINT"X-WAARDE          Y-WAARDE"
80 X=X0:K=0:IFN>8THEN N=8
90 GOSUB 200
100 T=Y:X=X+D
110 GOSUB 200
120 IF I=-1THEN150
130 IFY>TTHEN 100
140 GOTO 160
150 IFY<=T THEN100
160 X=X-2*D:D=D/10:K=K+1
170 PRINTX;TAB(16);T
180 IFK>N THEN PRINT:GOTO 300
190 GOTO 90
200 Y=8*X^3-20*X^2-2*X+5
210 RETURN
300 PRINT"MAAK TABEL MET START X0,STEP .1"
310 INPUT"VOOR CONTROLE;JA OF NEE";A$
320 IF A$<>"JA" THEN END
330 X=X0-.1: D=.1
```

```

335 PRINT "X-WAARDE"           Y-WAARDE"
340 FOR I=1 TO 20
350 X=X+D:GOSUB 200
360 PRINTX;TAB(16);Y
370 NEXT:PRINT:PRINT"TYP EEN LETTER OF"
375 PRINT"EINDIG MET RUN/STOP TOETS"
380 GETA$:IFA$=""THEN380
390 GOTO 335

```

## Voorbeeld 35

\*\*\*MAX., MIN. OF BUIGPUNT  $Y=F(X)$ \*\*\*

---

```

MAX.1  MIN.-1-1
START X0 0
STARTSTAP .1
AANTAL CIJFERS 6

```

X-WAARDE	Y-WAARDE
1.6	-16.896
1.71	-16.900416
1.714	-16.900893
1.7152	-16.9008943
1.71523	-16.9008943
1.715232	-16.9008943
1.7152319	-16.9008943

```

MAAK TABEL MET START X0,STEP .1
VOOR CONTROLE;JA OF NEEJA

```

X-WAARDE	Y-WAARDE
0	5
.1	4.608
.2	3.864
.3	2.816
.4	1.512
.5	0
.6	-1.67200001
.7	-3.45600001
.8	-5.30400001
.9	-7.16800001
1	-9.00000001
1.1	-10.752
1.2	-12.376
1.3	-13.824
1.4	-15.048
1.5	-16
1.6	-16.632
1.7	-16.896
1.8	-16.744
1.9	-16.128

TYP EEN LETTER OF  
EINDIG MET RUN/STOP TOETS

X-WAARDE	Y-WAARDE
2	-15
2.1	-13.312
2.2	-11.016
2.3	-8.06399999
2.4	-4.408
2.5	1.86264515E-09
2.6	5.20800001
2.7	11.264
2.8	18.216
2.9	26.1119999
3	34.9999999
3.1	44.9279998
3.2	55.9439999
3.3	68.0959997
3.4	81.4319997
3.5	95.9999996
3.6	111.847999
3.69999999	129.023999



## 36. Lengte van een curve, rotatie-oppervlak en -inhoud

Op regel 500 moet de functie van  $X$  worden gespecificeerd, voordat het programma wordt gestart. In bijgaand programma is gekozen voor  $Y = X$ .

Na het starten vraagt het programma om de startwaarde  $X_0$ , de eindwaarde  $X_{n-1}$  en het aantal stappen ( $N$ ).

Met deze gegevens maakt het programma de volgende berekeningen:

De lengte van de curve.

Het oppervlak waarover de curve roteert.

Het volume van de rotatiecurve.

Bij  $Y = X$  zal de curve een rechte lijn zijn, die bij roteren een kegel vormt, waarvan achtereenvolgens de oppervlakte en de inhoud worden berekend.

### Programma 36

```
10 PRINT"**** LENGTE VAN EEN CURVE ***"
20 PRINT"*OPPERVLAK ROTATIE VAN CURVE*"
25 PRINT"***VOLUME VAN ROTATIECURVE***"
30 PRINT"-----"
40 PRINT
50 PRINT"Y IS FUNCTIE VAN X IN LIJN 500"
55 REM ROTATIE T.O.V. DE X-AS
60 PRINT
65 REM N IN GROOTTEORDE 100 TOT 1000
70 INPUT"X0 STARTWAARDE";A
80 INPUT"XN EINDWAARDE ";B
90 INPUT"AANTAL STAPPEN";N
100 C=(B-A)/N
110 S=0:T=0:V=0:P=3.14159265
120 X=A+C*N
130 GOSUB 500
140 Y0=Y
150 N=N-1
160 X=A+C*N
170 GOSUB 500
180 GOSUB 300
190 Y0=Y
200 N=N-1
210 X=A+C*N
220 GOSUB 500
230 GOSUB 300
240 IF N<=0 THEN PRINT:GOTO 350
250 GOTO 190
300 U=SQR((Y0-Y)^2+C*C)
310 S=S+U:Y1=(Y0+Y)/2
320 T=T+2*P*Y1*U
330 V=V+P*Y1*Y1*C
340 RETURN
350 PRINT"LENGTE CURVE IS      ";S
360 PRINT
```

```

370 PRINT"OPPERVLAK ROTATIE IS ";T
380 PRINT
390 PRINT"VOLUME VAN ROTATIE IS";V
400 END
490 REM LIJN 500 OP VOOR Y=F(X)
500 Y=X
510 RETURN

```

### Voorbeeld 36.1

```

**** LENGTE VAN EEN CURVE ***
*OPPERVLAK ROTATIE VAN CURVE*
***VOLUME VAN ROTATIECURVE***
-----

```

Y IS FUNCTIE VAN X IN LIJN 500

DE FUNCTIE IS 500 Y=X  
=====

X0 STARTWAARDE 0  
XN EINDWAARDE 5  
AANTAL STAPPEN 100

LENGTE CURVE IS           7.07106779  
OPPERVLAK ROTATIE IS   111.072073  
VOLUME VAN ROTATIE IS 130.896421

### Voorbeeld 36.2

```

**** LENGTE VAN EEN CURVE ***
*OPPERVLAK ROTATIE VAN CURVE*
***VOLUME VAN ROTATIECURVE***
-----

```

Y IS FUNCTIE VAN X IN LIJN 500

DE FUNCTIE IS 500 Y=cos(X)  
=====

X0 STARTWAARDE 0  
XN EINDWAARDE 1.5708  
AANTAL STAPPEN 100

LENGTE CURVE IS           2.22144667  
OPPERVLAK ROTATIE IS   10.9624255  
VOLUME VAN ROTATIE IS 4.05863914

## 37. Systemen van differentiaalvergelijkingen

Voordat u het programma start kunt u op regelnummers 1000 en hoger de gewenste functies invullen.

Is het programma eenmaal gestart, dan zal het u vragen naar:

het aantal vergelijkingen,

de starttijd en eindtijd,

het aantal waarden (dit is tevens een indicatie voor de stapgrootte),

de stapvoorwaarde  $dX/dt$  voor alle functies en

de toe te passen rekenmethode.

U kunt kiezen uit de volgende drie rekenmethoden:

1. Euler.

2. Euler, echter nu met een vier keer zo hoge nauwkeurigheid.

3. Runge Kutta.

Het programma betekent voor de verschillende functies de bij iedere  $T$  behorende  $X$ -waarde.

De stapgrootte voor  $T = \frac{\text{eindtijd} - \text{starttijd}}{\text{aantal waarden}}$ .

### Programma 37

```
5 PRINT"SYSTEMEN VAN DIFFERENTIAALVERGELIJKINGEN"
7 PRINT"=====":PRINT
10 INPUT"AANTAL VERGELIJKINGEN:";N
20 INPUT"STARTTIJD: ";A
30 INPUT"EINDTIJD : ";E
40 INPUT"AANTAL PRINTS : ";H
45 H=(E-A)/H
50 PRINT"STARTVOORWAARDEN DX/DT : "
60 FORI=1TON
70 PRINT"X(";I;:INPUT")=";X(I)
80 NEXTI
90 INPUT"EULER 1,EULER 2,RUNGE KUTTA 4;TYP #:";A$
100 T=A
110 IF T>(E+H)THEN END
120 IF T<0 THEN END
130 PRINT:PRINT"T",
140 FORI=1TON
150 PRINT"X(";I;:PRINT")",
160 NEXTI:PRINT
170 PRINTT,
180 FORI=1TON
190 PRINTX(I),
200 NEXTI
220 IFA$="1"THENGOSUB 300
230 IFA$="2"THENGOSUB 400
240 IFA$="4"THENGOSUB 600
250 T=T+H: GOTO 110
```



```

300 GOSUB 1000
310 FOR I=1 TO N
320 XH(I)=X(I)
330 X(I)=X(I)+H*F(I)
340 K1(I)=F(I)
350 NEXT I
360 RETURN
400 GOSUB 300
410 T=T+H
420 GOSUB 1000
430 T=T-H
440 FOR I=1 TO N
450 X(I)=XH(I)+H*(F(I)+K1(I))/2
460 NEXT I
470 RETURN
600 GOSUB 300
610 FOR I=1 TO N
620 K1(I)=H*F(I)
630 X(I)=XH(I)+K1(I)/2
640 NEXT I
650 TH=T:T=T+H/2
660 GOSUB 1000
670 FOR I=1 TO N
680 K2(I)=H*F(I)
700 X(I)=XH(I)+K2(I)/2
710 NEXT I
720 GOSUB 1000
730 FOR I=1 TO N
740 K3(I)=H*F(I)
750 X(I)=XH(I)+K3(I)
760 NEXT I
770 T=TH+H
780 GOSUB 1000
790 T=TH
800 FOR I=1 TO N
810 K4(I)=H*F(I)
820 X(I)=XH(I)+(K1(I)+2*K2(I)+2*K3(I)+K4(I))/6
830 NEXT I
840 RETURN
985 REM =====
990 REM HIER KOMEN DE FUNCTIES(X,T)
995 REM START MET LIJN 1000
997 REM EINDIG MET RETURN
998 REM =====
1000 F(1)=X(2)
1010 F(2)=-X(1):RETURN
1020 ETC.
1500 RETURN

```

## Voorbeeld 37

### SYSTEMEN VAN DIFFERENTIAALVERGELIJKINGEN

AANTAL VERGELIJKINGEN: 2

STARTTIJD: 0

EINDTIJD: 1.5

AANTAL PRINTS: 20

STARTVOORWAARDEN DX/DT:

X( 1 ) = 0

X( 2 ) = 1

EULER 1, EULER 2, RUNGE KUTTA 4; TYP #: 2

T	X( 1 )	X( 2 )
0	0	1
T	X( 1 )	X( 2 )
.075	.075	.9971875
T	X( 1 )	X( 2 )
.15	.149578125	.98875791
T	X( 1 )	X( 2 )
.225	.22331428	.974758669
T	X( 1 )	X( 2 )
.3	.295793109	.955268589
T	X( 1 )	X( 2 )
.375	.366606335	.930397413
T	X( 1 )	X( 2 )
.45	.43535506	.900285195
T	X( 1 )	X( 2 )
.525	.501652014	.865101514
T	X( 1 )	X( 2 )
.6	.565123731	.825044515
T	X( 1 )	X( 2 )
.675	.625412659	.780339797
T	X( 1 )	X( 2 )
.75	.682179171	.731239142
T	X( 1 )	X( 2 )
.825	.735103478	.678019094
T	X( 1 )	X( 2 )
.9	.783887431	.620979405
T	X( 1 )	X( 2 )
.975	.828256204	.560441343
T	X( 1 )	X( 2 )
1.05	.867959834	.496745886
T	X( 1 )	X( 2 )
1.125	.902774638	.430251801
T	X( 1 )	X( 2 )
1.2	.93250447	.36133362
T	X( 1 )	X( 2 )
1.275	.956981822	.290379534
T	X( 1 )	X( 2 )
1.35	.976068776	.217789205
T	X( 1 )	X( 2 )
1.425	.989657773	.143971514
T	X( 1 )	X( 2 )
1.5	.997672224	.0693422615

## 38. Romberg / Gauss-integraal

Voor het programma wordt gestart moet de gewenste functie  $F_{(x)}$  op regel 500 worden ingevoerd.

Is het programma eenmaal gestart, dan zal het achtereenvolgens aan u vragen de ondergrens, de bovengrens, de selectie van Romberg of Gauss (1 of 2) en de maximale fout te specificeren.

Indien u Gauss selecteert, zal het programma u bovendien nog vragen een even getal ( $N$ ), dat ligt tussen 4 en 16, op te geven, waarmee de stapgrootte kan worden bepaald.

Met deze gegevens berekent het programma de integraal van de functie  $F_{(x)}$  tussen de onder- en bovengrens door middel van de methode Romberg of Gauss.

### Programma 38

```
10 PRINT"**ROMBERG/GAUSS INTEGRAAL**"
15 PRINT"=====
20 PRINT"LIJN 500 FX=.....:RETURN"
30 PRINT:PRINT"ONDERGRENS";:INPUTA
40 PRINT"BOVENGRENS";:INPUTB
50 PRINT"VOOR ROMBERG TYP 1,VOOR GAUSS 2";:INPUTR
60 PRINT"MAXIMALE FOUT";:INPUTER
70 L=ABS(INT(LOG(ER)/LOG(10)+.5)):L=10^L
80 IF R=2 THENQQ=1E15:GOTO240
90 M=1E15:C=0
100 G=(ABS(B)-ABS(A))/2
110 X=A:GOSUB500:C=C+FX
120 X=B:GOSUB500:C=C+FX
130 Q=(B-A)*M:G=(ABS(B)-ABS(A))/2
140 T=0:X=A+G:GOSUB500:S=FX
150 QQ=G*(C+(4*S)+(2*T))/3:PRINTQQ
160 IFABS(Q-QQ)<ERTHEN210
170 Q=QQ:T=T+S:G=G/2:S=0
180 FORU=ATOB-GSTEP2*G
190 X=U+G:GOSUB500:S=S+FX:NEXT
200 GOTO150
210 F=QQ+(Q-QQ)/15:PRINTF:R=INT(F*L+.5)/L:A$="ROMBERG"
220 PRINT"INTEGRAAL=";R,A$
230 PRINT:GOTO50
240 PRINT"TYP N (EVEN GETAL 4-16)";:INPUTN
250 H=(B-A)/N:QQ=Q
260 AA=A+H*(1-1/SQR(3))
270 BB=A+H*(1+1/SQR(3)):M=0:S=0
280 X=AA:GOSUB500:S=S+FX
290 X=BB:GOSUB500:S=S+FX
300 IFM=N/2-1THENPRINTN,S*H:GOTO340
310 AA=AA+2*H
320 BB=BB+2*H
330 M=M+1:GOTO280
```



```

340 Q=S*H
350 IFABS(Q-QQ)<ERTHEN R=INT(Q*L+.5)/L:A$="GAUSS":GOTO220
360 N=N*2:GOTO250
500 FX=SQR(1+SQR(X))
600 RETURN

```

## Voorbeeld 38

\*\*ROMBERG/GAUSS INTEGRAAL\*\*

=====

LIJN 500 FX=.....:RETURN

ONDERGRENS 0

BOVENGRENS 4

VOOR ROMBERG TYP 1,VOOR GAUSS 2 1

MAXIMALE FOUT 1E-03

5.96476447

6.03600654

6.06167829

6.07084695

6.07410658

6.0752625

6.07567183

6.07564454

INTEGRAAL= 6.076

ROMBERG

VOOR ROMBERG TYP 1,VOOR GAUSS 2 2

MAXIMALE FOUT 1E-03

TYP N (EVEN GETAL 4-16) 4

6.08585207

8 6.07947329

16 6.07717197

32 6.07634925

INTEGRAAL= 6.076

GAUSS

VOOR ROMBERG TYP 1,VOOR GAUSS 2 0

## 39. Discrete Fourier-analyse

Na te zijn gestart vraagt het programma u op te geven hoeveel punten ( $N$ ) u wilt gaan invoeren. Vervolgens wordt u gevraagd de waarde van de punten ( $X_0$  t/m  $X_{n-1}$ ) in te geven.

Door middel van de recursieve methode worden nu de harmonischen van de opgegeven punten berekend.

De berekening is hier in verband met de recursieve methode ontwikkeld voor een oneven aantal op onderling gelijke afstand liggende punten.

Het aantal te berekenen harmonischen is  $\frac{N}{2}$ .

### Programma 39

```
5 PRINT"DISCRETE FOURIER ANALYSE"
6 PRINT"=====":PRINT
10 PRINT"HARMONISCHE ANALYSE.EQUID.PUNTEN"
20 PRINT"RECURSIEVE METHODE":PRINT
30 PRINT"HOEVEEL PUNTEN ":INPUTN
40 DIMF(N)
45 PRINT
50 FORI=0TON-1
60 PRINT"X(";I;")":INPUTF(I)
70 NEXTI
120 C=1:S=0:P=0
160 C1=COS((2*PI)/N)
170 S1=SIN((2*PI)/N)
180 U2=0:U1=0:N1=N
190 FOR I=N1 TO 1 STEP-1
200 U0=F(I)+2*C*U1-U2
210 U2=U1:U1=U0
220 NEXT I
230 A=2*(F(0)+C*U1-U2)/N
240 B=2*S*U1/N
250 PRINT"HARM. :";P;TAB(11);INT(SQR(A*A+B*B)*10000+.5)/10000;
260 PRINTTAB(25);"[ABS.]"
280 IF P>N/2-1THEN END
290 Q=C1*C-S1*S
300 S=C1*S+S1*C
310 C=Q
320 P=P+1
330 GOTO 180
```

## Voorbeeld 39

HOEVEEL PUNTEN  
DISCRETE FOURIER ANALYSE  
=====

HARMONISCHE ANALYSE.EQUID.PUNTEN  
RECURSIEVE METHODE

HOEVEEL PUNTEN 11

X( 0 ) 2  
X( 1 ) 4  
X( 2 ) 5  
X( 3 ) 3  
X( 4 ) 6  
X( 5 ) 7  
X( 6 ) 1  
X( 7 ) 2  
X( 8 ) 4  
X( 9 ) 3  
X( 10 ) 1

HARM. : 0	6.9091	[ABS.]
HARM. : 1	1.527	[ABS.]
HARM. : 2	.6407	[ABS.]
HARM. : 3	1.856	[ABS.]
HARM. : 4	.8485	[ABS.]
HARM. : 5	.3665	[ABS.]

READY.



## 40. Fast Fourier-transformatie (FFT)

Het programma vraagt eerst om het aantal punten ( $N$ ).  $N$  moet een macht van 2 zijn. Daarna kunt u kiezen tussen transformatie ( $-1$ ) en inverse ( $+1$ ).

Met de data die op de regels 200 tot en met 259 staat voert het programma vervolgens de gekozen berekening uit. De data in bijgaand programma is  $X_{(i)} = i$  en  $Y_{(i)} = 17$ , waarin  $i$  de waarden heeft van 0 tot en met  $N-1$ .

*Inverse:* De absolute waarde van een imaginair getal waarvan  $X_{(i)}$  het reële en  $Y_{(i)}$  het imaginaire deel is wordt berekend.

*Transformatie:* Het getal wordt getransformeerd en het reële, absolute en imaginair deel worden berekend.

Door het intypen van een letter aan het einde van een berekening kan het programma opnieuw worden gestart.

### Programma 40

```
1 REM *****FAST FOURIER TRANSFORMATIE*****
2 REM *****MET INVERSE TRANSFORMATIE*****
4 I=0:J=0:B=0:C=0:U=0:V=0:R=0:M=0:T=0:P=0:N=0:D=0:F=0:A=0:Q=0:E=0
5 PRINT"FOURIER F.F.T.; -1=TRANSFORM;1=INVERSE":PRINT
6 INPUT"N*2 ";N:DIMX(N),Y(N):Q=1
7 FORI=1TO15
8 Q=Q*2
9 IF Q=NTHEN 11
10 NEXT:PRINT"N IS GEEN MACHT VAN 2":END
11 INPUT" +1 OR -1 ";D
12 PRINT"*BEN BEZIG MET DATAINVOER;200-259 *"
13 GOTO200
14 PRINT"*FFT BEREKENING IS GESTART!! *":A=N
20 P=PI:F=N:J=1
21 FORI=1TON
22 IFI>=JTHEN30
23 U=X(J):V=Y(J)
24 X(J)=X(I):Y(J)=Y(I)
25 X(I)=U:Y(I)=V
30 M=N/2
50 IFM>=JTHEN70
51 J=J-M
52 M=INT((M+1)/2)
53 GOTO 50
70 J=J+M
80 NEXT
85 R=1
90 IFR>=NTHEN130
100 E=2*R
101 FORM=1TOR
102 T=P*(D*(M-1))/R
103 B=COS(T):C=SIN(T)
104 FORI=MTON STEP E
```

```

105 J=I+R
106 U=B*X(J)+C*Y(J)
107 V=B*Y(J)-C*X(J)
108 X(J)=X(I)-U:Y(J)=Y(I)-V
109 X(I)=X(I)+U:Y(I)=Y(I)+V
110 NEXT
120 NEXT
125 R=E:GOTO90
130 IFD<0THEN161
140 FOR I=1TON
150 X(I)=X(I)/F
155 Y(I)=Y(I)/F
156 NEXT
161 PRINT"NR.      REEEL      IMAGINAIR      [ABS.]":IFT>0THENA=2*N
162 FORI=1TOA/2
165 PRINTINT(I-1);
170 PRINTTAB(6)INT(X(I)*1000+.5)/1000;TAB(16);INT(Y(I)*1000+.5)
    /1000;
171 PRINTTAB(29);INT(SQR(X(I)*X(I)+Y(I)*Y(I))*1000+.5)/1000
175 FORJ=1TO300:NEXT
176 NEXT
177 PRINT:PRINT"VOOR MEER TYP EEN LETTER"
178 GETA$:IFA$=""THEN178
180 INPUT"N ";N:INPUT"+1 OR -1 ";D|
    :IF D<0 THEN PRINT"FFT BEZIG!!":GOTO20
190 PRINT"INVERSE FFT IS GESTART!":GOTO20
200 FOR I=1TO N
210 X(I)=I:Y(I)=17
220 NEXT
260 GOTO 14

```

## Voorbeeld 40

\*\*\*\*FAST FOURIER TRANSFORMATIE\*\*\*\*  
 \*\*\*\*MET INVERSE TRANSFORMATIE\*\*\*\*  
 FOURIER F.F.T.: -1=TRANSFORM;1=INVERSE

N12 16

+1 OR -1 -1

\*BEN BEZIG MET DATAINVOER;200-259 \*

\*FFT BEREKENING IS GESTART!! \*

NR.	REEEL	IMAGINAIR	[ABS.]
0	136	272	304.105
1	-8	-40.219	41.007
2	-8	-19.314	20.905
3	-8	-11.973	14.4
4	-8	-8	11.314
5	-8	-5.345	9.622
6	-8	-3.314	8.659
7	-8	-1.591	8.157

VOOR MEER TYP EEN LETTER Q

N 16

+1 OR -1 1

INVERSE FFT IS GESTART!

NR.	REEEL	IMAGINAIR	[ABS.]
0	1	17	17.029
1	2	17	17.117
2	3	17	17.263
3	4	17	17.464
4	5	17	17.72
5	6	17	18.028
6	7	17	18.385
7	8	17	18.788
8	9	17	19.235
9	10	17	19.723
10	11	17	20.248
11	12	17	20.809
12	13	17	21.401
13	14	17	22.023
14	15	17	22.672
15	16	17	23.345

VOOR MEER TYP EEN LETTER WWWWWWW



## 41. Systeem van lineaire vergelijkingen

Het programma vraagt eerst naar het aantal vergelijkingen ( $N$ ) en daarna naar het aantal te berekenen onbekenden ( $P$ ), waarbij geldt dat  $P \leq N$ . Vervolgens vraagt het programma u de coëfficiënten in te voeren.

Hierna worden de onbekenden  $X_0$  tot en met  $X_n$  berekend.

Indien het aantal onbekenden kleiner is dan het aantal vergelijkingen, wordt daarbij de methode van de kleinste kwadraten gevolgd.

Uitbreiding is mogelijk tot vormen als:

$$A \log X_1 + B \log X_2 = 2 \log Y$$

mits de bijbehorende functies  $X_1$ ,  $X_2$  en  $Y$  worden geprogrammeerd.

### Programma 41

```
10 PRINT"SYSTEEM LINIAIRE VERGEKLIJINGEN"
20 PRINT"===== "
30 PRINT
40 PRINT"A11*X1+A12*X2+...A1P*XP=B1"
50 PRINT"A21*X1+A22*X2+...A2P*XP=B2"
60 PRINT"A.1*X1+A.2*X2+...A.P*XP=B."
70 PRINT"AN1*X1+AN2*X2+...ANP*XP=BN"
80 PRINT:PRINT"INDIEN MEER VERG. DAN GEZOCHTE"
90 PRINT"ONBEKENDEN, VOLGT KL.KWADRATEN BER."
100 PRINT:PRINT"AANTAL VERGELIJKINGEN ";:INPUTN
110 PRINT"HOEVEEL ONBEKENDEN ";:INPUTP
120 IFN<PTHENPRINT"TE WEINIG VERG. !":END
130 DIMA(N,P),AA(P,N),B(N),BB(P),X(N)
140 PRINT
150 FORI=1TON:FORJ=1TOP
160 PRINT"Typ A( ";I;J;" )=";:INPUTA(I,J)
170 NEXTJ
180 PRINT"Typ B( ";I;" )=";:INPUTB(I)
190 PRINT:NEXTI
200 IFN>PTHENGOSUB500
210 FORI=1TON-1
220 H=A(I,I):X(I)=0:HH=I
230 FORK=I+1TON:IFABS(H)<ABS(A(K,I))THENHH=K:NEXTK
240 IFHH=ITHEN270
250 FORK=1TON:U=A(HH,K):A(HH,K)=A(I,K):A(I,K)=U:NEXTK
260 U=B(HH):B(HH)=B(I):B(I)=U
270 FORK=I+1TON
280 D=A(K,I)/A(I,I)
290 FORL=1TON:A(K,L)=A(K,L)-D*A(I,L):NEXTL
300 B(K)=B(K)-D*B(I)
310 NEXTK:NEXTI
320 FORI=NT01STEP-1
330 E=0
340 FORK=1TON:E=E+X(K)*A(I,K):NEXTK
350 IFABS(A(I,I))<1E-10THENPRINT"A" I;I;" TE KLEIN":END
```

```

360 X(I)=(B(I)-E)/A(I,I)
370 NEXTI
380 PRINT:PRINT"HET RESULTAAT IS:"
390 FORI=1TON
400 PRINT"X(";I;")= ";X(I)
410 NEXTI:END
500 FORI=1TOP:FORJ=1TOP:FORK=1TON
510 AA(I,J)=AA(I,J)+A(K,I)*A(K,J)
520 NEXTK:NEXTJ:NEXTI
530 FORI=1TOP:FORJ=1TON
540 BB(I)=BB(I)+A(J,I)*B(J)
550 NEXTJ:NEXTI
560 FORI=1TOP:FORJ=1TOP
570 A(I,J)=AA(I,J):B(I)=BB(I)
580 NEXTJ:NEXTI
590 N=P:X(N)=0:RETURN

```

## Voorbeeld 41

SYSTEEM LINIAIRE VERGEKLIJINGEN  
=====

$A_{11} \cdot X_1 + A_{12} \cdot X_2 + \dots + A_{1P} \cdot X_P = B_1$   
 $A_{21} \cdot X_1 + A_{22} \cdot X_2 + \dots + A_{2P} \cdot X_P = B_2$   
 $A_{.1} \cdot X_1 + A_{.2} \cdot X_2 + \dots + A_{.P} \cdot X_P = B_{.}$   
 $A_{N1} \cdot X_1 + A_{N2} \cdot X_2 + \dots + A_{NP} \cdot X_P = B_N$

INDIEN MEER VERG. DAN GEZOCHTE  
ONBEKENDEN, VOLGT KL.KWADRATEN BER.

AANTAL VERGELIJKINGEN    4  
HOEVEEL ONBEKENDEN       4

TYP A( 1   1 )= 1  
TYP A( 1   2 )= 2  
TYP A( 1   3 )=-12  
TYP A( 1   4 )= 8  
TYP B(   1 )= 27

TYP A( 2   1 )= 5  
TYP A( 2   2 )= 4  
TYP A( 2   3 )= 7  
TYP A( 2   4 )=-2  
TYP B(   2 )= 4

TYP A( 3   1 )=-3  
TYP A( 3   2 )= 7  
TYP A( 3   3 )= 9  
TYP A( 3   4 )= 5  
TYP B(   3 )= 11

TYP A( 4   1 )= 6  
TYP A( 4   2 )=-12  
TYP A( 4   3 )=-8  
TYP A( 4   4 )= 3  
TYP B(   4 )= 49

HET RESULTAAT IS:  
X( 1 )= 3  
X( 2 )= -2  
X( 3 )= 1  
X( 4 )= 5

## 42. Coëfficiënten polynomen

Nadat u het aantal punten ( $N$ ) hebt opgegeven, zal het programma u vragen de waarde van die punten ( $X, Y$ ) te specificeren.

De orde van de polynoom is het aantal punten min 1 ( $N - 1$ ). De maximum orde is 9, het maximum aantal punten is dus 10.

Het programma berekent de coëfficiënten waarvoor alle punten voldoen aan de vergelijking:

$$Y = A_0 + A_1X + A_2X^2 + \dots + A_nX^n$$

### Programma 42

```
10 PRINT"COEFFICIENTEN POLYNOMEN"
20 PRINT"=====
30 REM PUNTEN COLLOCEREN MET POLYNOM
40 REM DE ORDE V.H.POL.IS HET AANTAL
50 REM PUNTEN MINUS 1.MAX ORDER 9
60 PRINT"TYP HET AANTAL X,Y PUNTEN";:INPUTN
70 FOR I=1TON
80 PRINT"PUNT";I;" X,Y = ";:INPUTX(I),Y(I)
90 NEXTI
100 FORI=0TON:A(I)=0:B(I)=0:C(I)=1:NEXTI
110 A(0)=1
120 FORK=1TON:C(K)=0:C(K-1)=1:D=1:E=1
130 FORJ=1TON
140 IFC(J)=0THEN175
150 B=X(J)
160 FORI=DTO1STEP-1:A(I)=A(I-1)-B*A(I):NEXTI
170 A(0)=-B*A(0):D=D+1:E=E*(X(K)-X(J))
175 NEXTJ
180 FORJ=0TON-1:B(J)=B(J)+A(J)*Y(K)/E
190 A(J)=0
200 NEXTJ
210 A(0)=1:NEXTK
220 PRINT:PRINT"DE COEFFICIENTEN EN MACHTEN VAN X ZIJN:":PRINT
230 FOR I=N-1TO 0 STEP-1
235 B(I)=INT(B(I)*10000+.5)/10000
240 PRINTB(I);TAB(12)"MAAL X TOT DE MACHT";I;
250 IF I=0 THEN 270
260 PRINT" EN"
270 NEXT I:PRINT:PRINT" IS GELIJK AAN Y"
280 END
```



## Voorbeeld 42.1

### COEFFICIENTEN POLYNOMEN =====

TYP	HET AANTAL	X,Y	PUNTEN	5
PUNT 1	X,Y =	0		576
PUNT 2	X,Y =	1		326
PUNT 3	X,Y =	2		92
PUNT 4	X,Y =	3		-162
PUNT 5	X,Y =	4		-400

DE COEFFICIENTEN EN MACHTEN VAN X ZIJN:

3	MAAL X TOT DE MACHT 4	EN
-24	MAAL X TOT DE MACHT 3	EN
59	MAAL X TOT DE MACHT 2	EN
-288	MAAL X TOT DE MACHT 1	EN
576	MAAL X TOT DE MACHT 0	
IS GELIJK AAN Y		

## Voorbeeld 42.2

### COEFFICIENTEN POLYNOMEN =====

PUNTEN COLLOCEREN MET POLYNOOM  
DE ORDER V.H.POL.IS HET AANTAL  
PUNTEN MINUS 1.MAX ORDER 9

TYP	HET AANTAL	X,Y	PUNTEN	6
PUNT 1	X,Y =	1		8
PUNT 2	X,Y =	3		4
PUNT 3	X,Y =	2.5		5
PUNT 4	X,Y =	4		7
PUNT 5	X,Y =	8		10
PUNT 6	X,Y =	15		13.5

DE COEFFICIENTEN EN MACHTEN VAN X ZIJN:

.0225	MAAL X TOT DE MACHT 5	EN
-.6731	MAAL X TOT DE MACHT 4	EN
6.576	MAAL X TOT DE MACHT 3	EN
-25.6007	MAAL X TOT DE MACHT 2	EN
39.1164	MAAL X TOT DE MACHT 1	EN
-11.4412	MAAL X TOT DE MACHT 0	
IS GELIJK AAN Y		

# Appendix

In deze appendix zijn de volgende conversietabellen opgenomen:

Tabel 1: String-functies.

Tabel 2: Rekenkundige en logische functies.

Tabel 3: Statements.

De tabellen kunnen worden gebruikt bij het aanpassen van programma's die op een ander type computer zijn geschreven dan de in uw bezit zijnde computer.

Het gebruik van de tabellen gaat als volgt:

In het aan te passen programma ziet u een statement waarvan u denkt dat die niet op uw computer voorkomt. U zoekt de statement op in de tabellen, controleert op welke computers die statement voorkomt, en indien hij inderdaad op uw computer niet voorkomt dan kunt u in de kolom 'overeenkomstige statements' vinden of er vervangende statements zijn. Zo ja, dan kunt u de voor uw computer geldende vervangende statement in het aan te passen programma gebruiken.

Statements en functies die alleen op één computer voorkomen, en waarvoor geen vervangende statement of functie bestaat, zijn in deze tabellen meestal niet opgenomen.

De auteur van dit boek heeft getracht de programma's zoveel mogelijk in converteerbare statements te schrijven.

Tabel 1. String-functies.

string-functies	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige string-functies
\$ n									X	CHR\$(n)
ASC(str)	X	X	X	X	X	X		X		CH
CH									X	ASC(str)
CHR\$(n)	X	X	X	X	X	X	X	X		\$ (n)
CODE							X			
FRE('')					X					FRE(n\$)
FRE(n\$)			X			X				FRE('')
GET	X	X		X				X		INKEY\$
INKEY\$						X	X			GET
LEFT\$(str,n)	X	X	X	X	X	X		X	X	
LEN(str)	X	X	X	X	X	X	X	X	X	
MID\$(str,m,n)	X	X	X	X	X	X		X	X	
RIGHT\$(str,n)	X	X	X	X	X	X		X	X	
slicing							X			geeft LEFT\$, MID\$ en RIGHT\$ functies
STR\$(n)	X	X	X	X	X	X	X	X	X	
STRING\$(n, kar)					X	X				
VAL(str)	X	X	X	X	X	X	X	X	X	
VARPTR n		X			X	X				

Tabel 2. Rekenkundige en logische functies.

functies	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige functies
& <i>n</i>									X	HEX\$( <i>n</i> )
%									X	CINT, FIX
?FRE(0)	X									FRE
ABS	X	X	X	X	X	X	X	X	X	
ACS							X			
AND	X	X	X	X	X	X	X		X	
ASN							X			
ATN	X	X	X	X	X	X	X	X	X	
CALLM..., ( <i>x</i> )		X								LINK <i>n</i> , USR( <i>x</i> )
CDBL					X	X				
CINT					X	X				%
COS <i>n</i>	X	X	X	X	X	X	X	X	X	
CSGN					X	X				
CUR		X								POS
DEG									X	
ERL					X	X				
ERR					X	X				
EXP <i>n</i>	X	X	X	X	X	X	X	X	X	
FIX					X	X				%
FLT									X	
FRAC		X								
FRE		X	X	X	X	X		X		?FRE(0)
HEX\$( <i>n</i> )		X			X					& <i>n</i>
HTN									X	
INT	X	X	X	X	X	X	X	X		
LINK <i>n</i>									X	CALLM..., ( <i>x</i> ), USR( <i>x</i> )
LN							X			LOG
LOG	X	X	X	X	X	X		X	X	LN
LOGT		X								
MOD	X	X			X					<i>x</i> % <i>y</i>
NOT	X	X	X	X	X	X	X			
OCT\$( <i>n</i> )					X					
OR	X	X	X	X	X	X	X		X	
POS	X		X	X	X	X				CUR
RAD									X	
RND		X							X	RND(0), RND( <i>neg</i> )
RND% <i>n</i>									X	
RND(0)			X	X	X	X	X	X		RND( <i>neg</i> ), RND
RND( <i>n</i> )					X	X				RND( <i>pos</i> )
RND( <i>neg</i> )	X									RND(0), RND
RND( <i>pos</i> )	X									RND( <i>n</i> )
SCRN( <i>x</i> , <i>y</i> )	X	X								
SGN	X	X	X	X	X	X	X	X	X	
SIN <i>n</i>	X	X	X	X	X	X	X	X	X	
SPC <i>m</i>	X	X	X		X					
SPC <i>n</i>				X				X		
SQR( <i>n</i> )	X	X	X	X	X	X	X	X	X	
SWAP					X					
TAN( <i>n</i> )	X	X	X	X	X	X	X	X	X	
USR( <i>x</i> )	X		X	X	X	X	X	X		CALLM..., ( <i>x</i> ), LINK <i>n</i>
<i>x</i> % <i>y</i>									X	MOD



Tabel 3. Statements.

statements	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige statements
\$									X	DEFSTR, IMPSTR
%									X	DEFDBL
?m=n									X	POKE m,n
?n									X	IN(poort), IN #n, INP, INP(n)
AT							X			CURSOR m,n
BEEP x,y									X	SOUND
CALL	X									CALLM, SYS, LINK
CALLM		X								CALL, SYS, LINK
CLEAR n									X	MODE n
CLEAR 0									X	MODE 0, TEXT, GR
CLOAD*					X					INPUT #n,m, RECALL, LOADA, LOAD, GET A
CLOSE				X				X		SHUT
CLS						X	X			HOME, PRINT \$12
COLOR n									X	HCOLOR = n
COLOR = n	X									COLOUR n
COLOUR n									X	COLOR = n
COUNT									X	
COUNT n									X	TAB
CSAVE*					X					SAVE A, STORE
CURSOR m,n		X						X		AT
DATA	X	X	X	X	X	X		X	X	
DEF FN (naam)	X			X	X			X		
DEFDBL					X	X				%
DEFINT					X	X				IMPINT
DEFSNG					X	X				IMPFPPT
DEFSTR						X				IMPSTR, \$
DIM var(n)	X	X	X	X	X	X	X	X	X	
DO...UNTIL									X	
DOT		X								SET (x,y), PLOT, HPLLOT, PLOT k,x,y
DRAW		X								
DSP n	X									
END	X	X	X	X	X	X	X	X	X	
ERROR					X	X				
EXT									X	
FILL		X								
FIN									X	OPEN
FOR...TO...STEP	X	X	X	X	X	X	X	X	X	
GET				X		X			X	
GET A									X	INPUT #n,m, RECALL, LOADA, CLOAD*, LOAD
GOSUB n	X	X	X	X	X	X	X	X	X	
GOTO n	X	X	X	X	X	X	X	X	X	
GR	X									CLEAR 0
HCOLOR = n	X									COLOR n
HOME	X									CLS, PRINT \$12
HPLLOT	X									DOT, PLOT, PLOT k,x,y
HTAB (n)	X									
IF...GOSUB n	X	X	X	X	X	X			X	
IF...GOTO n	X	X	X	X	X	X		X	X	
IF...THEN...	X	X	X	X	X	X	X	X	X	
IF...THEN...ELSE					X	X			X	
IMPFPPT		X								DEFSNG
IMPINT		X								DEFINT
IMPSTR		X								DEFSTR, \$
IN (poort)			X		X					IN #n, INP, INP(n), ?n
IN #n	X									IN (poort), INP, INP(n), ?n
INP		X								IN (poort), IN #n, INP (n), ?n
INP (n)						X				IN (poort), IN #n, INP, ?n
INPUT n							X			INPUT "...",n
INPUT "...",n	X	X	X	X	X	X		X	X	INPUT n
INPUT #n,m			X	X		X		X		RECALL, LOADA, CLOAD*, LOAD, GET A
LET	X	X	X	X	X	X	X	X	X	

Tabel 3. Statements (vervolg).

statements	APPLE II	DAI	EXIDY SORC.	PET/CBM	P2000	TRS80	ZX81	VIC20	ACORN ATOM	overeenkomstige statements
LINK									X	CALL, CALLM, SYS
LOAD		X					X			INPUT # <i>n,m</i> , RECALL, LOADA, CLOAD*, GET A
LOADA		X			X	X	X			INPUT # <i>n,m</i> , RECALL, CLOAD*, LOAD, GET A
LPRINT					X	X	X			PRINT \$2, PRINT =
MODE <i>n</i>		X								CLEAR <i>n</i>
MODE 0		X								TEXT, CLEAR 0
MOVE									X	
NEXT	X	X	X	X	X	X	X	X	X	
ON..ERROR GOTO <i>n</i>					X	X			X	ONERR
ON..GOSUB <i>n</i>	X	X	X	X	X	X		X	X	
ON..GOTO <i>n</i>	X	X	X	X	X	X		X	X	
ONERR	X									ON..ERROR GOTO <i>n</i>
OPEN				X				X		FIN
OUT ( <i>poort</i> , <i>n</i> )		X			X	X				PR # <i>n</i> , PRINT # <i>n</i>
PAUSE							X			WAIT
PDL	X	X								
PEEK ( <i>n</i> )	X	X	X	X	X	X	X	X		PRINT ? <i>n</i>
PLOT	X						X			SET ( <i>x,y</i> ), HPLOT, DOT, PLOT <i>k,x,y</i>
PLOT <i>k,x,yxy</i>									X	RESET ( <i>x,y</i> ), UNPLOT, SET ( <i>x,y</i> ), HPLOT, DOT
POINT						X				
POKE <i>m,n</i>	X	X	X	X	X	X	X	X		? <i>m,n</i>
PR # <i>n</i>	X				X	X				OUT ( <i>poort</i> , <i>n</i> ), PRINT # <i>n</i>
PRINT	X	X	X	X	X	X	X	X	X	
PRINT \$12									X	CLS, HOME
PRINT \$2									X	PRINT =, LPRINT
PRINT =				X				X		SAVEA, STORE, LPRINT
PRINT ? <i>n</i>									X	PEEK ( <i>n</i> )
PRINT @ <i>n</i>						X				
PRINT USING					X	X				
PRINT # <i>n</i>				X				X		OUT ( <i>poort</i> , <i>n</i> ), PR # <i>n</i>
PTR ( <i>n</i> )									X	
PUT									X	
READ	X	X	X	X	X	X	X	X	X	
RECALL	X									INPUT # <i>n,m</i> , LOADA, CLOAD*, LOAD, GET A
REM	X	X	X	X	X	X	X	X	X	
RESET ( <i>x,y</i> )							X			UNPLOT, PLOT <i>k,x,y</i>
RESTORE	X	X	X	X		X		X	X	RSTORE <i>n</i>
RESUME	X				X	X				
RETURN	X	X	X	X	X	X	X	X	X	
RSTORE <i>n</i>					X					RESTORE
SAVEA		X								STORE, PRINT =
SET ( <i>x,y</i> )						X				PLOT, HPLOT, DOT, PLOT <i>k,x,y</i>
SGET									X	
SHUT									X	CLOSE
SOUND		X								BEEP <i>x,y</i>
SPEED	X									
STOP	X	X	X	X	X	X	X	X	X	
STORE	X									SAVEA, PRINT =
SYS								X		CALL, CALLM, LINK
TAB	X	X	X	X	X	X	X	X		COUNT <i>n</i>
TEXT	X									MODE 0, CLEAR 0
TOP									X	
UNPLOT							X			RESET ( <i>x,y</i> ), PLOT <i>k,x,y</i>
VLIN..AT	X									
VTAB ( <i>n</i> )	X									
WAIT	X	X	X	X	X			X	X	PAUSE




## **BASIC voor beginners**

J.P.A. van Prooyen

BASIC is de meest populaire taal onder de microcomputergebruikers. Dit boek is bestemd voor diegenen die direct aan de slag willen. Het geeft geen ellenlange inleidingen over de werking van de computer of over moeilijke programmeer technieken, maar biedt direct de behandeling van de BASIC-instructies, afgewisseld met programmavoorbeelden.

94 pagina's, formaat 14,5 × 21 cm.  
ISBN 90 201 1257 0




## **BASIC en huiscomputers**

K.L. Boon

Dit boek maakt de potentiële koper of gebruiker van een microcomputer vertrouwd met de begrippen en eigenschappen van die computer. Omdat BASIC dé programmeertaal is voor huiscomputers, bevat dit boek tevens een zeer uitvoerige cursus in die taal.

Door de uitgewerkte programma-voorbeelden kan de lezer zijn kennis testen en zal hij gestimuleerd worden om zelf te gaan programmeren.

152 pagina's, formaat 14,5 × 21,5 cm.  
ISBN 90 201 1305 4





## **PASCAL voor iedereen**


K.L. Boon

Naast het populaire BASIC, als programmeertaal voor microcomputers, gaat ook PASCAL een goede toekomst tegemoet.

PASCAL heeft betere uitdrukkingsmogelijkheden, terwijl ook meer contrôlemogelijkheden aanwezig zijn.

Dit boek geeft een eerste introductie tot PASCAL. Geen moeilijke programma's dus, maar een eenvoudige behandeling, die iedereen de mogelijkheid biedt om de taal te leren.

168 pagina's, formaat 17 × 23,5 cm.  
ISBN 90 201 1425 5



## **PASCAL**


### **Inleiding tot gestructureerd programmeren**

W. Findlay/D.A. Watt

Dit boek behandelt de programmeermethoden die een programmeur in staat stellen een goed overzicht te behouden, ook wanneer het programma groot en ingewikkeld is. Deze methode van programmeren wordt aangeduid met gestructureerd programmeren.

De behandeling van dit gestructureerd programmeren gaat samen met een gedegen cursus in de taal PASCAL, waarbij ervan wordt uitgegaan dat de lezer nog geen voorkennis van PASCAL heeft.

401 pagina's, formaat 14,5 × 21 cm.  
ISBN 90 201 1473 5



## **BASIC-computerspellen**


Kluwer Software-reeks

M.Th.A.M. Vijftigchild

Dit boek is bedoeld om de computer zijn nut te laten bewijzen als ontspannend element.

De programma's in dit boek variëren van kort tot redelijk uitgebreid en hebben ondanks hun luchtig karakter toch een leerzame achtergrond. Door de gestructureerde opbouw is men namelijk snel in staat de werking te doorzien en eventuele aanpassingen naar eigen wens aan te brengen.

138 pagina's, formaat 17,5 x 25,5 cm  
ISBN 90 201 1601 0



### **Andere titels op het gebied van computers en automatiseren**

ISBN	AUTEUR	TITEL
1603 7	Blume/Dillmann	Automatiseren met robots
1593 6	Dirksen	Microcomputers
1332 1	Dirksen	Computeroriëntatie
1590 1	Doppenberg	Basic-programma's voor elektronici
1576 6	Heijer den/Tolsma	Datacommunicatie
1331 3	Lörincz	Fortran V
1424 7	Schutte	Basic-programma's voor huiscomputers
1599 5	Schutte	Basic-programma's voor wetenschap en techniek
1515 4	Sickler	ZX 81
1595 2	Sickler	VIC 20
1577 4	Smulders	Acorn Atom
1324 0	Teleac	Microprocessors deel 1
1369 0	Teleac	Microprocessors deel 2
1589 8	Tracton	Basic-subroutines
1519 7	Vijftigchild	Basic-programma's voor school en bedrijf
1604 5	Voorburg	Toepassingen en spellen voor de ZX 81
1477 8	Wilmink	CP/M operating system
1472 7	Wilmink/Boon	Microcomputergids '82
1128 0	Zaks	Microcomputers voor hobby en werk
1129 9	Zaks	Microprocessors; van chip tot systeem





de berekening nodig heeft op een duidelijke manier in het Nederlands aan u opvragen. Dit komt het bedieningsgemak zeer ten goede.

Terwijl de technici onder u met dit boek een groot aantal kant en klare probleemoplossingen in handen krijgen, laat dit boek de programmeurs zien wat de mogelijkheden van BASIC op het terrein van de wetenschap en techniek zijn.



---

H.H. Schutte

**BASIC-programma's  
voor wetenschap en techniek**

---

42 origineel Nederlandstalige  
programma's

---